
第６章　コマンドリファレンス

ADVFSCの文法

MPC-684に搭載されているADVFSC （アドバンス ト ・ フェージッ ク） はBASICと よ く似たインタプ リ タです

が、マシン制御用と して開発されたものでその文法・ コマンドなどの仕様も一般的なBASICとは幾分異なっ

ています。

使用できる文字

ADVFSCで使用できる文字は半角の英数字・記号・日本語です。コン ト ロールコードは使用できません。ADV

FSCで用いる演算子・変数名・数値と しての数字は全て半角文字です。 アルファベッ トは大文字、小文字の

どちらでも使えますが、コマンド ・関数は全て大文字に変換されます。日本語を使いコ メン ト を書く と読み

やすいプログラムにな り ます。

プログラムモードとダイレクトモード

ターミナルソフ トからコマンドを入力して直接MPCを動作させるこ とができます。 これをダイレク トモー

ド といいます。 コマンドの行頭に文番号をつける と MPCはプログラムと解釈してその行をメモ リーに格納

します。 これがプログラムモードです。 ほとんどのコマンドはどちらでも使用できます。

文番号

ADVFSCで使用できる文番号は1～65534までの整数です。 プログラムは1つのタスクの中では条件分岐など

が無い限り文番号の小さい順から実行されます。

一行の文字数

一行に書く こ とのできる文字列は文番号やコ メン ト文も含めて半角文字で78文字以内です。1行78文字を超

えない範囲でコロン （:） で区切るこ とによ り一行に幾つも文を書く こ とができます（これをマルチステート

メン ト と言います）。

一文の文字列数

1つの文に書く こ とのできる文字列は8個以内です。

引き数が幾つも指定できるコマンドや、 論理結合でも8個を超えるこ とはできません。

#ON 0
#PRINT SW(192) 直接コマンドを実行します。
#A=10

#10 ON 0
20 PRINT SW(192) プログラムとしてメモリーに格納されます。
30 A=10

10 ON 0 : ON 1 : ON 2
文

文番号

 78文字以内

10 ON 0 1 2 3 4 5 6
一文の文字列は8個以内

コメント文

シングルクォート （'） をつける とその後ろは実行されません。 1行をそっ く り コ メ ン トにした り、実行文の

後ろにコ メン ト を書いた りするこ とができます。コ メン ト文の文字数は半角で37文字以内です。日本語も使

えます。 コ メン ト文の中にコロン （:） やカンマ （，） は使用できません。 コロンはデリ ミ タ と して扱われそ

れ以降は別の文と して解釈されます。 カンマはスペースに置き換えられます。

ラベル

GOTO、GOSUBの飛び先、FORKの実行位置などは全てラベルで記述します。 ラベルには次の制約があ り ま

す。

･ ラベルは先頭にアスタ リ スク （*） の付く文字列です。

･ ラベルは行の先頭になくてはなり ません。

･ ラベルは半角37文字以内です。

･ ラベルは重複はできません。

ラベルには半角英数字の他、日本語も使用できます。LISTコマンドやRUNなどの引き数にも ラベルを使う こ

とができます。 VLISTコマンドで使用状況が見れます。

10 FORK 1 *task1
20 GOSUB *case1
1000 *case1
1010 FOR i=0 TO 255
        |
LIST *case1

コマンド

ADVFSCの基本はコマンド と引き数です。コマンドは、必ず大文字のアルファベッ ト と数字になっています。

小文字で入力しても自動的に大文字に変換されます。コマンドには必ず引き数が必要です。引き数の数はコ

マンドによ り異な り ますが最大で7個まで定数、変数、関数、代入式、式で与えるこ とができます。IF～THEN

～END_IF、 DO～LOOP、 WHILEなどの制御文が要求する引き数は条件式と呼ばれる特別な式です。

ON 0
ON a
ON A=B+C
PRINT SW(192)
PRINT A$
PRINT 123+&HFF 

関　数

関数は引き数に対してある処理をして数値を与える働きをします。 MPC の代表的な関数に SW(n) があ り ま

す。SW(n)はnポートの状態を読み取ってON状態なら1、OFF状態なら0を与えます。関数の引き数には変数、

配列変数、 定数、 式が使えます。

A=SW(0)
A=SW(B)
A=SW(B+C+D)
PRINT SW(0)

10 ' この行はコメント文です。
20 ON 0 '出力0をオン
100 'コメント :ON 10

コ メン トにならず実行されます。

変数・配列変数

9文字以内の文字列で表現されます。 変数の大きさは4byte整数で-2147483647～2147483647の範囲を表現す

るこ とができます。変数は2000個まで定義するこ とができます。特別な変数と して配列変数があ り ます。配

列はDIMコマンドによって定義されます。 確保できる配列は15配列までで、 データの総和が10000個までで

す。DIM a(100)と定義する とa(0)～a(99)まで確保されます。変数名、配列変数名で使用できる文字は半角英数

字、 さ らに最初の文字はアルファベッ トに限り ます。

ローカル変数

語尾に ‘！’ マークを付けた変数はローカル変数と して扱われます。ローカル変数とはタスク単位の変数で

す。同じ名前でもタスク毎に違う メモ リエ リ アに割り当てられるので、１つのサブルーチンを複数のタスク

で共有するこ とができます。

A!=B!+C! /* ‘!’を付ければローカル変数

文字列変数

変数名の語尾がダラー （$） の変数は文字列変数です。文字列変数には文字列を格納するこ とができます。文

字列は80文字長で32個まで使用できます。

DO
  INPUT str$
  PRINT str$
LOOP

文字列変数名で使用できる文字は半角英数字、 さ らに最初の文字はアルファベッ トに限り ます。

定　数

定数にはプログラムに記述された数値とCONSTコマンドで定数化された変数とがあ り ます。 数値と しては

自然数型、 ヘキサ型 （16進）、 バイナリ型 （2進） があ り 16とい う数は次の様な記述になり ます。

自然数型 16

ヘキサ型 &H10

バイナリ型 &B10000

定数化変数は定数をシンボルと して扱う もので、 次の様に使います。

CONST sol 1
CONST sensor 192

このコマンドを実行する とその変数は定数とみなされて数値を変更しよ う とする とエラーメ ッセージが表

わされます｡プログラム中での再定義もできません。 定数化変数名で使用できる文字は半角英数字、 さ らに

最初の文字はアルファベッ トに限り ます。

文字列定数

ダブルクォート （"） で囲まれた文字列は文字列定義です。

str$="ACCEL"
str$=str$+"CORP"
str$=""

変数・配列変数・文字列変数・定数・文字列定数の参照と初期化

変数・配列変数・文字列変数・定数・文字列定数は、VLISTコマンドによ り使用状況を見るこ とができます。

一度定義したものはNEWコマンドでク リ アされます。 プログラムのロード時にはディ スクから読み込む前

にNEWを実行します。

代入式

代入式の書式は

変数＝式　 または　変数　 または　定数　 または　関数

です。

A=&H0F|&HF0
B=C+D
E=IN(24)

ADVFSCでは代入式を引き数と して扱う こ とができるため、独立した計算を減らすこ とができます。つま り

計算をしながらコマンドを実行できる とい う こ とです。例えば次のONコマンドのよ うに、 Aの値を1つずつ

加算しながらポート をONするこ とができます。

  A=0
DO
  ON A=A+1
LOOP UNTIL A==10

式

式とは関数、定数、変数を演算子で結び付けたものをいいます。ADVFSCの演算子には次のものがあ り ます。

式には、 括弧 （） も許されており、 通常の算術的なルールを守っています。

a=3000:b=4000:c=30:d=40

PRINT res=SQR(SQ(a)+SQ(b))-SQR(c*a+d*b)

条件式

条件式はIF文、 DO～LOOP文、 WHILE～WEND文で使用される条件を判断する式です。 条件式はAND,OR,

XORの論理結合式で結び論理演算するこ と もできます。

A,Bが式でも使用するこ とができます。 次は実際の使用例です。

演算子 書　式 内　容

==

<>

>=

>

<

<=

A==B

A<>B

A>=B

A>B

A<B

A<=B

AとBが等しい

AとBが等し くない

AがBよ り大きいかまたは等しい

AがBよ り大きい

AがBよ り小さい

AがBよ り小さいかまたは等しい

※印については特殊演算子の項を参照して ください。

演算子 演算内容 書式例

+

-

*

/

%

&

¦

¥

；

̃

,

加算

減算

乗算

除算

剰余 （モジュロ）

論理積 （and）

論理和 （or）

排他的論理和 （xor）

シフ ト  (※)

ワード ロード (※)

ワード ロード (※)

A=B+C

A=B-C

A=B*C

A=B/C

A=B%C

A=B&C

A=B¦C

A=B¥C

A=B;C

A=B̃C

A=(B,C)

A B A&B

0 0

0 1

1 0

1 1

0

0

0

1

A B A¦B

0 0

0 1

1 0

1 1

0

1

1

1

A B A¥B

0 0

0 1

1 0

1 1

0

1

1

0

① WHILE SW(i+192)==1 : WEND

② IF i>24 THEN
     WHILE SW(i+192)==0 : WEND
     ELSE
     TIME 100
   END_IF

③ IF IOR(&H1D0)<>tsk03 OR IOR (&H1D1)<>NOT(tsk03)
   GOTO *rr
   END_IF

④ t0=timer
   DO WHILE t0==timer
     SWAP
   LOOP

①i＋192で示されたポートがONとなっている間、 無限ループになり ます。

②iの値によって条件分岐をしています。

③論理結合の例です。 ORで結合されたどちらかの条件が成立すればGOTO *rrが実行されます。

④カレンダータイマーをポーリ ングして1秒待ちをしています。

ADVFSCでは数値演算の他に、文字列演算・文字列比較条件式なども扱う こ とができます。 これらは文字列

変数、 文字列定数、 あるいは文字列型関数と併せて使用します。

IF str$=="abc" THEN : GOTO *rr : END_IF
DO UNTIL A$==CHR$(&H0D)
  A$=INP$#0(1)
LOOP

パルス発生コマンドの停止条件と して等しい場合 （==） を与える時はプログラムを書く際に必ずイコール

（=） を2つ並べて下さい。

文字列の演算

文字列の演算には加算と代入がサポート されています。また、文字列を作り出す関数もいくつかあ り ますが

これらは文字列変数と同様に使用するこ とができます｡

A$="HEX=&h"+HEX$(16)
PRINT A$
HEX=&h10

条件式と しての文字列比較も可能で次の演算ができます。

演算子 書　式 内　容

==

<>

>=

>

<

<=

A$==B$

A$<>B$

A$>=B$

A$>B$

A$<B$

A$<=B$

A$とB$がアスキーコード と して等しい

A$とB$がアスキーコード と して等し くない

A$がB$よ り アスキーコード と して大きいかまたは等しい

A$がB$よ り アスキーコード と して大きい

A$がB$よ り アスキーコード と して小さい

A$がB$よ り アスキーコード と して小さいかまたは等しい

RMOV 500 1000 2000 UNTIL SW(192)=1

RMOV 500 1000 2000 UNTIL SW(192)==1

論理結合式

演算式は次のよ うな論理演算子を用いて複数の条件を組み合わせるこ とができます。

DO
LOOP UNTIL A==1 AND B==1

パルス発生の停止条件では論理結合ができません。

予約変数

time$ date$

タ ッチパネル(GPシ リーズ)のカレンダーの文字列です。time$が時間、data$が日付の文字列です。MBK-SH/RS

でGPと接続した時のみ有効です。

PR time$
12:24:31
PR date$
92/11/17

timer この変数は時間を全て秒になおした数です。 00:01:00の場合timerは60とな り ます。

SYSCLK パワーオンよ り5msec毎に+1される変数です。

TASKN タスクナンバーを知る変数です。

特殊演算子

特殊なデータ操作を行う演算子です。

【 ; 】シフト                     A=B;C

Bを8ビッ ト算術左シフ ト して、 Cのワード とORします。

用途)  MPG-314の論理演算

例)
#a=1;2
#prx a
 0102
#a=1;2;3;4
#prx a
 01020304

【 ~ 】ワードロード               A=B~C

Cを上位ワード、 Bを下位ワードに格納します。

用途) メモ リ IO、 MBKの型指定

例)
#a=1~2
#prx a
 00020001

【 , 】ワードロード               A=(B,C)

Bを上位ワード、Cを下位ワードに格納します。関数や配列の二次元化に使用します。()の中のみ演算子と し

て機能します。

用途） MPG-314の論理演算、 二次元配列

AND

OR

かつ

または

MOVE 1000 1000 1000 UNTIL SW(192)==1 AND SW(-1)==1
結合不可
6-6



例)
#a=(1,2)
#prx a
 00010002

特殊な例ですが、 次のよ うにも使用可能です。

10 a=1
20 b=2
30 c=3
40 d=4
50 prx e=((a;b),(c;d))
run
01020304

コマンドリファレンスの書式

(1) コマンド、 関数の名前です。 入力は大文字アルファベッ トで行います。 小文字で入力しても自動的に

大文字に変換されます(予約変数を除く )。

(2) コマンド、 関数の分類です。

(3) コマンド、 関数の機能です。

(4) 引き数の与え方を表します。 []は省略可能です。 引き数が複数ある場合は1つ以上のスペースで区切り

ます。 

(5) コマンド、 関数についての内容や使用方法を解説しています。

(6) 関連するコマンド、 関数です。

ON   ←(1)
I/O  ←(2) 出力オン  ←(3)

■書　式

ON A1 [A2 A3 A4 A5 A6] ←(4)
　 　A1～A6:出力ポートナンバー

■解　説

 ・ ・ ・ ・ ・ ・ ・ ・ ・ ・ ←(5)
 関連:OFF,OUT ←(6)

MPG-314のコマンドに関して

概　略

MPG-314は、制御用コンピュータMPC-684用のパルス発生ボードです。 NOVA電子製MCX-314が搭載され4

軸分の独立したPGと直線補間・円弧補間ド ライバが一つシステムにま とめられたもので、多様かつ高度なパ

ルス発生が可能です。 従来のユーザにはMPG-68K/405ユーザが旧来のプログラムで最小限の修正でそのま

ま使用できるよ うになっています。

さ らに、細かい制御が可能な専用コマンド(MPG314ネ ィティブコマンド)も用意されています。専用コマンド

の中にはMCX-314に対するレジスタアクセスコマンド も含まれているため、 MCX-314のすべての機能を引

き出すこ とができます。MPG-314は同一システムに10枚まで使用するこ とができます。ボードはPGコマンド

で各タスクに割り当てて使用します。 アドレスはボード前面のDSW1の値を 10倍して&H400 を加えた値で

す。 DSW1=2の場合は&H420です。 割り当て方法は次の通りです。

PG &H400 : ボードアドレス&H400(DSW1=0)のMPG-314を現在のタスクで使用します。

PG &h410 15 : ボードアドレス&H410(DSW1=1)のMPG-314をタスク15で使用します。

免責事項

MPG-314 を使用したプログラムは複雑かつ多様な制御に対応できますが、 その多様性のため思わぬ誤作動

を引き起こす可能性もあ り ます。 製作されたプログラムはすべて実行して動作を確認の上実用に供して く

ださい。コマンドの仕様書のみを根拠に、動作するであろ う と机上で製作されたプログラムをそのまま適用

しないでください。たとえば、エラー処理の回復プログラムはコーディングされただけで一回も実行される

こ とな く実際の装置に組み込まれている場合があ り大変危険です。また、コマンドには多様な引数入力につ

いてバグ、弊害を発生させる可能性があ り ます。このバグについてもユーザプログラムの試験実行によ りバ

グ回避使用対応し、当方まで御指摘ください。当社ではバグの修正は責任を以って迅速に対応しますが、バ

グによる損害補償には応ずるこ とができません。

MPG-68K 互換コマンド

MPG-314は従来のソフ ト を変更なく使用できるよ うにするために、 ほとんどのMPG-68Kコマンドに対応し

ています。 このためMPG-68K(405)からMPG-314への移行は簡単です。

また、追加機能と してXYUが動作中でもZ軸の制御が独立してできるこ と、またパルス発生中でもHPT()/X(0)

～U(0)のボード上のIO関数、 コマンドが使えるよ うに改善されています。

注意事項と して、エラーフラグが立つよ う な入力を加える と、エラーメ ッセージを表示して制御インタプ リ

タが停止します。 このため、エラー回復操作などをソフ トで対応する必要のある場合は、互換コマンドを使

用しないか(専用コマンドで対応)、 エラー入力を接続しないでください。

互換性の注意

･ ア ド レスが&H400,410,420～＆H490となっています。PG &HE0などのMPG-68K/405アド レスでは使用でき

ません。

･ MPG-314のPULSEコマンドは405互換です。 このため、 パルスデューティの決定はできません。

･ GO,RMコマンドは同時4軸スタートのパルス発生ですが、直線補間にはなり ません。MPG-314では4軸直線

補間がハードの制約から不可能となっています。また、GO,RMを使用する場合は4軸のACCEL設定が同一

である必要があ り ます。

･ XPLS,WPLSは使用できません。 しかし、 ネィティブコマンドでは同等の機能を実現できます。

･ HOME,HOMZコマンドはS1側に近傍センサ、S2にZ相(C相)入力を想定しています。 また、近傍センサのみ

で原点復帰を完結する場合は、S2側のみ使用して ください。これ以外の複雑な原点復帰はMPG-314専用コ

マンドのHOMEを参照して ください。

･ MPG-405にあ りMPG-68Kに無いコマンド、 TR,SETXやACCELの405専用入力方式は使用できません。

互換性には注意しており ますが細部に仕様の相違やバグの可能性もあ り ます。 ユーザで動作を検証の上実

用に供して ください。

互換（従来）のパルスコマンドとの関係

次表はMPG-68K/405のパルス発生に関するコマンド ・ 関数です。

これらはMPCのパルス制御にとって基礎となるもので、 MPG-314でも これらを使用するこ とができます。

MPG-314専用コマンド も これらの互換コマンドを踏まえた上で使用して ください。例えば、単軸制御などM

PG-314機能を使う場合には、専用コマンド ・拡張ステート メ ン ト を使いますが、現在点の取得は表中のX～Z

を用います。

conventional commands

(*1)MPG-314ではDDA4軸制御がハード的に不可能なため疑似的に4軸同時パルス発生をしています。

主なコマンドの概要

詳細は各コマンド解説をご覧下さい。

■PG
タスク とMPGを引当てます。 タスクで実行された命令は引当てられたMPGに対して発行されます。

PGは全てのパルス発生コマンドに先立ち実行しなければなり ません。

例えば、 タスク0で "PG &HE4 2" または、 タスク2の中で "PG &HE4" とすれば、 タスク2で実行されるコマ

ンドはボードアドレス&HE4に設定されたMPGに対して発行されます。

ボードアドレスはMPG-68K/405は &HE0 &HE4 -- 、 MPG-314は &H400 &H410 -- とな り ます。

■ACCEL
加減速テーブル （最大スピード、 加減速領域パルス数、 立ち上がりパルス数） を作成します。

汎用コマンドでは台形駆動です。作成されたデータはMPG側で保持しており、MOVE ・ RMOV等の移動命令

に適用されます。

MPG-314に対しては軸選択、 S字加減速パラ メータが拡張されています。

ACCEL    ･･加減速ﾃｰﾌﾞﾙの作成
BSY      ･･ﾊﾟﾙｽ発生状態入力
CLRPOS   ･･現在位置ｸﾘｱ
CURPOS   ･･現在位置表示
FEDD     ･･ｽﾋﾟｰﾄﾞ設定
FEDH     ･･ｽﾋﾟｰﾄﾞ設定
FEDT     ･･ｽﾋﾟｰﾄﾞ設定
FEDZ     ･･ｽﾋﾟｰﾄﾞ設定
FEED     ･･ｽﾋﾟｰﾄﾞ設定
GO       ･･4軸同時ﾊﾟﾙｽ発生 (*1)
HOME     ･･原点復帰
HOMZ     ･･原点復帰
JMPZ     ･･ｹﾞｰﾄﾓｰｼｮﾝ移動
JUMP     ･･ｹﾞｰﾄﾓｰｼｮﾝ移動
LIMZ     ･･ｹﾞｰﾄﾓｰｼｮﾝ規制
MOVE     ･･XYU絶対座標移動
MOVZ     ･･Z絶対座標移動
P        ･･点ﾃﾞｰﾀ
PALET1   ･･ﾊﾟﾚｯﾄ宣言
PALET2   ･･ﾊﾟﾚｯﾄ宣言
PALET3   ･･ﾊﾟﾚｯﾄ宣言
PALET4   ･･ﾊﾟﾚｯﾄ宣言
PG       ･･PG宣言
PGSEL    ･･PGﾎﾞｰﾄﾞ選択

PL1      ･･ﾊﾟﾚｯﾄﾎﾟｲﾝﾄ
PL2      ･･ﾊﾟﾚｯﾄﾎﾟｲﾝﾄ
PL3      ･･ﾊﾟﾚｯﾄﾎﾟｲﾝﾄ
PL4      ･･ﾊﾟﾚｯﾄﾎﾟｲﾝﾄ
PULSE    ･･定速ﾊﾟﾙｽ発生
Q_PAUSE  ･･ｸｲｯｸﾎﾟｰｽﾞ
RANGE    ･･動作範囲の制限
RM       ･･4軸相対座標移動 (*1)
RMOV     ･･XYU軸相対座標移動
RMVZ     ･･Z軸相対座標移動
SET      ･･ｲﾝﾁﾝｸﾞ量設定
SETP     ･･点ﾃﾞｰﾀ設定
SETPOS   ･･現在位置変更
SHMZ     ･･原点復帰ﾓｰﾄﾞ設定
SHOM     ･･原点復帰ﾓｰﾄﾞ設定
STOP     ･･ﾊﾟﾙｽ発生停止
TEACH    ･･ﾃｨｰﾁﾓｰﾄﾞ
U        ･･U軸点ﾃﾞｰﾀ
X        ･･X軸点ﾃﾞｰﾀ
Y        ･･Y軸点ﾃﾞｰﾀ
Z        ･･Z軸点ﾃﾞｰﾀ
NEWP     ･･点ﾃﾞｰﾀ初期化
PLIST    ･･点ﾃﾞｰﾀ表示
HOUT     ･･ﾊﾟﾙｽﾎﾞｰﾄ汎用出力
HPT      ･･ﾊﾟﾙｽﾎﾞｰﾄﾞ原点ｾﾝｻﾎﾟｰﾄ入力

■FEED
ACCELコマンドで設定した加減速テーブルが富士山だとする と、その何合目まで登るかを決めるのがFEED

です。

"FEED 0" で頂上まで登り ます。 FEEDはACCEL設定後に実行します。

MPG-314では軸指定と速度指定の分解能が拡張されています。

■MOVE
XYU絶対座標移動です。 最も一般的で使用頻度の高い移動コマンドです。

"MOVE 1000 2000 3000"とすれば座標 X=1000 Y=2000 U=3000 へ移動します。

パラ メータには変数 ・定数 ・ ポイン トデータを与えるこ とができます。

絶対座標移動の314専用コマンド と しては "MOVL、 MOVS、 MOVT"があ り ます。

■RMOV
XYU軸相対座標移動で、 MOVEと同じ く使用頻度の高いコマンドです。

パラ メータで与えられたパルス数だけ現在位置から移動します。

パラ メータには変数 ・定数を与えるこ とができます。

相対座標移動の314専用コマンド と しては "RMVL、 RMVS、 RMVT"があ り ます。

■P(n),X(n),Y(n),U(n),Z(n)
点データを設定 ・参照する関数です。 P(n)は4軸分、 X(n)～Z(n)は各軸の位置を返します。

nには1～10000の点番号または0を設定します。

0は現在位置を表すもので、 "nowXpos=X(0)"など と軸ごとに現在位置を変数に読み込むこ とができます。

現在位置の設定は"SETPOS"です。

ティーチングポイン トは"PLIST(省略形PLS)"で一覧表示します。

また、 X(n)～Z(n)は配列変数と しても使えます(nは1～10000)。

MPG-314に対してはn=-1でエンコーダカウンタ値を返します。

■TEACH (省略形T)
FTMWでティーチングするコマンドです。 キーボードのX,Y,U,Zキーでジ ョ グ移動し、 Pキーで教示します。

    #T

     PG[0,400] X= -150 Y= 150 U= 0 Z= 0  dx= 50 dy= 50 du= 50 dz= 50

    点番号を指定して下さい　P100

テ ィーチモードはQキーで終了します。

MPGやモータ ド ライバの動作確認にも利用できます。

■HOUT
MPG-68K/405では汎用出力の制御。

MPG-314では汎用出力、 ド ラ イバSON信号の制御のほか、 レジスタ制御に用いられます。

■HPT(n)
MPG-68K/405では原点センサポートの読み込み、

MPG-314では原点センサ、 インポジシ ョ ン、 アラーム入力を一括で読込ます。

MPG-314 専用コマンド (314 ネィティブコマンド )概略

MPG-314を用いて新しい装置を設計製作する場合は、ネイティブコマンドを使用して ください。任意の軸の

組み合わせで二軸および三軸直線補間可能です。 また、 単軸パルス発生も軸ごとに独立して制御できます。

ネィティブコマンド と互換コマンドのもっと も大きな相違は、 ネィティブコマンドはコマンドの実行(MCX

-314へのデータ設定)を終える とただちにコマンドから抜け出すこ とです。このため、パルス発生中の条件停

止や速度変更などをインタプ リ タ上で記述するこ とができます。

10        PG &H410
20        ACCEL X_A 8000
30        STPS X_A 0
40        RMVS X_A 10000 /* MPG-314を設定するとただちにコマンドから出る
50        WAIT X(0)>5000 /*リアルタイムで現在位置の監視が可能
60        FEED X_A 200 /*途中で速度変更
70        TIME 1000
80        STOP X_A STP_D /*途中停止
90        WAIT RR(X_A)==0 /*動作完了検出
100       CP
#run
 X= 7309 Y= 0 U= 0 Z= 0
#        

制　約

しかし次のよ うな使い方はハードの制約上できません。

･ XY軸、 ZU軸などのよ うに二組の補間制御を同時に行う こ と。 同時でなければ可能です。

･ 連続補間の途中で軸の組み合わせを変更するこ と。

･ 絶対位置移動(MOVS,MOVLなど)は、軸が停止する前に実行できない。(現在位置が確定していない

ため)

･ 移動コマンド （パルス発生） は、エラー発生時にパルス出力を停止しますが、インタプ リ タは実行

継続とな り ます。 エラー発生時に実行をhaltするには、 ERR_PAUSEコマンドを使います。

互換コマンドとネィティブコマンドの混在

MPG互換モード コマンド と 314 ネィテ ィブコマンド混在は可能です。 MPG互換モード コマンドから , ネィ

テ ィブコマンドに移る場合はインタプ リ タの制御を開放しないので矛盾しませんが、 ネィテ ィブコマンド

から互換モードに移る場合は条件待ちが必要です。 

move 1000 1000 0 /* このコマンドはパルス発生終了まで抜けてこない。
movl 1000 0 0 1000 /* MOVL実行問題なし

movl 10000 10000 0 0 /* パルス発生終了を待たずに次ぎの行へ
wait rr(0)&(X_A|Y_A)=0 /* この条件待ちを外すと次のmove 0 0 0がただちに実行され
move 0 0 0    て誤作動する。

ビット定数

MPC-684では、常用される機能のためコマンドを理解しやすいものにと り ま とめていますが、細部の機能設

定も可能なよ うに、 レジスタ も直接アクセス可能と しています。

また、レジスタ設定は多様なビッ ト定数が必要とな り ますが、代表的なものを予約定数と してあらかじめ68

4に登録してあ り ます。

次はレジスタ関連の関数 ・ コマンドの使用例です。

･ X軸のパルス発生終了を待つにはには WAIT RR(X_A)=0 

･ X軸の発生中のパルスを停止しさせるには STOP X_A STP_D

･ Y軸の速度を変更するには FEED Y_A 128

･ Z軸のパルス出力をパルス/方向信号方式にする HOUT MD_DPLS (Z_A;NOP,2)

注:()中は使用コマンド

MCX-314 の重要なレジスタとアクセス方法 (参考 )

■読み取り用

RR0: パルス発生とエラーの監視

パルス発生中の監視とエラーの監視用のレジスタです。

関数RR(0)で参照するこ とができます。 下位4ビッ トがビジー、 次の4ビッ トがエラーです。

次のよ うに使用します。 エラーは次のRR1の上位8ビッ トの反映です。

･ パルス発生の終了待ち WAIT RR(0)&X_A=0  も し くは WAIT RR(X_A)=0

･ 他のタスクのRR0読み取り RR(&H410)

RR1:軸毎のエラーステータス1

X,Y,U,Zそれぞれか持つステータスです。 関数RR(軸定数,1)とい う形式で読み取るこ とができます。

このレジスタの上位8ビッ トはパルス発生がどのよ うに終了したかを表します。

RR()関数で読み取り ますが次のよ うな使用方法とな り ます。

･ X軸リ ミ ッ トエラーの参照 LMT=&h3000&RR(X_A,1)

NOP ノーオペレーション 軸選択のレジスタ設定に使用

VOID 入力無効定数

CLR_ERR  (HOUT) Wr0 コマンドエラー解除

STP_I,STP_D (STOP) Wr0 停止コマンド

X_A,Y_A,U_A,Z_A,ALL_A Wr0軸選択定数 ALL_AはACCEL等で全軸

CW,CCW 円弧補間方向指定    連続補間に使用します

DS_DACL,EN_DACL 自動減速無効、有効  連続補間に使用します

IN0_ON,IN1_ON,IN2_ON,
IN3_ON,IN0_OFF,IN1_OFF,
IN2_OFF,IN3_OFF(STOP,HOME)

wr1の設定パラメータです。停止入力をON/OFF状態で有効にします。HOME
X_A,STOP X_A のタイプのコマンドで使用します。

INP_ON,INP_OFF(INSET_314) MCX-314/wr2の設定パラメータです。インポジョンを有効にしてON/OFF
を決定します。

ALM_ON,ALM_OFF(INSET_314) MCX-314/wr2の設定パラメータです。アラーム入力を有効にしてON/OFF
を指定します。

LMT_ON,LMT_OFF(INSET_314) MCX-314/wr2の設定パラメータです。リミット入力のON/OFF設定

SLMT_ON MCX-314/wr2の設定パラメータです。RANGEコマンドによるソフトリミッ
トの有効化定数

MD_2PLS wr2の設定パラメータです。CW/CCWパルス発生指定

MD_DPLS wr2の設定パラメータです。DIR,パルス発生指定
6-13



RR2:軸毎のエラーステータス2

X,Y,U,Zそれぞれが持つエラー入力の参照レジスタです。 ALAMやLMT信号を直接参照するにはこのレジス

タを参照します。

RR3: 軸毎のエラーステータス3

X,Y,U,Zそれぞれが持つカウンタの比較情報です。MCX-314には軸ごとに比較用レジスタ と比較器があ り ま

す。 これによ り特定の場所で正確な信号を発生させるこ と も可能です。

■設定用

WR0:コマンドレジスタ

軸選択とMCX-314に対するコマンドレジスタです。 たとえば

HOUT X_A;STP_D

とい う記述では、 X_A;STD_D とい う演算によって&H0126 とい う数値が生成されますがHOUTはこれをWR0

への直接書き込みと して扱います。WR0は軸選択レジスタ と しても使用されるため、次のWR1,WR2への書き

込み時にもダ ミーライ トが必要とな り ます。

WR1:モードレジスタ1

主に IN0 ～ IN3 のド ラ イブ停止検出信号の設定と して使用します。 WR1 への書き込みは次のよ う なフォー

マッ トにな り ます。

HOUT 2 (X_A;NOP,1)

2はWR2に書き込む値でIN0がLOWで停止という条件です。()のの中の二番目の引数きX_A;NOPでWR0に対し

て軸設定をします。,の後ろの番号がレジスタ番号です。レジスタ番号にボードアドレスを加える と任意のア

ドレスのMPG-314にアクセスするこ とができます。

WR2:モードレジスタ2

リ ミ ッ ト検出やインポジシ ョ ン検出の設定を行う レジスタです。 しかしながら このレジスタにはパルス出

力モードの設定ビッ ト も含まれていますので注意して再設定して ください。

HOUT INP_ON+MD_2PLS (X_A;NOP,2)

こ こではINP_ON+MD_2PLSをWR2に設定する値と しています。 MD_2PLSはパルス出力モードをCW/CCW負

論理方式(デフォルト )に設定するビッ ト定数ですが、も し も この値を加えておかないとパルス出力のモード

設定フ ィルードがすべて0とな り、パルス出力が正論理に変わってしまいます。(X_A;NOP,2)はWR1の場合と

同様の入力でX軸選択でWR2を選んでいます。

WR3:モードレジスタ3

下位3ビッ トが加減速度のモード設定とな り ます。初期値と してこのレジスタは4(自動減速、対称減速、S字

加減速度有効)が設定されています。 このほかにはOUT4～7の出力モード設定があ り汎用出力と して使用す

るか、 位置信号の出力と して使うかを決定できます。

 浮動小数点演算(オプション コプロ演算)

MPC-684はオプシ ョ ンのコプロセッサを装着するこ とによ り浮動小数点演算(倍精度)が可能にな り ます。し

かしながらMPC-684のインタプ リ タ(ADVFSC)は整数形式しか取り扱う こ とができないため、浮動小数点演

算のサポートは一般の演算から切り離された独立した形式とな り ます。 コプロ演算で必要なコマンドは次

の四種類です。 また、 コプロ という単一のIOでの計算のためマルチタスクには対応できません(複数のタス

クからの同時実行はできません)。 このためコマンドsetf,getfによ りセマフォの獲得、開放を組み込んであ り

ます。

SETF fn1,fn2,fn3.. 変数、 定数内部をコプロに引き渡します。 (セマフォ獲得)

GETF fn1,fn2,fn3.. コプロのデータをMPCの変数に引き渡します(セマフォ開放)

CALF "式" コプロ演算

PRF fn1,fn2,fn3.. コプロ内部データの表示 fpの番号を指定します。

コプロ内部にはユーザで使用できる浮動小数点レジスタがFP0--FP4まであ り ます。このFP0～FP4を変数と

して連続演算に使用します。SETFはデータをこのFPnにデータを引き渡すコマンドです。与えられた引数は

順にFP0,FP1..に引き渡します。

SETF 0 3 4 /* FP0に0,FP1に3,FP2に4をセットします。

GETFはFPnからデータを取り出すコマンドで与えられた変数に順にFP0,FP1.. の値を取り出してセッ ト し

ます。 この時数値は整数化され丸められます。 注)

SETF 0 3 4 : GETF A B C : PR A B C

結果は0,3,4とな り ます。

CALFは引数を文字列であらわされた計算式です。 FPnおよび、 作業用のメモ リ変数M0-M9,N0-N9の間の演

算を実行します。 定数もそのまま記述するこ とができます。 括弧は一重のみ使用できます。、 */(乗除)は優

先されませんので （） で囲います。 Fnの指定は大文字でも小文字でも有効です。

SETF 0 3 4 : CALF "F0=F1*F1+(F2*F2)" : GETF A : PR A

結果は25とな り ます。

複数の演算を順次行う場合は、； で区切り ます。

SETF 0 3 4 : CALF "F0=F1*F1+(F2*F2);F0=qF0" : GETF A : PR A

結果は5とな り ます。

注)GETFの整数化丸めは、コプロが決定し10進数の四捨五入とは異なり ます。10進数の四捨五入によって整

数化するには、 GETF 45 A B等のよ うに最初の引数に45を与えて下さい。

'q'は単項演算子です。 単項演算子には次のよ うなものがあ り ます。

単項演算子:右のレジスタに対してそれぞれの単項演算を行います。

sin/cos などの三角関数はラジアンで扱われます。 このため度での扱いが必要な場合は変換が必要になり ま

す。 この時にr,dの単項演算子が必要にな り ます。 次の例ではSin(45°)を求めています。

SETF 0 45 : CALF "f0=rf1;f0=sf0" : PRF 0
# Fp0 7.0710678118654752e-001

定数の記述では17桁までの数値を扱う こ とができます。また少数点も記述できます。しかし、E-10とい う指

数表現を追加するこ とはできませんので、 0.00000123などの場合は、 M0=123/100000000と します。

calf "f0=1000.1234*10000" : getf A : pr A

結果は10001234です。

アキュムレータ

calfの演算は複雑な括弧や優先順位に対応できないために、演算を細かく区切る必要があ り ます。しかし、ア

キュムレータ とい う概念を用いる と連続演算を記述するこ とができます。 アキュムレータは演算の途中結

果がはいっているレジスタです。 たとえばcalf "f0=1+2+3"の場合1+2+3は一旦アキュレータの上に保持され

ており、アキュムレータから結果をf0に転送します。単項演算子は単独で用いる と このアキュムレータに作

用します。 アキュレータはprf 7で参照するこ とができます。

たとえば次二つの演算は、 まった く同じ意味で、 と もに結果は5にな り ます。

CALF "F0=q(b*3+b*4)
CALF "f0=b3+b4q"

下行の演算ではb3+b4と演算が完結した後に単独でqが現れています。 このためqはアキュムレータの値の平

方根をと り ます。

二次方程式2X*X+7*X+4=0の解は次のよ うに求めるこ とができます。 F1=a,F2=b,F3=cと しています。

10 a=2 : b=7 : c=4
20 SETF 0 a b c
40 CALF "F0=-4*F1*F3+bF2q+(-1*F2)/2/F1"
50 PRF 0
#run
Fp0 -7.1922359359558485e-001

こ こでは 二次方程式の一つの解F0=(-1*F2+root(F2*F2-4*F1*F3))/(2*F1)を実行しています。 この式には二

重括弧が含まれCALFでは計算できませんが、順序をおきなおすこ とによ り括弧を最小限にして演算してい

ます。注意と してbが2の場合この計算は正しい答えを返すこ とができません。これは平方根をと る中身が負

記号 意味 表記 演算内容

s sin sF0 F0=sin(F0)

c cos cF0 F0=cos(F0)

t tan tF2 F2=tan(F2)

a atan aF3 F3=atan(F3)

q √ qF0 F0=√F0

l (int) iF0 F0=int(F0)

p π pF6 F6=π

e e eF4 F4=e

z zero zF4 F4=0

b n*n bF3 F3=F3*F3

r radian rF4 F4=F4*π/180( fp5 使用)

d degree dF4 F4=F4*180/π (fp5 使用)

の値とな り解が虚数領域にはいるためです。 平方根演算は対象が正の値であるこ とを確認する必要があ り

ます。

計算事例

<<M変数を用いた例>>

40 CALF "M0=300;M1=400"
50 CALF "F3=bM0+bM1"
60 PRF 3
70 CALF "M2=b300+b400;F2=qM2+1"
80 PRF 2

次はsin/cosを求めるもので度－＞ラジアン変換を含みます。

90 'sin cos
100 SETF 0 0 30 60
110 CALF "F0=s(rF2)*1000;F1=c(rF3)*1000"
120 GETF sinv cosv
130 PRINT sinv cosv

次はatanでラジアン－＞度変換を含みます

140 ' atan
150 SETF 0 1
160 CALF "F0=d(aF1)*1000"
170 GETF atanv
180 PRINT atanv 

次は点p(1)をp(2)基点でdeg回転する演算です。

ラジアン変換とsin/cosに括弧を用いないのはこの方が高速演算だからです。 次の演算で2.3m秒です。

220 SETP 1 200 100
230 SETP 2 100 100
240 deg=45
250 SETF X(1) Y(1) X(2) Y(2) deg
260 CALF "M0=F0-F2;M1=F1-F3;M4=rF4c;M5=rF4s"
270 CALF "F0=M0*M4-(M1*M5)+F2"
280 CALF "F1=M0*M5+(M1*M4)+F3"
290 GETF xco yco
310 PRINT "x=" xco "y=" yco

組み込み関数について

MPC-684には次のよ うな組み込み計算コマンドがあ り ます。 いずれもコプロを使用しています。

形　式 計算内容 実行例 結　果

SIN A1 A2 ADR(A3) A3=sin(A1/10000)*A2 SIN 300000 1000 ADR(a) a<-500

COS A1 A2 ADR(A3) A3=cos(A1/10000)*A2 COS 600000 1000 ADR(b) b<-500

TAN A1 A2 ADR(A3) A3=tan(A1/10000)*A2 TAN 450000 1000 ADR(c) 1000 c<-100

ATAN A1 A2 ADR(A3) A3=atan(A1/1000)*A2 ATAN 1000 100 ADR(d) 4500 d<-4500

ATAN2 A1 A2 ADR(A3) A3=atan2(A1/A2)*10000 ATAN2 10000 10000 ADR(e) e<-450000

GETDG n m ADR(A3) A3= 成す角度(P(n)->P(m))*10000 SETP 100 0 0 0 0

SETP 101 1000 1000 0 0

GETDG 100 101 ADR(f)

f<-450000

AFFIN n m l deg l=回転(nをm中心でdeg/10000°) AFFIN 101 100 102 450000 p(102)<=0 1414 0 0

コマンドリファレンス

@
演算 論理否定

■書　式

@(n)

  n:変数、 定数

■解　説

MPCのSW入力はビッ ト値を返しますがNOT関数でこの論理否定をと る と1は&HFFFFFFFEとい う値になっ

てしまいます。 @は1と0を反転する関数です。

pr @(1)
0

?
I/O ビットの読み込み(HSW代替)

■書　式

?(n)

  n:変数、 定数

■解　説

入力ポートの複雑な論理をと る場合、 HSW(n) とい う記述は長すぎて式の長さが37文字以下という制限にす

ぐに抵触してしまいます。 ?は式を短く書くための代替関数です。

  s1=@(?(-1)¥?(-2)) は s1=&h1&NOT(HSW(-1)¥HSW(-2)) と同じです

ABS
演算 絶対値

■書　式

ABS(n)

■解　説

nの絶対値を返します。

#aho=ABS(-1000)
#PRINT aho
  1000

ACCEL
パルス（MPG-314専用） 加減速テーブルの作成

■書　式

ACCEL [n] [s] max [long min]

  n:軸選択予約定数 X_A～Z_A も し くは論理和。 省略時、 全軸に対する加速度、 速度設定

  s:S字加減速パラ メータ -1～-50。 省略時、 台形制御

  max:最大スピード 1～4Mpps(円弧2M)

  long:加減速領域パルス数

  min:立ち上がりパルス数

引数省略時 設定パラ メータ表示

■解　説

ACCELコマンドはこれまでmax、 long、 minパラ メータのほかに軸選択、 S字加減速の設定を追加するこ とが

できます。

引数を全て省略する と設定内容を表示します。

ACCEL実行後はFEED=0にな り ます。

ACCEL U_A -30 10000 /* U軸、S字30%、MAX10Kpps
ACCEL X_A|U_A -1 10000 /* XとU軸を同時に設定

※注意

･ minは1ppsから指定可能ですが微小距離移動等で著し く遅くなる場合があるのでむやみに10pps以下に設

定しないで下さい。

･ S字のパラ メータを大き く し過ぎる と加減速域が大き くな り、 タ ク ト タイム遅延の原因にな り ます。装置

の状況によ り異な り ますが、 大き くても20%～30%が妥当な範囲と思われます。

ACCEL
パルス（MPG-68K互換） 加減速テーブルの作成

■書　式

ACCEL max [long min]

  max:最大スピード

  long:加減速領域パルス数

  min:立ち上がりパルス数

    100≦min≦max≦100000

    long≦10000

■解　説

※パラ メーターを省略する と現在の設定値が表示されます。

※このパラ メーターの与え方はMPG-68K,MPG-405のどちらにも使えます。

■684のACCELでの加速距離 ・加速時間

pls

pps

long long

max (feed 0)

max (feed 64)

次の図は3kppsでスタート して15kppsまで加速 （24msec）全体は102msecの間に1250パルス移動する という も

のです。

例えば前記のよ うな条件が与えられている場合にACCELコマンドには次のよ うな計算によってパラ メータ

を設定します。

ﾊﾟﾙｽ数mは m=t*15k/2 より 時間tは　t=2m/15k

加速度は ｽﾋﾟｰﾄﾞ/時間 ですから　15k/t=15k/(2m/15k)=15k*15k/2m

加速度は500kですから　500k=15k*15k/2m

これよりtのあいだのﾊﾟﾙｽ数mは　m=15k*15k/(2*500k)=225ﾊﾟﾙｽ

同様にﾊﾟﾙｽ数 a は　a=3k*3k/(2*500k)=9ﾊﾟﾙｽ

ﾊﾟﾗﾒｰﾀｰとして与えるﾊﾟﾙｽ数は b ですから　b=m-a=216ﾊﾟﾙｽ

この結果よ りACCELは(ACCEL 15000 216 3000)と します。 このよ うにアクセルを設定した後で RMOV 1250

0 0とパルス発生すれば前の図のよ う な加減速移動します。 加速度と距離あるいは時間の関係は次のよ うに

な り ます。これは0から加速した距離Lを前提と しています。等加速の場合移動距離が三角形の面積となるた

めこのよ う な関係が成り立ちます。

加速度=(MAXSPD)*(MAXSPD)/(2*L)

加速時間=2*L/(MAXSPD)

移動距離=L=t*MAXSPD/2

    

■ACCELとFEEDの計算について
ソフ ト ウェアによる加減速テーブルの作成や中間のFEEDのテーブルには次の計算を実施しています。

ACCEL MAX L MIN を指定した場合のFEED nに対応する加速距離の計算

Ln=((128-k)*MIN+MAX*k)*L*k/(MIN+MAX)/4096   (k=64-n)

n= 0(k=64) Ln=L n=64(k=0 ) Ln=0

またpパルス目のタイマー値は次の式よ りn(p)を求めている。

f(p)=sqrt(min^2+(max^2-min^2)*p/L)

t(p)=16000000/f(p)=22*n(p)+120

この計算の元になるのは加速度α=(MAX^2-MIN^2)/2Lであるこ と、また距離とスピードの関係が次のよ うに

なるこ とから得られます。

p=1/2αt^2+MIN*t

f(p)=α*t

  tについて解いたpの関数をα*tに代入してf(p)を求めます。

15k

24msec

a
b

102msec
t

m=a+b

500000pps/s=(15k-3k)/0.024

(加速度)

a,b,cはパルス数(面積に相当)

L

t

■ACCEL実行スピード実測値
ACCEL コマンドの実行スピードは各パラ メーターの数値によ り変化します。 同じ最高スピードでも加速距

離や立ち上がりパルス数で変わり ます。 斜辺が滑らかな加減速にするほど演算に時間がかかり ます。

10         SYSCLK=0
20       ACCEL 50000
30         A=SYSCLK
40       ACCEL 
50       PRINT A*5 "msecです"
60     '
70         SYSCLK=0
80       ACCEL 50000 10000 100
90         A=SYSCLK
100      ACCEL 
110      PRINT A*5 "msecです"
120    '
130        SYSCLK=0
140      ACCEL 50000 1000 4000
150        A=SYSCLK
160      ACCEL 
170      PRINT A*5 "msecです"
180    '
190        SYSCLK=0
200      ACCEL 100000
210        A=SYSCLK
220      ACCEL 
230      PRINT A*5 "msecです"
240    '
250        SYSCLK=0
260      ACCEL 100000 10000 100
270        A=SYSCLK
280      ACCEL 
290      PRINT A*5 "msecです"
300    '
310        SYSCLK=0
320      ACCEL 100000 1000 5000
330        A=SYSCLK
340      ACCEL 
350      PRINT A*5 "msecです"
#RUN

最大スピード 50000  加速距離 2500  最小スピード 2500
805 msecです
最大スピード 50000  加速距離 10000  最小スピード 100
3155 msecです
最大スピード 50000  加速距離 1000  最小スピード 4000
335 msecです
最大スピード 100000  加速距離 5000  最小スピード 5000
1970 msecです
最大スピード 100000  加速距離 10000  最小スピード 100
4695 msecです
最大スピード 100000  加速距離 1000  最小スピード 5000
355 msecです

ADD_MBK
MBK-SH データインクリメント

■書　式 

ADD_MBK a dtadr

  a: 加算値

  dtadr: MBKデータアドレス

■解　説

dtadrの値にaを加算します。

#S_MBK 1 100
#PR MBK(100)
1
#ADD_MBK 1 100 /* DT100の値に1を足す
#PR MBK(100)
2

10 CONST counter1 100 /* カウンタの定義(ワード)
20 CONST counter2 102~Lng /* カウンタの定義(ロング)
30 CONST counter3 104~Int /* カウンタの定義(符号付ワード)
100 ADD_MBK 1 counter1 /* counte1 インクリメント
110 ADD_MBK 2 counter2 /* counter2 2加算
120 ADD_MBK -1 counter3 /* counter3 デクリメント

ADR
ユーザーコマンド アドレス取得

■書　式

usercom ADR(v) [A1 A2 A3]

  usercom:ユーザーコマンド

  v:変数 ・ 配列

  A1 A2 A3:ユーザーコマンド引き数

■解　説

ADR()は変数・配列のアドレスを得る関数です。 マシン語で作成したユーザーコマンドの結果をインタプ リ

タに戻すのに使用します。次の例ではマシン語で作成したプログラムをMPC-684にダウンロード してUSER

COM0に” adrtst” と して登録し、 プログラムでは結果を変数dataに格納しています。

マシン語プログラム

     TEXT
adrtst: 次のプログラムの・・・
     movea.l D0,A0 ・・・D0に変数dataのアドレスが入る
     add.l  D2,D1 ・・・IがD1に、I*10がD2に。それを加算しD1へ
     move.l D1,(A0) ・・・D1をD0の指すアドレス、つまり変数dataへ
     clr.l  D0
     rts
     end
-----------------------------
#comset 0 "adrtst" &hbf100
#comset
       USERCOMO -->ADRTST Entry Address->&H000BF100
10     FOR l=0 TO 5
20       adrtst ADR(data) l l*10 ・・・adrtstの結果をdataに格納
30       PRINT 
40     NEXT l
6-22



RUN
0
11
22
33
44
55

ALT 
I/O ON/OFF反転

■書　式

ALT A

  A:出力ポートナンバー

■解　説

I/O操作コマンドでI/O反転です。 ONされていたポート をOFFしOFFされていたポート をONします。

ALT 0 2 4
#ON 0
#OFF 2
#OFF 4
#ALT 0 2 4 <-- 0をOFF、2をON、4をON

1コマンドで設定できる出力は最大7つです。

AND 
演算 論理結合式(論理積) 

■書　式

（式） AND (式)

■解　説

式と式を論理積結合します。

IF SW(A)==1 AND SW(B)==1 THEN
  (制御文)
END_IF

DO
  (制御文)
LOOP UNTIL A==1 AND B==1

※注意

パルス発生の停止条件では論理結合はできません （下線部）

MOVE -- UNTIL A==1 AND B==1

AR$ 
文字列 文字配列

■書　式

AR$(n)

  n:配列数

■解　説

文字配列です。nはDIM_AR$のmで指定した値までと します（自動チェッ ク しないので注意して ください）。他

の文字列と同じよ うに扱う こ とができます。

【使用例】 長さ35byteの文字列 （Null含む） を4000個使用します。 この場合、使用できる最大点番号は1250ま

でにな り ます。

10        DIM_AR$ 35 4000 : '半角35文字 0～3999設定
20        FOR i=0 TO 3999
30            AR$(i)="ABC"+STR$(i*1000)
40        NEXT i
50        DIM_AR$  : '確認表示
60        FOR i=0 TO 3999
70          IF (i<5) OR (i>3995) THEN
80            PRINT i AR$(i)
90          END_IF
100       NEXT i
#run
Length=35 Count =4000 P(MAX)=1250
0 ABC0
1 ABC1000
2 ABC2000
3 ABC3000
4 ABC4000
3996 ABC3996000
3997 ABC3997000
3998 ABC3998000
3999 ABC3999000

ASC
文字列 文字からコードへ変換

■書　式

ASC(s$)

  s$:文字列

■解　説

文字列の先頭文字のアスキーコードを返します。

#LIST
10     a$="ABC"
20   IF ASC(a$)==&H41 THEN
30     PRINT "aho"
40   END_IF
50     a$="baka"
60   PRX ASC(a$)
#run
aho
 0062 <---bのアスキーコード
#

ATAN
演算 三角関数

COS参照

ATAN2
演算 三角関数

COS参照

BREAK
制御文 制御フロー終了 (IF文からBREAK) 

■書　式

BREAK

■解　説

一つ上位のIF文から直接抜け出ます。

IF --- THEN
  ---
ELSE 
  IF b==1 THEN 
    ---
    BREAK /* ▽へｼﾞｬﾝﾌﾟ
  END_IF 
  ---
END_IF 
(▽)

※注意

必ずEND_IFの直前に配置

BREAK
制御文 制御フロー終了 (繰り返し文からBREAK) 

■書　式

BREAK LOOP

BREAK WEND

BREAK NEXT

■解　説

DO-LOOP、 WHILE-WEND、 FOR-NEXTからの抜け出ます。

FOR i=1 TO 1000
  j=0
  DO UNTIL j==10
    j=j+1
    IF i==500 THEN
      BREAK NEXT /* ▽へｼﾞｬﾝﾌﾟ

    END_IF
  LOOP
NEXT i
(▽)

※注意

引数 LOOP,NEXT,WEND は予約定数と して定義されているので大文字で間違いのないよ うに設定してくだ

さい。

BSY
パルス（MPG-68K互換） パルス発生状態入力

■書　式

BSY （n）

  n:タスク番号

    -1,0≦n≦31

■解　説

BSY関数はMPGの動作状態を与えます。nの値が-1の場合は自己タスク、その他の値であればそれぞれのタス

クのMPGの動作状態です。 返される値は次の通りで、 それぞれ次のよ う な意味を持ちます。

BSY(n)==0 MPG動作中

BSY(n)==1 MPG正常停止

BSY(n)==256 MPG減速停止

BSY(n)==512 MPG即停止

BSY(n)==1024 DS_MPG

     PG &HEO 1 ’タスク1にPGボードを選択
     FORK 1 *task1
     DO
'
     WAIT BSY(1)==1 ’MPG動作待
'
       a=0
      DO UNTIL a==1 OR a==2
        INPUT”1か2のキーを押して下さい”a
      LOOP
      IF BSY (1)<>1 THEN
        STOP a 1
        SELECT_CASE BSY(1) ’MPGの停止
           CASE 256 : PRINT ”減速停止”
           CASE 512 : PRINT ”即停止”
           CASE_ELSE  : PRINT ”何かへん”
        END_SELECT
        ELSE
        PRINT ”もう止まってるよ”
     END_IF
     LOOP
'
'    ==============================
*task1
     ACCEL 5000
     FEED 0
     SETP 0 0 0 0 0 ’現在の点を0 0 0 0とする
     DO
      MOVE 50000 50000 50000
      TIME 2000
      MOVE 0 0 0
      TIME 2000
     LOOP

CALF
コプロ演算 演算

■書　式

CALF "式"

  引数は文字列

■解　説

FPnおよび、 作業用のメモ リ変数M0-M9の間の演算を実行します。

定数もそのまま記述するこ とができます。

括弧は一重のみ可。 */(乗除)は優先されないので()で囲います。

Fnの指定は大文字でも小文字でも有効です。

SETF 0 3 4
CALF "F0=F1*F1+(F2*F2)"
GETF A
PR A
  結果は25

例)

複数の演算を順次行う場合は、 ;で区切る。

SETF 0 3 4
CALF "F0=F1*F1+(F2*F2);F0=qF0"
GETF A
PR A
  結果は5

※注意

'q'は単項演算子。

CASE
制御文 多値分岐

SELECT_CASE参照

CASE_ELSE<CEL>
制御文 多値分岐

SELECT_CASE参照

CH_MOVS
パルス (MPG-314専用) MOVS到達点途中変更

■書　式

単軸指定

CH_MOVS n c

  n:軸指定予約定数 X_A～Z_A

  c:座標値

複数軸指定

CH_MOVS x y u x

  x y u x:各軸座標値。 指定無=VOID予約定数。

■解　説

パルスが終了している場合には継ぎ足し移動します。MOVSと必ず対で使用します。RMVSのあとではま と も

に動作しません。

下記は、実験と して5000パルスを超えたら行き先を変更しています。その時、既にそのポイン ト を超えてい

た り、 減速域がない場合は急停止にな り ます(①)。 パルス量が伸びる場合は、 減速停止になり ます②。

MOVS X_A 10000
WAIT X(0)>5000
'CH_MOVS X_A 5000  /* ①この場合、急停止になる
CH_MOVS X_A 15000  /* ②この場合、減速停止になる
WAIT RR(X_A)=0
CP

複数軸指定の場合は次のよ うな記述です。

MOVS 10000 10000 VOID VOID
WAIT X(0)>5000
CH_MOVS 15000 13000 VOID VOID
WAIT RR(ALL_A)=0
CP

CHR$
文字列 コードから文字へ変換

■書　式

CHR$(n)

  n:1byte長数

■解　説

nをアスキーコードに見なした文字を返します。 次の例ではAを表示します。

PRINT CHR$(&H41)
PRINT CHR$(65)
PRINT CHR$(&B01000001)

CLRPOS<CLP>
パルス（MPG-68K互換） 現在位置クリア

■書　式

CLRPOS

■解　説

現在位置のク リ アです。 XYUZ各値が強制的に0と されます。

10   SHOM &H15 100 　’原点復帰動作設定
20   HOME &H15 100 100 100 　’メカ原点
30   MOVE 500 500 500 　’電気原点
40   CLRPOS 　’現在点をクリアする。
50   DO
60     PRINT ”ここは” P(0)
70     MOVE 1000 1000 1000
80     PRINT ”ここは” P(0)
90     MOVE 0 0 0
100  LOOP

CMND
MPG-3202 X3202コマンド実行

■書　式

CMND code

  code:X3203アド レス と命令コードの和。

■解　説

MPG-3202のパルス発生IC 「X3203」 の命令を実行します。

MPG-3202関係コマンド ・ 関数:CMND,REG,REG3,ST_REG

参照: 「MPG-3202 製品別マニュアル」　 ･･･web参照または営業係へ請求下さい。

CMND &H106 /* X3202#1のｺﾏﾝﾄﾞ6実行

CNFG#0
RS-232 通信モード設定

■書　式

CNFG#0

  CH0(J2)の通信モード設定です。

CNFG#2

  CH2(J1)の通信モード設定です。

各パラ メータは次の文字列から選択し次のフォーマ文字列定数と して与えます。

CNFG#n "[BAUD][WORD][PRTY][STOP][XCNT]"

[BAUD] 19200,9600,4800,2400,1200,600

[WORD] b8,b7 （8ビ ッ ト ・ 7ビッ ト ）

[PRTY] pe,po,pn （偶 ・ 奇 ・ 無）

[STOP] s1,s2 （1ビ ッ ト ・ 2ビッ ト ）

[XCNT] XON,NONE （有り ・ 無し）

＊パラ メーターを省略する と現在の設定値が表示されます。

■解　説

#cnfg#0
CNFG#0 "9600b8pns1NONE"
#cnfg#2
CNFG#2 "9600b8pns1XON"
#
10 CNFG#0 "9600b8pns1NONE"
20 CNFG#2 "9600b8pns1XON"
#

CNFG#n はRS-232ポートの通信条件の設定とバッファのク リ アーを行います。

PRINT#n の直後に CNFG#n する と送信途中でバッファの内容がク リ アされてしまいます。 次のプログラム

の70行から *RXTASK に戻すと CNFG#n が LOOP されて不具合が生じます。

10      *RXTASK
20        CNFG#0 "2400b8pns1NONE"
25      *RXTASK1
30          A$="" :   B$=""
40        TIME 50
50        INPUT#0 B$
60        PRINT B$
70        IF B$=="TESTTEST" THEN : PRINT#0 "1234¥n" : GOTO *RXTASK1 : END_IF 
80        END 
#

CNFG#2
RS-232 通信モード設定

CNFG#0参照

CNT
デバッグ 実行継続

■書　式

CNT [n]

  n:ブレークポイン ト

■解　説

ブレークポイン トで停止したプログラムの再開。 nは次のブレークポイン トで略すとブレーク無しです。

COMSET
ユーザーコマンド コマンド名設定

■書　式

COMSET n s$ [adrs]

  n:ユーザーコマンド番号

    0≦n≦9

  s$:コマンド名

  adrs:実行番地

＊パラ メーターを省略する と現在の設定値が表示されます。

■解　説

USERCOM0～USERCOM9のコマンド名と対応するマシン語プログラムの配置位置を定めます。 adrsを省略

する と、 コマンド名だけが登録されマシン語プログラムは実行されません。また、COMSETする と登録され

たコマンドを一覧表示します。

CONST
演算 変数の定数化

■書　式

CONST v n

  v:変数名

  n:数値

■解　説

変数の定数化コマンド。おもにI/Oのシンボル化に使用します。定数化される とデータを変更するこ とはで

きません。 変数に戻す時はNEW語プログラム再編集するかロード して下さい。

10   CONST SOL1 0
20   CONST SOL2 1
30   CONST SOL3 2
40   ON SOL1 SOL2 SOL3 ’変数不可の例です。
50     SOL1=1
#RUN
#50 ・・・定数は変更できません

ON 0をON SOL1と記述できます。

CONT
タスク操作 タスク継続

■書　式

CONT A

  A:タス ク番号

■解　説

PAUSEされたタスクの再開。

       FORK 1 *task1
       DO
        WAIT SW(192)==1
        PAUSE 1
        WAIT SW(192)==0
        CONT 1
       LOOP
*task1
       DO
        ON 0
        TIME 100
        OFF 0
        TIME 100
       LOOP

COS
演算 三角関数

■ユーザコマンドを利用した、 三角関数のサポート
三角関数SIN/COS/TAN/ATANをユーザコマンド と して使用するこ とができます。 この演算は68000用コン

パイラのライブラ リ コールとなっているため、マルチタスクでは使用できません。各演算にはマルチタスク

の停止が含まれていますので、 注意して使用して ください。引き数は整数しか扱う こ とができませんので、

入力で10000倍の度、 第二引き数を出力する数のスケールファ ク ターと しています。

sin A1 A2 adr(A3) A3=sin(A1/10000) *A2 引き数は度です

cos A1 A2 adr(A3) A3=cos(A1/10000) *A2 引き数は度です

tan A1 A2 adr(A3) A3=tan(A1/10000) *A2 引き数は度です

atan A1 A2 adr(A3) A3=atan(A1/1000) *A2 結果は度です。

atan2 A1 A2 adr(A3) A3=atan2(A1/A2) *10000 結果は度です。

SIN 300000 1000 ADR(A)
PR A
 500
ATAN 1000 1000 ADR(A)
PR A
 45000

斜辺の長さ と高さから角度を求めるには

B=sqr(L^2-H^2)

から

θ=arctan(H/sqr(L^2-H^2)

で得られます。 MPC-684ではatan2で計算できます。

atan2 A1 A2 adr(A3)

A1:高さ

A2:斜辺の長さ

A3:戻り値が格納されている変数 （単位:度）

ATAN2のプログラム例

10          H=10000
20      'L=14142 '45度
30      'L=11547 '60度
40          L=20000 '30度
50          X1=L*L
60          X2=H*H
70          X3=X1-X2
80          X4=SQR(X3)
90        ATAN2 H X4 ADR(A)
100       PRINT A
RUN
300007 ←10000倍されています。（単位:度）

ATAN2の実行スピード

10          H=10000
20      'L=14142 '45度
30      'L=11547 '60度
40          L=20000 '30度
50        SINGLE 
60          OT=SYSCLK

y

xB

θ

L
H

70        FOR C=1 TO 1000
80            X1=L*L
90            X2=H*H
100           X3=X1-X2
110           X4=SQR(X3)
120         ATAN2 H X4 ADR(A)
130       NEXT C
140         NT=SYSCLK
150       MULTI 
160       PRINT (NT-OT)*5 "msec"
170       PRINT A
RUN
5790 msec ←結果=1000回で5.8秒
300007

CURPOS<CP>
パルス（MPG-68K互換） 現在位置表示

■書　式

CURPOS　 （省略形　CP）

■解　説

現在位置を表示します。 現在位置の表示方法は次にもあ り ますがCP最もシンプルです。

T （テ ィーチモードで現在位置を確認できる）

PR P(0) （PRINT文で現在位置を表示する）

SETP 0 （SETPコマンドで0を指定する）

#SETP 0 100 200 300 400 現在位置の設定
#CP 表示
X=100 Y=200 U=300 Z=400

date$
カレンダー 日付文字列取得

■書　式

date$  (小文字で書いてください)

■解　説

MBK-SHまたはMBK-RSとデジタルGPが接続されている とGPの年月日が入り ます。

#d$=date$
#pr d$
2001/10/23

DELETE<DEL>
編集  プログラムの削除

■書　式

DELETE n [m]

  n,m:文番号

■解　説

指定番号プログラムの削除

DEL 20 20番削除
DEL 20 40 文番号20から文番号40を削除する

DIM
演算  配列宣言

■書　式

DIM A(n)

  A:配列名

  n:配列サイズ

■解　説

配列宣言。宣言可能な大きさはmモデル(出荷時)で10000データです。宣言された配列の総和がこれを越える

と実行時にエラー表示されます。 また、 定義できる配列は15個までです。

DIM AHO(100)　BAKA(20)

AHO(0)～AHO(99)、 BAKA(0)～BAKA(19)の配列を宣言しました。

DIM
演算 配列宣言(二次元)

■書　式

DIM A(m,n)

  A:配列名

  m,n:配列サイズ

■解　説

MPC-684では二次元配列宣言が可能です。

DIM a(2,3)
a(2,1)=100
PRINT a(2,1)

※注意

扱いは二次元でも中身は一次元です。

  A(0,0)=A(0)
  A(0,1)=A(1)
  A(0,2)=A(2)

DIM_AR$
文字列 文字列配列宣言

■書　式

DIM_AR$ n m

  n:文字長

  m:確保数

■解　説

文字列配列を定義します。長さnの文字列をm個P(13000)側から確保します。引数をいれないと現在の設定値

と使用可能な点の最大指定値を表示します。

AR$参照

DIMCPY
文字列 配列データのコピー

■書　式

DIMCPY arrys(n) arryd(m) c

  arrys(n):コピー元配列、 点データ。 nはコピー先頭位置

  arryd(m): コピー先配列、 点データ。 mはコピー先頭位置

  c: コピー数 1～5000

DIMCPY var arryd(m) c

  var:変数、 定数

  arryd(m): コピー先配列、 点データ。 mはコピー先頭位置

  c: コピー数 1～5000

■解　説

点データや配列データを一括複写します。方向はarrys(n)からarryd(m)へとな り ます。複写数の上限チェッ ク

は行き先配列に則してしており ませんので、 不用意に大きな数をcに指定しないでください。 配列や点デー

タエ リ アを越えて複写する とデータが破壊されます。

arryは点データを含みます。

varを指定する とvarの値にしたがって同じ値で配列を埋めます。

DIMCPY A(1) X(10) 10   /* A(1)～A(10)の値をX(10)～X(19)にコピーする。

DO..LOOP
制御文 繰り返し

■書　式

DO [UNTIL/WHILE]

LOOP [UNTIL/WHILE]

■解　説

DOとLOOPの数は同じで対をなしていなければなり ません。UNTIL、WHILEをDOの後ろに置くか、LOOPの

後ろに置くかでDO～LOOPを実行する回数が違います。 注意して下さい。

DO～LOOP使用例

DO
  DO
    FOR I=0 TO 47
      ON I
      TIME 50
      OFF I
      TIME 50
      IF A==0 THEN : LOOP  : END_IF
  NEXT I
LOOP

DUMP
デバッグ メモリ表示

■書　式

DUMP [adr]

  adr:ア ド レス

■解　説

メモ リの内容を表示します。

ELSE
制御文 条件分岐

IF参照

END
制御文 プログラムの停止

■書　式

END

■解　説

実行停止。実行中のタスクがENDにぶつかればそのタスクは自動的に停止されます。 メ インタスクではRUN

を終了しコマンドモード とな り ます。ENDで停止されたプログラムはMONによる表示で 「停止」 と表示され

ます。
6-36



END_IF <EIF>
制御文 条件分岐

IF参照

END_SELECT <ESL>
制御文 多値分岐

SELECT_CASE参照

ERASE
メンテナンス フラッシュ ROMのプログラム消去

■書　式

ERASE

■解　説

初期化コマンドです。 フラ ッシュ メモ リ （FROM） 内のプログラムをク リ アします。

ダイレク ト コマンドのみの使用です。

#ERASE
*
#

MPCINIT参照

ERR_PAUSE
パルス（MPG-314専用） エラー発生時のタスク制御

■書　式

ERR_PAUSE n 

  n:軸指定予約定数  複数監視する場合は和をと る

■解　説

エラーフラグを検出する と タスクはポーズします。 ただし メ インタスク (タスク 0)はエラーメ ッセージを出

力して停止します。 エラーのク リ アは HOUT X_A;CLR_ERR &h400 とい う よ うに行います。

ERR_PAUSE X_A|Z_A /*XとZのエラーを監視する。

FAST
制御文 SWAP機能停止

■書　式

FAST

■解　説

ADVFSCではマルチタク スの効率的な実行の為にIF文の条件外動作にSWAPを組み込んであり ます。これは、

変数のチェッ ク と IF文の組み合わせではタスクのスイッ チの頻度が低下し、全体と しての実行効率が下がる

為です。FASTコマンド はこの機能を停止するものです。シングルタスクのみの応用の時に使用して下さい。

FCLK
システム クロックスピード変更

■書　式

FCLK n

   n:変数、 定数

     nの値 ｸﾛｯｸ(MHz)

     -5 11.53  MPC-68K相当

     -4 12.58

     -3 13.63

     -2 14.64

     -1 15.73

      0 16.78

      1 17.82

      2 18.87

      3 19.92

      4 20.97

      5 22.02  デフォルト

      6 23.06  要冷却

      7 24.12  要冷却ﾞ

■解　説

MPC-684のCPUク ロ ッ クスピードを変更します。MPC-68Kからのプログラム移植で、実行速度に依存するタ

イ ミ ングがある、 などの場合はクロ ッ クスピードを調整して ください。 電源OFFで初期状態に戻り ます。

FEDD
パルス（MPG-68K互換） スピード設定

■書　式

FEDD n

  n:スピード設定/変数 ・定数

     0(最高速)≦n≦64

＊パラ メーターを省略する と現在の設定値が表示されます。

■解　説

JUMPコマンドのゲートモーシ ョ ンの下降時のスピード設定です。

引き数を省略する と現在の設定値が表示されます。 MPG-405はU軸

に有効です。

FEDH
パルス（MPG-68K互換） スピード設定

■書　式

FEDH n

  n:スピード設定/変数 ・定数

     0(最高速)≦n≦64

＊パラ メーターを省略する と現在の設定値が表示されます。

■解　説

HOME,HOMZ コマンドでの原点復帰に先立つ原点退避移動の

スピード設定です。引き数を省略する と現在の設定値が表示さ

れます。

FEDT
パルス（MPG-68K互換） スピード設定

■書　式

FEDT n

  n:スピード設定/変数 ・定数

     0(最高速)≦n≦64

＊パラ メーターを省略する と現在の設定値が表示されます。

■解　説

ティ ーチモード でのインチング移動のスピード 設定です。引き数を省略すると 現在の設定値が表示されます。

FEDZ
パルス（MPG-68K互換） スピード設定

■書　式

FEDZ n

  n:スピード設定/変数 ・定数

     0(最高速)≦n≦64

＊パラ メーターを省略する と現在の設定値が表示されます。

上昇

FEDZ

下降

FEDD

FEED

JUMP P(2)

P(1) P(2)

SHOM,SHMZで設定　FEDHで設定

HOME動作

原点 始点 退避点

■解　説

JUMPでの下降移動を除く全てのZ軸に対するスピード設定です。 引き数を省略する と現在の設定値が表示

されます。 MPG-405はU軸に有効です。

FEED
パルス（MPG-314専用） スピード設定

■書　式

FEED n m

  n:軸指定予約定数 X_A～Z_A も し くは論理和

  m:速度指定 0～255

■解　説

ACCELで設定された最高スピードに対して256段階で速度を指定します。パルス発生中でも変更可能です。nを

省略すると MPG-68KのFEED互換と なり XYUに対する速度設定と なり ます。ACCEL実行で初期値(0)になり ます。

FEED Z_A 128       /*Z軸を半分のスピードにします
FEED Z_A 0 X_A 32  /*連続設定もできます
FEED Z_A|X_A 0     /*論理和もできます(Z,X軸 FEED=0)

注意:S字を指定している と行き先の決まっているパルス発生では増速が無効になり ます。

FEED
パルス（MPG-314専用） スピード設定(微細設定)

■書　式

FEED n VOID¦m

  n:軸指定予約定数 X_A～Z_A

  m:速度指定 1(最低)～8000(最大)

■解　説

速度指定に予約定数VOIDをorすると 設定範囲は1～8000と なり 、MCX314の速度レジスタに直接書き込みます。

ACCELの最高速度が8000以下の場合は指定値がそのままppsにな り ます。 ACCELの指定値が8000を超える

場合は最高速に対して8000等分されたppsにな り ます。

ACCEL 50000 /* 最高速8000超
FEED X_A VOID|4000 /*この場合25Kpps (50K/8K*4K)

FEED
パルス（MPG-68K互換） スピード設定

■書　式

FEED n

  n:スピード設定/変数 ・定数

     0(最高速)≦n≦64

＊パラ メーターを省略する と現在の設定値が表示されます。

■解　説

MOVE,RMOV,JUMP,GO,RMの移動に対するスピード設定です。 スピード設定はACCELコマンドによって作

成された加減速テーブルに対するスピードテーブルを使用します。 スピードテーブルとはACCELで指定さ

れたスピード範囲について64分割し、それぞれに対応する加速距離を定義したものです。例えばACCEL　 50

00 1000とすれば、FEED 0では、1000の加速距離値が入っています。 これはFEED 0が最大スピードを意味し

ており、先のACCELによればこれは1000パルスで最高のスピードに達する と しているからです。逆にFEED

64 ではその値は 1 で、 加速テーブルを登らないという こ とにな り ます。 この為、 移動距離が少ない時には

FEEDをいく ら設定しても目的のスピードに達しないこ とがあ り ます。 この例でもRMOV 1000 0  0と しても

最大スピードに達する前に減速に入ってしま うため期待通りの速度が得られないわけです。

FIND
デバッグ 文字列検索

■書　式

FIND ["string"]

  string:検索する文字列

■解　説

プログラムのラベル重複、変数名などを検索します。引数無しでラベル重複、""で文字列を与える とそれが

含まれる行を表示します。変数名は大小文字を区別します。複数行発見した場合、16行で表示を停止、CRキー

で継続になり ます。

10      'FIND TEST
20        CONST PORT0 24
30      *LABEL
40        FOR count=0 TO 23
50          ON count
60          TIME 100
70          OFF count
80        NEXT count
90      *LABEL
#FIND ←ラベル重複を調べます
*LABEL[90][30]
#FIND "PORT0" ←PORT0という文字を含む行を探します
20        CONST PORT0 24 ←1行発見
#FIND "Port0" ←大小文字は区別されます
# ←みつかりませんでした
#FIND "count" ←countという変数を探します
40        FOR count=0 TO 23
50        ON count
70        OFF count
80        NEXT count
#FIND "FOR" ←コマンドの場合、探せないこともあります
40        FOR count=0 TO 23 ←FORは見つかりましたが、
#FIND "TO" ←TOは探せません。
#

FIX
デバッグ フラッシュ ROMへ書き込み

■書　式

FIX

■解　説

フラ ッシュ ROM （FROM） へプログラムを書込みます。 RUNは書込み後プログラムを実行しますが、 FIXは

実行しません。 ダイレク ト コマンドのみの使用です。

#FIX
*-----+++++Ok
#

RUN参照
6-41



FOR..NEXT
制御文 繰り返し

■書　式

FOR～TO～[STEP]

NEXT

■解　説

数指定繰り返し制御文。加算量はSTEPによって与えるこ とができます。STEP以下が省略されれば、1ずつ増

えるこ とにな り ます。 また、 FOR文のループからGOTO文で飛び出すこ とは許されています。

FOR j=0 TO 80 STEP 20
   FOR k=0 TO 100
      PRINT i j k
   NEXT k
NEXT j

FORK
タスク操作 タスク実行

■書　式

FORK n *ﾗﾍﾞﾙ

  n:タスク番号

     1≦n≦31

■解　説

他タスク実行コマンド｡マルチタス ク とはタ イムシェア リ ングによ り複数のプログラムを一定時間毎に順

次実行し、同時に複数のプログラムが実行されているかのよ うに見せかけるものです。ADVFSCにはこの機

能が組み込まれており、同時実行できるプログラムの数は32です。機械制御ではI/Oのポーリ ング・タイマー

待ちなど、それぞれのプログラムは比較的暇なこ とが多く、 このマルチタスクが有効に動作します。ですか

ら、演算のよ うにタイマーの要素が全くない。プログラムをマルチタスクで実施しても何等メ リ ッ トはあ り

ません。 マルチタスクが効率よ く動作し意味をなすのはそれぞれのタスクにTIME,WAIT,SW(n),IN(n)などの

コマンドや関数が含まれている場合です。 SW(n),IN(n)にはノ イズ除去のために5msecのタイマーが組み込ま

れているためにTIMEがあるのと同じ意味を持ちます。

     FORK 1 *intval
     FORK 2 *p_port
*p_port
     DO
        FOR i01=0 TO 3
           P_ON i01
           TIME 100
           WAIT P_SW(i01)==1
        NEXT i01
     LOOP
*intval
     FORK 6 *b_out
     TIME 4000
     GOTO *intval

■タスク32について

タスク32は リ アルタイム(5̃7m秒程度)で反応する特殊なタスクです。

タスク32は常に優先的に実行されます。 (task0->task32->task2->task32->task3--->)

タス ク32で実行できるのはSENSE_SW 文のみです。また、タスク32中のSENSE_SWの引数にSW(),IN()などタイ

マ要素を含む関数は使えません。 HSW(),HIN() ?(),@()や変数のみ使用と して ください。

タスク32はかならず以下のよ うにラベルとENDで挟んで途中はSENSE_SW文のみです。 一行で250μ秒です

から記述は4行程度にして ください。 急がない仕事にやたらに使う と、 どんどん全体が遅くな り ます。 用が

済んだらquitして ください。

FORK 32,*sense

500 *sense
510 SENSE_SW ?(-1)&?(0) (OFF_P,0)
550 SENSE_SW ?(-1)&@(0) (ON_P,0)
555 SENSE_SW ?(-1)&?(-2) (X_A;STP_I,&H410)
556 SENSE_SW ?(-1)&?(-3) (Y_A;STP_I,&H410) (OFF_P,-3) /* 二回目は動作しないよ
560 END    うにしている。

FORK 32は最初に実行して ください。する とその後FORKしたタスクがすべて5m秒のライフタイムにな り ま

す。 これによ り タスク32の反応速度を確保しています。

QUIT 32しても5m秒のライフタイム設定は解除されません。 パワーオン リセッ トでデフォル トに戻り ます。

任意にライフタイムを戻すのにLIFE_TIMEコマンドは有効です。

FREE
編集 残りメモリの表示

■書　式

FREE

■解　説

メモ リ残量表示。 NEW後500kbyteです。 ADVFSCでは1行の消費メモ リが約20byteのため約25000ステップ記

述可能です。

GET_VAL
文字列 文字列からの数値自動取り出し

■書　式

GET_VAL str arry(n)

  str:文字列

  arry(n):配列。 nは取り込んだデータの複写先の先頭。

■解　説

文字列中にある数字を数値と して自動的に配列に取り込みます。

数字列中の小数点は無視して連続した数字列と扱います。

例)文字列からの取り出し

a$="b--99999.+321-456aho-874.1234e12b"
GET_VAL a$ aho(1)
FOR i=0 TO 8
 PRINT i : aho(i)
NEXT i 
#run
0 0
1 -99999
2 321
3 -456
4 -8741234
5 12
6 0
7 0
8 0 

例)バージ ョ ンデータの1行目から数字を取り込みます。

#VER
MPC-684 ADVFSC(r)eREV-3.81x
 BASIC like + multi tasking
 Created by ACCEL Co.'~2001

10        DIM a(2)
20        GET_VAL VER$ a(0)
30        FOR i=0 TO 1
40          PRINT i a(i)
50        NEXT i
#run
0 -684    /* MPC-684
1 -381    /* REV-3.81x

GETF
コプロ演算 データ取り出し

■書　式

GETF [45] fn1,fn2,fn3,..

  fn1,fn2,fn3,..:変数

■解　説

コプロからデータを取り出します。

与えられた変数に順番にコプロの不動小数点レジスタFP0,FP1..の値を取り出してセッ ト します。

SETF 0 3 4
GETF A B C
PR A B C
  結果は0,3,4

※[45]は、 １ ０進数での四捨五入整数化の為のオプシ ョ ンです。

GO
パルス（MPG-68K互換） 4軸同時パルス発生

■書　式

GO x y u z

GO P(n)

※このコマンドはMPG-405には使えません。

■解　説

GOコマンドは4軸同時のパルス発生です。 これはXYロボッ ト を構成した場合に斜め組み込みなどに有効で

す。GOコマンドは絶対位置移動です。これに対して相対4軸移動であるRMがあ り ます。GOコマンドはMOVE

に比べてスピードが遅くなるのは、ACCELコマンドが作成する加減速テーブルは3軸同時に見合ったものと

なっている為です。

GO P(1) 点P(1)にダイレクトに移動します。
GO 50 100 200 400 絶対座標X=50,Y=100,U=200,Z=400にダイレクトに移動します。

GOSUB
制御文 サブルーチンコール

■書　式

GOSUB *ﾗﾍﾞﾙ

■解　説

サブルーチン実行 （文番号は許されていません）

  GOSUB *p_port
  END
*p_port
     DO
      FOR i01=0 TO3
          P_ON i01
         TIME 100
         WAIT P_SW(i01)<>1
         P_OFF i01
         TIME 100
         WAIT P_SW(i01)==1
      NEXT i01
     LOOP
     RETURN

GOSUB
制御文 サブルーチンコール(引数渡し)

■書　式

GOSUB *ラベル [arg1 arg2 - -]

  arg:変数、 定数

■解　説

argをサブルーチンの引数と します。 ローカル変数と組合わせる と タスク間でのサブルーチンの共有が可能

とな り ます。 受け側は _VAR var1,var2 - - です。

10  GOSUB *SUB 1 2 3 /*渡す
20  END
30  *SUB /* ラベル!はローカル変数
40  _VAR A! B! C! /*受け取る
50  PRINT A! B! C!
60  RETURN
#RUN
1 2 3

※GOSUBのネス トは各タスク約170です。

GOTO
制御文 無条件分岐

■書　式

GOTO *ﾗﾍﾞﾙ

■解　説

無条件分岐 （文番号は許されていません）

*intval
   FORK 6 *b_out
   TIME 4000
   GOTO *intval

HEX$
文字列 ヘキサ表現文字列変換

■書　式

HEX$(n)

■解　説

数値のヘキサ表現文字列への変換

PRINT HEX$(100) 64となります。

a$="aho"
b$=HEX$(ABC(a$))
PRINT b$
61 ”a”のアスキーコードが文字列としてb$に入りました。

HIN
I/O パラレル入力

■書　式

HIN(n)

■解　説

I/Oのパラレル入力 （フ ィルタなし）。 IN(n)と同様ですが、 IN(n)のよ う なノ イズフ ィルタ機能はあ り ません。

nはポートの指定、 8bit単位で区切られてます。 例えばポートの0～7まではバンク0とな り ます。

PRINT HIN(0)&&H0F 0～3ビットを検査します。

HOME
パルス（MPG-314専用） 定則低速原点復帰

■書　式

HOME n rate cond [n1 rate1 cond]

  n:軸指定予約定数 X_A～Z_A

  rate:原点復帰速度 pps

  cond:入力指定 INx_ON～INx_OFFの論理和

■解　説

最初の引数に軸指定定数を設定する と 1軸ごとの原点復帰コマンドにな り ます。 引数にパルスレート と IN0

～IN3のどの入力の信号を使用するかを決定する値を設定できます。 条件はIN0_ON～IN3_OFFが用意されて

おり、 和をと る こ とによ り複数の信号によ り停止させるこ とができます。 HOMEはACCELで設定したパラ

メータを破壊するため、 PRSET_ACCELで復旧して下さい。 また、 座標値のク リ ア、 プ リセッ トは行わない

のでSTPSで別途実施します。

HOME X_A 10 IN0_ON Y_A -10 IN1_ON /* X軸CW10ppsIN0がオンになるまで
/* Y軸CCW10ppsIN1がオンになるまで

HOME
パルス（MPG-314専用） 高速加減速原点復帰

■書　式

HOME n cond ±F [n1 cond1 ±F1]

  n:軸指定予約定数 X_A～Z_A

  cond:入力指定 INx_ON～INx_OFFの論理和

  F:feed 1～255  符号は回転方向 +:CW,-:CCW

■解　説

第1引数に軸、 第2引数に停止条件を指定する と 1軸毎の高速原点復帰コマンドにな り ます。 引数にパルス

レート と IN0-IN3のどの入力の信号を使用するかを決定する値を設定できます。 条件はIN0_ON～N3_OFFが

用意されており論理和をと る こ とによ り複数の信号によ り停止させるこ とができます。

HOME X_A IN0_ON -125 Z_A IN0_ON ?200 /* X軸 IN0使用 FEED125 CCW
/* Z軸 IN0使用 FEED200 CCW

※注意

原点復帰時に指定した停止条件は原点復帰後も有効となるためSTOP n VOIDで条件をク リ アして下さい。

HOME
パルス（MPG-68K互換） 原点復帰

■書　式

HOME patn [x y u]

  patn:原点センサーパターン

  x y u:退避移動量 （パルス）

＊SHOM<ENTER>またはSHMZ<ENTER>で現在の設定状況が表示されます。

■解　説

XYU同時3軸で原点復帰します。 出力するパルスパタンとパルスレートはSHOMで定めた通りです。 そして

pathで指定された入力条件となるまでパルス発生し続けます。

例えば、 HOME &H5とすればXS1が1、

XS2が0、YS1が1、YS2が0となるまで原

点復帰し続けます（1=ON、0=OFF）。も

し こ こでUS1、 US12のいずれかが1に

なっている と条件と しては US1， US2

がと もに0ですから U軸の原点復帰も

加わり ます。 2番目以降の引き数は退

避移動のパルス量です。初期状態では

100パルスずつセッ ト されているので

必要に応じて設定します。 1度設定さ

れる と 2度目からは引き数がなくても

設定され値だけ退避します。又、 この

退避移動のスピードはFEDHで指定し

ます。HOMEコマンド終了時のポイン

トはX=0， Y=0， U=0にな り ます。

例1）

SHOM &H5 5000 X軸CW、Y軸CW方向、5000pps
HOME &H5 200 200 0 退避量200パルス、XS1、YS1がONになるまで原点復帰動作

例2）

SHOM &B1010 500 <--Y軸:CCW,X軸CCW方向に原点復帰、ｽﾋﾟｰﾄﾞ:500pps
HOME &B0101 <--XS1とYS1がONになるまで原点復帰

【Q】 原点復帰の前に原点センサーの入力を確認して、 も し原点センサーがON状態ならOFFになるまで退避

移動をしてから、 再び原点復帰動作を行いたい。

【A】 次のプログラムは原点復帰の前に HPT() 関数で原点センサーの状態を確認しています。 も し原点セン

サーがすべてOFFでなければCCW方向に原点センサーがOFFするまで移動します。

10       IF HPT(0)<>0 THEN : GOSUB *TAIHI : END_IF
20       SHOM &B101 500 <--Y:CW,X:CW
30       HOME &B1010 100 100 0 <--YS2,XS2がONまでﾊﾟﾙｽ発生
40       END
50     *TAIHI
60       SHOM &B1010 500 <--Y:CCW,X:CCW
70       HOME 0 0 0 0 <--ｾﾝｻｰがALL OFFまでﾊﾟﾙｽ発生
80       TIME 500
90       RETURN

J4ｺﾈｸﾀｰﾋﾟﾝ番 8 7 6 5 4 3 2 1

MSB 7 6 5 4 3 2 1 0 LSB

信号名 - - US2 US1 YS2 YS1 XS2 XS1

XS1がON,XS2がOFFまでCW

原点 始点

始点

退避移動

　  X軸CCW

200ﾊﾟﾙｽCCW

YS1がON,

YS2がOFFまでCW

　            Y軸CCW

200ﾊﾟﾙｽCCW     退避移動

HOMZ
パルス（MPG-68K互換） 原点復帰

■書　式

HOMZ patn [z]

  z:退避移動 （パルス）

＊SHOM<ENTER>またはSHMZ<ENTER>で現在の設定状況が表示されます。

■解　説

Z軸の原点復帰をします。出力するパルスパタンとパルスレートはSHMZで定めた通りです。そしてpatnで指

定された入力条件となるまでパルス発生し続けます。

例えば、HOMZ &H1とすればZS1が1,2が0になるまで原点復帰し続けます。2番目以降の引き数は退避移動の

パルス量です。初期状態では100パルスにセッ ト されているので必要に応じて設定します。1度設定される と

2度目からは引き数がなくても設定された値だけ退避します。又、この退避移動のスピードはFEDHで指定し

ます。 HOMZコマンド終了時のポイン トはZ=0にな り ます。

SHOM &H1 5000 Z軸CW方向、5000pps
HOME &H1 200 退避量200パルス、ZS1がONになるまで原点復帰動作

HOUT
パルス（MPG-314専用） ポート出力

■書　式

HOUT p

  p:出力データ1～255

■解　説

HOUTを1～255の範囲の引数で使用する とOP1～4,X-SON～Z-SONのパラレル設定とな り ます。

下位がOP1～4の設定で従来のコマンド と互換性があ り ます。

HOUT
パルス（MPG-314専用）  レジスタ制御

■書　式

HOUT n;cmnd

  n:軸選択予約定数 X_A～Z_A

  cmnd:コマンド予約定数

■解　説

MCX314のレジスタWR0を制御します。

HOUT X_A;STP_D /* X軸に対して減速停止命令を実行します(STOP X_A STP_Dと同じ)
HOUT Y_A;CLR_ERR /* Y軸のエラービットをクリアします

J4ｺﾈｸﾀｰﾋﾟﾝ番 8 7

MSB 7 6 5 4 3 2 1 0 LSB

信号名 - - - - - - ZS2 ZS1

HOUT
パルス（MPG-68K互換） パルスボード汎用出力

■書　式

HOUT patn

■解　説

MPC-684、 MPG-68KのJ4コネク タには原点入力とあわせて4点の出力ポートがあ り ます。 HOUTはこの出力

に対してパラレル出力します。 尚このコマンドはMOVEやACCELと同じ くPGコマンドひきあてられたタス

ク、 も し くはPGSELによって選択されたMPGに対して有効です。 MPG動作中には実行できません。

PG &HE0 1 <--アドレス&HE0のMPGをタスク1に引き当て
FORK 1 *task1 <--タスク1の起動
（略）
*TASK1
     HOUT &H3 <--アドレス&HE0のMPGのJ4コネクタOP 1、OP 2をON

HPT
パルス（MPG-314専用） 入力ポート読み込み

■書　式

HPT(0)

■解　説

MPG-314の入力を読み込みます。

#PRX HPT(0)
  00000000 ←表示データ
  H7 ← H0 ←桁

返す値を8桁のヘキサで表現する と、 次のよ うに入力ポートの値がそれぞれの桁に反映されます。

H0: XS1,XS2(X-IN0,X-IN1),YS1,YS2(Y-IN0,Y-IN1) (コネク タJ4)

H1: US1,US2(U-IN0,U-IN1),ZS1,ZS2(Z-IN0,Z-IN1) (コネク タJ4)

H2: X-IN2～Z-IN2(差動入力) (コネク タJ1)

H3: X-IN3～Z-IN3 (コネク タJ2)

H4: X-INPOS～Z-INPOS (コネク タJ4)

H5: X-ALM～Z-ALM (コネク タJ2)

例)X-INP(位置決め完了)待ち

WAIT HPT(0)&&H10000<>0

例)X-IN3オン待ち

WAIT HPT(0)&&H1000<>0

HPT
パルス（MPG-68K互換） 原点センサポート入力

■書　式

HPT(n)

  n:入力ビッ ト番号

     n=0:パラレル入力

     n=1～8:ビ ッ ト入力

■解　説

HPT()はMPG-68Kの原点入力を読む関数で、 原点センサーの

I/Oチェッ クに利用できます。 入力状態はTEACHモードのH

操作でも表示されます。 HOUT,HPT()はTEACHかPGSELで選

択されたMPGに対して有効です。

PGSELでMPGを選んで

PRX HPT(0)
PRINT HPT(1)
HOUT &H3

＊MPC-684のJ4コネクター入出力はP_SW参照 

HSW
I/O ビットの読み込み

■書　式

HSW(n)

  n:ポー ト番号

■解　説

I/Oのビッ ト入力 （フ ィルタなし）。 nはポート番号の指定です。 SW(n)と同様ですが、 SW(n)のよ う なノ イズ

フ ィルタ機能はあ り ません。 返す値は1がON状態、 OFFで0とな り ます。 ポート番号に出力ポート を指定す

る と出力ポートの状態を得るこ と もできます。

ON 0 0ﾎﾟｰﾄ ON
PRINT HSW(0) 0ﾎﾟｰﾄ の状態表示
 1

ON -1 ﾒﾓﾘｰI/O ON
PRINT HSW(-1) ﾒﾓﾘｰI/Oの状態表示
 1

J4ピン番号 コマンド ・ 関数

 1

 2

 3

 4

 5

 6

 7

 8

HPT(1)

HPT(2)

HPT(3)

HPT(4)       HPT(0)   入力

HPT(5)

HPT(6)

HPT(7)

HPT(8)

 9

10

11

12

   

     HOUT patn        出力

IF
制御文 条件分岐

■書　式

IF　条件式　THEN

ELSE

END_IF<EIF>

■解　説

条件分岐文｡条件式で、指定された式が成立すればIF～ELSE間が実行され非成立であれば、ELSE～END_IF間

のみが実行されます。 次の例ではiが10以下とそ うでない場合に分類し10以下であればPRINT "i<=10" iを実

行します。 10を除く10以上であれば、 ELSEよ り下を実行するわけですがこの中にもIF文があ り iが13の時に

限りPRINT "i=13" iを実行します。

FOR i=0 TO 20
  IF i<=10 THEN
    PRINT "i<=10" i
    ELSE
    IF i==13 THEN
      PRINT "i=13" i
    END_IF
    PRINT "i>10" i
  END_IF
NEXT i

IN
I/O パラレル入力

■書　式

IN(n)

  n:バンクナンバー

■解　説

I/Oのパラレル入力。HIN(n)と同様ですが、 2度読みのノ イズフ ィルタ機能付きです。 これは、チャタやノ イ

ズによる誤読みと り を防ぐためのものです。nはポートのバンク指定で、8bit単位で区切られています。例え

ばポートの0～7まではバンク0とな り ます。I/Oサイズ指定予約定数を使って入力バイ ト数を指定できます。

IN(n~Lng) /* 4バイト入力
IN(n~Wrd) /* 2バイト入力 
IN(n~Int) /* 2バイト符号付入力

下記はメモ リ I/Oのパラレル入出力の例です。

#OUT &H12 -1 /* メモリI/Oのバンク1へ1バイト出力
#OUT &H34 -2 /*         〃         2へ1バイト出力
#OUT &H56 -3 /*         〃         3へ1バイト出力
#OUT &H78 -4 /*         〃         4へ1バイト出力
#PRX IN(-1) /* メモリI/Oのバンク1から1バイト入力
 0012
#PRX IN(-1~Lng) /* メモリI/Oのバンク1～4入力
 78563412
#PRX IN(-1~Wrd) /* メモリI/Oのバンク1～2入力
 3412

#OUT -1 -1~Int /* メモリI/Oのバンク1～2へ出力
#PR IN(-1~Wrd) /* メモリI/Oバンク1～2入力
65535

#PR IN(-1~Int) /* メモリI/Oバンク1～2符号付入力
-1

INCHK_314
パルス（MPG-314専用） 入力モニタ

■書　式

INCHK_314

■解　説

INCHK_314はMPG-314の入力ポート をモニタするコマンドです。標準I/Oに対するIOコマンド と同等の機能

を持ちます。

INP$#0
RS-232 n文字読み込み

INPUT$参照

INP$#2
RS-232 n文字読み込み

INPUT$参照

INPBLK#
RS-232 バイナリ固定フォーマット入力

■書　式

INPBLK#n v1 v2 v3

  n:ポー ト番号0,2

  v1,v2,v3:変数

■解　説

データのバイナリ レシーブです。input#は文字列解析・計算が入りデータの取得に時間がかかり ますがinpblk

#はバイナリ固定フォーマッ トでデータ受け取るこ とによ り高率アップします。

送信フォーマッ ト

[STX B0 B1 B2 B3 B4 B5 B6 SUM]

   STX = &H2

   B0～B6 = ﾊﾞｲﾅﾘﾃﾞｰﾀ

   SUM = ﾁｪｯｸｻﾑ (0-((B0+B1+B2+B3+B4+B5+B6) and &HFF))

受信したデータは変数v1～v3に数値と して入力されます。 チェ ッ クサムでエラーになる とv3の値が255にな

り ます。

         (上位)  (下位)

v1  ←   B0,B1,B2

v2  ←   B3,B4,B5

v3  ←   B6

パソ コンから数値12345を送信する。 10進12345は&H3039です。 送り側データは次の通りです。

&H02 &H00 &H30 &H39 &H00 &H00 &H00 &H00 &H97

MPCの変数には次の様に入力されます。

prx v1
 3039 ←10進12345
prx v2
 0000
prx v3
 0000 ←通常は送信ﾃﾞｰﾀと一致。ﾁｪｯｸｻﾑｴﾗｰで&HFF(255)

つま り、 3byte （±8388607） データを同時に2つ、 チェッ クサム付きで受信可能です。

INPUT
RS-232 データ入力

■書　式

INPUT [s$,v]

INPUT#0 [s$,v]

INPUT#2 [s$,v]

■解　説

CH1数値/文字列の入力コマンド

CH0数値/文字列の入力コマンド

CH2数値/文字列の入力コマンド

INPUT 文は無手順系の通信コマンドでターミネータ と してスペース、 タブ、 CR を使用するコマンドです。

CH1のINPUTコマンドはエコーバッ クがあ り ますがCH0,CH2についてはエコーがあ り ません。 また、INPUT

文も文字列を表示するこ とができます。文字列定数・定数に対してはPRINTコマンド と同様出力と して動作

します。

INPUT#2 "Please Set Any string ? " a$
PRINT#2 "echo--" a$+CHR$(13)+CHR$(10)

INPUT "Please Set Any string ? " a$
PRINT "echo--" a$

■INPUTのターミネーター
MPC-684のRS-232のINPUT コマンドはスペース、 タブ、 キャ

リ ッジ リ ターンをターミネーターと して扱います。次のサンプ

ルプログラムはMPC-684を2台を接続して作り ました。

（1）　 スペース/タブコードの入ったデータ
（1-1) 文字列の場合その1　 スペースの入っている文字列

送信側

 LIST 0
5        CNFG#2 "9600b8pns1NONE"        CNFGコマンドでRS-232初期化
7        TIME 100
10       PRINT#2 "AB CDE¥r"

受信側

 LIST 0
10       CNFG#2 "9600b8pns1NONE"
20         A$="" :   B$=""
25       TIME 50
30       INPUT#2 A$ B$

結果 スペースで区切られ2つの文字列になる

PRINT A$
AB

MPC-684

CH2

MPC-684

CH2

#PRINT B$
CDE

 (1-2) 文字列の場合その2　 スペースが入らない文字列

送信側

 LIST 0
5        CNFG#2 "9600b8pns1NONE"
7        TIME 100
10       PRINT#2 "ABCDE¥r"

受信側

 LIST 0
10       CNFG#2 "9600b8pns1NONE"
20         A$="" :   B$=""
25       TIME 50
30       INPUT#2 A$
結果 とーぜん1つの文字列として扱われる
PRINT A$
ABCDE
#

(1-3) 定数 ・ 変数の場合その1　1つの定数

送信側

 LIST 0
5        CNFG#2 "9600b8pns1NONE"
7        TIME 100
10       PRINT#2 12345 "¥r"

受信側

 LIST 0
10       CNFG#2 "9600b8pns1NONE"
20         A=0
25       TIME 50
30       INPUT#2 A

結果

PRINT A
12345
#

(1-4) 定数 ・ 変数の場合その2　データ間に半角スペースがある

送信側

 LIST 0
5        CNFG#2 "9600b8pns1NONE"
7        TIME 100
10       PRINT#2 123 45 "¥r"

受信側

 LIST 0
10       CNFG#2 "9600b8pns1NONE"
20         A=0
25       TIME 50
30       INPUT#2 A

結果 1つの数値と して入力 （これは出力側の都合）

PRINT A
12345
#

(1-5) 定数 ・ 変数の場合その3　 スペースコードが入っている

送信側

 LIST 0
5        CNFG#2 "9600b8pns1NONE"
7        TIME 100
10       PRINT#2 123 " " 45 "¥r"

受信側

 LIST 0
10       CNFG#2 "9600b8pns1NONE"

20         A=0:B=0
25       TIME 50
30       INPUT#2 A B

結果 スペースで区切り2つの数値を入力

PRINT A
123
#PRINT B
45
#

RS 2 RS-232バッファーを表示

RS受信

       00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00
       00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00
       00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00
       00 00 00 00 00 00 00 00 00 31 32 33 20 34 35 0D

RS送信

       00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00
       00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00
       00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00
       00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00
#

INPUTは文字列でも数値でもスペース(&H20)、TAB(&H09)、CR(&H0D)をターミ ネーターと して同じに扱います。

スペースの入った文字列を1つの文字列変数に格納したい場合は次の様にINP$で受信するのが能率的です。

送信側

 LIST 0
5        CNFG#2 "9600b8pns1NONE"
7        TIME 100
10       PRINT#2 "ab cde¥r"

受信側

 LIST 0
10       CNFG#2 "9600b8pns1NONE"
20         A$="" :   B$=""
25       TIME 50
30       DO UNTIL A$==CHR$(&HD) CRまで1文字づつ入力します。
35           A$=INP$#2(1)
40           B$=B$+A$ 文字列の結合
50       LOOP

結果

PRINT B$ 1つの文字列変数に格納されています
ab cde

（2）　 コン ト ロールコードの場合は
例えばバッ クスペースコードが含まれているデータをINPUTで入力する と き数値と して入力するか、文字列

と して入力するかで扱いが違います。

(2-1) バッ クスペースの入ったデータを数値と して入力する と．．．

送信側

[1][ﾊﾞｯｸｽﾍﾟｰｽ][2][CR]　と送信
#PRINT#2 1 CHR$(&H8) 2 "¥r"

受信側バッファの内容 （ﾊﾞｯﾌｧ先頭は"1"(&H31)になっている）

       00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00
       00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00
       00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00
       00 00 00 00 00 00 00 00 00 00 00 00 31<08 32 0D

これを数値と して読み込むと

#input#2 a
#print a ...ﾊﾞｯｸｽﾍﾟｰｽをﾀｰﾐﾈｰﾀｰとして1だけをaに入力
1

受信側バッファの内容 （ﾊﾞｯﾌｧ先頭は"2"(&H32)になった）

       00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00
       00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00
       00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00
       00 00 00 00 00 00 00 00 00 00 00 00 31 08 32<0D
#input#2 b
#print b ...CRをﾀｰﾐﾈｰﾀｰとして2をbに入力
2

つま り、 数値と して入力する と数字以外のコードはターミネーターと して扱われます。

(2-2) 文字列と して入力する と ・ ・ ・ ・

送信側

#print#2 1 chr$(&h8) 2 "¥r"

受信側バッファ内容

       00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00
       00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00
       00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00
       00 00 00 00 00 00 00 00 00 00 00 00 31<08 32 0D
#input#2 a$ ...文字列変数a$に入力
#print a$　　　　　
2 ...2しか入らない

受信側バッファ内容

       00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00
       00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00
       00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00
       00 00 00 00 00 00 00 00 00 00 00 00 31 08 32 0D

つま り、 文字列と して入力する とﾊﾞｯｸｽﾍﾟｰｽと して解釈される。

送信側

#PRINT#2 "ABC" CHR$(&H8) "DEFG" "¥r"
受信側
#INPUT#2 A$
#PRINT A$
ABDEFG <--画面上"C"の上に"D"が重ね書きされた。

■INPUTのLFコード （&HA） の扱いについて
INPUTはLFを普通の文字列と同様に扱い、 RS-232のバッファから削除したり しません。

CR ・ LF(¥n)でターミネート されてる文字列を入力する とCRまで読み込みLFはバッファに残されます。 そし

て次に読み込む文字列の先頭はLFにな り ます。 先頭にLFがある文字列を文字列変数にINPUTで読み込むと

その文字列変数の先頭にもLFが入りいますが、 PRINTコマンドでその文字列を画面表示をする とLFは無視

されるのでLFが無いよ うに見えます （例1）。 LFが先頭にある文字列を数値変数と して読み込むとLFでキャ

ンセルされて変数は常に0になってしまいます （例1）。 （例3） では文字列変数に読み込みLFを削除して数値

に変換しています。接続されている機器と送受信タイ ミ ングのインターロ ッ ク されている場合はINPUTで入

力した後、 CNFGコマンドでバッファをク リ アーしてしま うのが一番簡単です。

（例1） 文字列変数に読み込む場合

CNFG#2 "9600b8pns1NONE" <--CNFGで通信条件設定とﾊﾞｯﾌｧｸﾘｱｰ
#print#2 "123¥n" <--1回目送信
#rs 2

RS受信

        00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 
        00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 
        00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 
        00 00 00 00 00 00 00 00 00 00 00 31<32 33 0D 0A 
6-57



RS送信

        00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 
        00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 
        00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 
        00 00 00 00 00 00 00 00 00 00 00 31 32 33 0D 0A 
#input#2 str$ <--CRまで読み込む、LFはバッファに残す
#pr str$
123 <--1回目は変数の内容、文字数OK、
#pr len(str$)
3
#print#2 "456¥n" <--2回目送信
#rs 2

RS受信

        00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 
        00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 
        00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 
        00 00 00 00 00 00 31 32 33 0D 0A<34 35 36 0D 0A 

RS送信

        00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 
        00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 
        00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 
        00 00 00 00 00 00 31 32 33 0D 0A 34 35 36 0D 0A 
#input#2 str$ <--前回のLFから読み込む
#pr str$
456 <--文字列として読み込み表示するとまとも？
#pr len(str$)
4 <--文字数はLFを含む数

（例2） 数値変数に読み込む場合

CNFG#2 "9600b8pns1NONE"
#print#2 "123¥n" <--1回目送信
#rs 2

RS受信

        00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 
        00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 
        00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 
        00 00 00 00 00 00 00 00 00 00 00 31<32 33 0D 0A 

RS送信

        00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 
        00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 
        00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 
        00 00 00 00 00 00 00 00 00 00 00 31 32 33 0D 0A 
#input#2 n <--CRまで読み込む、LFはバッファに残す
#pr n
123 <--1回目はOK
#print#2 "456¥n" <--2回目送信
#rs 2

RS受信

        00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 
        00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 
        00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 
        00 00 00 00 00 00 31 32 33 0D 0A<34 35 36 0D 0A 

RS送信

        00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 
        00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 
        00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 
        00 00 00 00 00 00 31 32 33 0D 0A 34 35 36 0D 0A 
#input#2 n <--LFから読み込む
#pr n
0 <--0???

（例3） 文字列変数に読み込みLFを削除して、 数値に変換

10        CNFG#2 "9600b8pns1NONE"
20        FOR i=1 TO 3

30          PUT#2 &HA &HA
40          PRINT#2 "123¥n"
50          TIME 20
60          INPUT#2 st$
70          GOSUB *lf_remove
80          PRINT st$ " " LEN(st$) VAL(s$)
90        NEXT i
100       END 
110     *lf_remove
120         s$=""
130       FOR j=0 TO LEN(st$)-1
140         STRCPY st$ tmp$ j 1
150         IF tmp$<>CHR$(&HA) THEN :   s$=s$+tmp$ : END_IF 
160       NEXT j
170       RETURN 
#
RUN
123 5 123
123 6 123
123 6 123

前記実験はMPC-684のコネク ター J1の8,9番ピンをシ ョート して行いました。

INPUT$
RS-232 n文字読み取り

■書　式

INP$#0(n)

INP$#2(n)

INPUT$(n)

■解　説

INP$#0:RS-232 CH0よ り n文字取り出す

INP$#2:RS-232 CH2よ り n文字取り出す

INPUT$:RS-232 CH1よ り n文字取り出す

INP$#0()， INP$#()， INPUT$()は各ポート よ り指定された文字だけ取り出す関数です。通常は次のよ うに文字

列変数へ複写して使用します。も し、バッファに指定された文字列が受け取られていなければ数が満たされ

るまでポーリ ングします。

A$=INP$#0(10) CH0より10文字取り出してA$へコピー
A$=INP$#2(10) CH2より10文字取り出してA$へコピー
A$=INPUT$(10) ターミナルソフトからの入力です。

INPUT#0
RS-232 データ入力

INPUT参照

INPUT#2
RS-232 データ入力

INPUT参照

INSET_314
パルス（MPG-314専用） 入力ポート機能設定

■書　式

INSET_314 n cond

  n:軸指定予約定数

  cond:入力指定

■解　説

入力ポートの機能を設定します。 最後に実行したINSET_314が有効にな り ます。

パラ メータで与えられたもの以外の設定はリセッ ト されます。

INSET_314 X_A ALM_ON¦INP_ON 
X軸に対して入力設定します。アラームを有効にしてON状態をアラームと します。また、インポジシ ョ ンを

有効にしてONでインポズ成立と します。

INSET_314 ALL_A ALM_ON¦INP_OFF
すべての軸に対して入力設定します。 アラームはONで有効、 INPOSはOFFで有効と します。

IO
I/O I/Oモニタ

■書　式

IO [n]

  n:ポー ト番号

■解　説

I/O状態ビッ トマップ表現。I/Oの状態を画面に表示します。引き数で指定された値のポート番号のバンクか

ら 64点表示し、 0.5秒毎リ フレッシュされます。 任意のキィーを押すと停止します。 実際のI/Oチェッ クは

FTMWのIOC機能の方が便利です。

IOR
I/O バス読み込み

■書　式

IOR(adrs)

  adrsよ り読みだし （バイ ト読みだし）

     0≦adrs≦&HFFFF

IOW n adrs

  adrsにnを書き込み （バイ ト書き込み）

     0≦adrs≦&HFFFF

30

■解　説

MPC-684はPC-98のI/Oバスに準拠したバス構造となっています。この為、PC-98シ リーズに用意されている

I/Oボードをこのコマンドで直接読み書きする事が出来ます。

IOW tsk03 &H1D0
IOW NOT(tsk03) &H1D1
TIME 1
IF IOR(&H1D1)<>tsk03 OR IOR(&H1D1)<>NOT(tsk03)&&HFF THEN
      GOTO *rr
END_IF

IOR(adrs)
I/Oバスアドレスadrsよ りバイ ト アクセスします。但し、MPC-684はハード的な制約から奇数アドレスに対す

るバイ ト読みだしがワード呼び出しになってしまいます。 （A0が常に0の為）

IOW n adrs
I/Oバスアドレスadrsに対してバイ ト書き込みします。 但しMPC-684はハード的な制約から奇数アドレスに

対するバイ ト書き込みがワード書き込みになってしまいます。 （A0が常に0の為）

IOW
I/O  バス書き込み

IOR参照

JMPZ
パルス（MPG-68K互換） ゲートモーション移動

■書　式

JMPZ P(n)

  n:ポイン トナンバー

     1≦n≦10000 （mモデル）

■解　説

JMPZは調整用コマンドです。 ゲートモーシ ョ ンを途中で終了してZ軸の下降移動を実施しません。 JMPZは

ファンクシ ョ ンキ ィーでサポート されている他、 ティーチモードの’ A’ によっても実施されます。

JUMP
パルス（MPG-68K互換） ゲートモーション移動

■書　式

JUMP P(n) [a]

  n:ポイン トナンバー

     1≦n≦10000 （mモデル）

■解　説

JUMPはゲートモーシ ョ ンです。 JUMPはファンクシ ョ ンキ ィーでサポー

ト されている他、 ティーチモードの"J"によっても実施されます。 又、 第

2引き数を指定する と値分Z軸のパルス発生に加算されます。 JUMP P(1) -

10は目的となるポイン ト P(1)よ り -10パルス上空で停止とな り ます。
10ﾊﾟﾙｽ

P(1)

JUMP P(1) -10

MOVE P(1)

LD_M
MBK-SH メモリ一括コピー (MPC→MBK)

■書　式

LD_M array mbktop count [opt]

  array:MPC点ﾃﾞｰﾀの先頭(XYUZ(1～13000))、 配列変数の先頭

  mbktop:MBKﾒﾓﾘ先頭ｱﾄﾞﾚｽ。 20≦mbktop

  count:ｺﾋﾟｰ数

  opt:4=ﾛﾝｸﾞﾜｰﾄﾞで1ｱﾄﾞﾚｽおき、 Lng=ﾛﾝｸﾞﾜｰﾄﾞで連続

■解　説

配列から MBKのデータエ リ アにコピーします。 パラ メータはアドレスの使用範囲を超えねよ うに設定して

下さい。

LD_M X(9020) 20 683 /* データX(9020)からDT20へ683個ワードコピー
LD_M test(10) 500 10 /* 配列変数test(10)からDT500へ10個ワードコピー
LD_M X(9020) 20 342 4 /* 点データX(9020)からDT20へ342個ロングコピー
LD_M X(9000) 500 5 Lng /* 点データX(9000)からDT500へ5個ロングコピー (連続)
LD_M test(0) 500 5 Lng /* 配列変数test(0)からDT500へ5個ロングコピー (連続)

LEN
文字列 文字数取得

■書　式

LEN(s$)

  s$:文字列

■解　説

文字列s$の長さを与える。

PRINT LEN("ABC")
3
S$="ABC"
PRINT LEN(S$)
3

LET
演算  式実行

■書　式

[LET] 代入式

■解　説

LETは式を実行するコマンドです。 LETは、 入力してもLIST時には表示されません。 又、 式のみを記述して

も内部で自動的に式と して判断します。

数値演算
数値演算は、=()の記号、及び+,-,*,/,%,&,¦,¥の演算子で記述する事が出来ます。()の中式は優先的に演算され*,

/,%,&,¦,¥は+,-よ り優先される演算となっています。

+ 加算

- 減算

* 乗算

/ 除算

% 余算

& 論理積 （and）

¦ 論理和 （or）

¥ 排他的論理和 （xor）

#LIST
10     dsw=(&HF0&IN(24))/16*10+&HF&IN(24)
20   PRINT "dsw=" dsw
110    a=3000 :   b=4000 :   c=30 :   d=40
120    res=SQR(SQ(a)+SQ(b))-SQR(c*a+d*b)
130  PRINT "res=" res
#run
dsw=55
res=4500

この例は、 入力ポートに接続されたデジスイ ッチの内容を読み取る演算と一般の計算の例です。 120の行に

記述されたSQR及びSQはそれぞれ平方根と自乗 （2乗） を意味しています。 又、 定数と しての&HF0,&HFは

ヘキサ表現を意味しています。 ビッ ト表現の場合は&B□□と します。

文字列演算
ADVFSCでは文字列の演算もできます。 許される演算子は、 加算のみです。

#LIST
40    aho$="HEX->"+"&H"+HEX$(100*100+1)
50  PRINT aho$
80  INPUT ”何か入力して！”　a$
90    b$=”あなたが入力したのは－＞”
100   c$=b$+a$
110 PRINT c$
#run
HEX->&H2711
何か入力して！ 123
あなたが入力したのは－＞123

文字列演算中でも 40の行にある通り文字列を得る関数の中身である限り数値演算も可能です。 HEX$()は与

えられた数値のヘキサ表現の文字列を与えます。

LIFE_TIME
タスク操作 タスク寿命を設定

■書　式

LIFE_TIME taskn m 

  taskn:ﾀｽｸ番号

  m:寿命値  1～20。 2m秒単位

■解　説

MPC-684のタスクのライフタイムは約22m秒となっています。この時間を変更するコマンド と して、LIFE_TI

MEがあ り ます。FORK文のあとで実行します。次の例ではタスク1が2m秒でタイムアウ ト し、タスク0に実行

権を引き渡します。

5 FORK 1 *aho
6 LIFE_TIME 1 1
10 DO
20 LOOP
100 *aho
110 DO
120 LOOP 

※参考    ラ ウンド ロビンのベースタイム  22msec      ディ スパッチオーバーヘッ ド    0.2msec

LIMZ
パルス（MPG-68K互換）  ゲートモーション規制

■書　式

LIMZ a

＊パラ メーターを省略する と現在の設定値が表示されます。

■解　説

LIMZはJUMP,JMPZのZ軸の上限を指定します。例えば、初期状態で

点P1=(1000,1000,100,2000)、点P2=(2000,3000,1000,3000)と し点P1の

位置から点P2の位置にJUMPも し くは、JMPZする と このコマンドの

最初の移動はMOVZ 0です。 こ こで、 LIMZ 500と指定しておく と最

初の移動がMOVZ 500とな り ます。LIMZの値を適切に設定する事に

よ り、 Z軸移動にかかる無駄な時間を減らす事が出来ます。

LIMZ 0

JUMP P(2)

P(1) P(2)

500ﾊﾟﾙｽLIMZ 500

LIST
編集 プログラム表示

■書　式

LIST [n m]

LIST [*ﾗﾍﾞﾙ m]

  n:表示開始文字番号 （省略する と前回の続き）

  m:表示行数 （初期値）

■解　説

プログラムの表示

LIST 続きを表示

LIST 文番号 文番号よ り表示

LIST *ラベル ラベルよ り表示

LIST *ラベル 4 ラベルよ り4行表示

LIST 0 0 全部表示

表示行数mは記憶されます。

LOF
RS-232 バッファの文字数

■書　式

LOF(n)

  n:RS-232ポートチャンネルナンバー

■解　説

CHnのバッファ中のキャラ ク タの数の取得

(このプログラムはCH1をループバックさせていないと動作しません)
     CNFG#0 "9600b8pns1NONE"
     FORK 1 *aho
     DO
        PRINT LOF(0) ’バッファは255byte
        TIME 500 ’255byteを越えると読み捨て
     LOOP 　
*aho
     PUT#0 &H41 ’&H41=A
     DO
        TIME 100
        PUT#0 &H30 ’&H30=0
     LOOP

LOG
デバッグ プログラムポート出力記録

■書　式

LOG [0]

■解　説

LOG<ent>とする とRS-232 CH1(プログラムポート )の出力記録を表示します。 これによ り自動実行時のラン

タイムエラーの事後確認ができます。 記録メモ リは1kバイ トの リ ングバッファです。 LOG 0<ent>とする と

バッファをク リ アします。

※注意  タスク0でのランタイムエラーのメ ッセージは記録できません。

LOOP
制御文 繰り返し

DO.LOOP参照

M_RMVS
パルス（MPG-314専用）  非対称加減速移動

■書　式

M_RMVS n pls slow_length [adj]

  n: X_A～Z_Aのいずれか一つ

  pls: 移動距離

  slow_length: 減速距離

  adj: 減速開始位置微調整

■解　説

減速度ファ ク タを与え減速度を決定して非対称加減速度と します。 内部演算を整合させるため直前に実行

されたACCELのパラ メータを使用します。 加減速の比は加速距離と反比例関係にあるこ とによ り算出しま

す。

減速度=L*加速度/減速距離。

adjは引きずりあるいは尻切れが発生した場合の微調整用です。 引きずりは加速距離と減速距離の和が移動

距離よ り大きいとかならず発生します。 また、一度M_RMVSを実行する とパルス発生モードが変わってしま

いその後のパルス発生に影響します。

このため用済みになったらM_RMVS X_A (パルス引数無し)を実行するかACCELを実行して通常モードに戻

します。

例)800パルスで加速して8kppsにな り残り 10000パルスを減速する。

ACCEL 8000 800 10
M_RMVS X_A 10800 10000
WAIT RR(X_A)==0 /*停止待ち
M_RMVS X_A /*ﾓｰﾄﾞｸﾘｱ

例)10000パルスで加速して8kppsにな り残り800パルスを減速する。

ACCEL 8000 10000 10
M_RMVS X_A 10800 800
WAIT RR(X_A)==0 /*停止待ち
ACCEL 8000 /*ﾓｰﾄﾞｸﾘｱ

800     10000

8000pps

(この例の場合は定速区間は無い)

MBK
MBK-SH データエリア読み出し

■書　式

ワード(2バイ ト )読み出し

MBK(dtadr)

  dtadr:MBKレジスタアドレス

符号付ワード(2バイ ト )読み出し

MBK(dtadr̃Int)

  dtadr: データアドレス

  Int: 予約定数

符号付ロング(4バイ ト )読み出し

MBK(dtadr̃Lng)

  dtadr: データアドレス

  Lng: 予約定数

■解　説

MBK-SHのデータエ リ アからデータを読み出します

p=MBK(0) /* DT0000(表示画面番号)読み出し
r=MBK(21~Int) /* DT0021を符号付で読み出し
r=MBK(699~Lng) /* DT0699(下位)とDT700(上位)をﾛﾝｸﾞ読み出し

MBK
MBK-SH バックアップ状態の確認

■書　式

MBK(-1)

■解　説

MBKのデータはMBK関数、S_MBKコマンド実行時にMPC-684のS-RAMにバッ クアップされ、電源投入時に復

元されます。 も し もバッ クアップデータが破損した場はMBKにデータは復元されません。

MBKデータが正常だったかど うかは、 MBK(-1)関数で知るこ とができます。

MBK(-1)が0であれば、データは正常でバッ クアップデータが復帰されています。も しデータが破損している

と、破損している最初のMBKデータの番号を返します。 また、MBKのデータはク リ ア状態になっています。

#PR MBK(-1)
0

MEM
メンテナンス メモリーテスト

■書　式

MEM

■解　説

メモ リーテス トです。RAMの動作が疑わしい時に使用します。プログラムエ リ アのRAMをプログラムを破壊

するこ とな くテス ト します。 このコマンドでプログラムが壊れた り、 エラーが出力されればメモ リ不良で

す。購入後1年以内であれば返品して下さい。MEM<ENTER>する と メモ リ テス ト スタート と表示されます。メ

モ リ テス ト終了と表示されるまで待ちます。

MON
デバッグ 停止状態確認

■書　式

MON [1]

■解　説

各タスクの状態モニタ。 通常は<CTRL>＋<A>によって停止した状態をモニターするものですがサブタスク

によって常時表示する事も可能です。 引き数を与える と0.5秒毎に実行文番号を表示します。

#LIST
10   FORK 1 *task1
20   FORK 2 *task2
30   DO
40     FOR i=0 TO 7 : ON i TIME 100 : OFF i : NEXT i
50   LOOP
60 *task1
70   DO
80     FOR j=8 TO 15 : ON j : TIME 100 : OFF j : NEXT j
90   LOOP
100 *task2
110   DO
120     FOR k=16 TO 23 : ON k : TIME 100 : OFF k : NEXT k
130   LOOP
#run
      *0      [40] <-- <CTRL>+<A>で停止
      *1      [80]       *2  [120]
#mon
      *0      [40]
      *1      [80]       *2  [120]
#

■プログラム停止時とMON実行時のタスク番号後ろの 「!」 の意味。
タイムスライス内(20mSec)にSWAPが無いタスクを表します。これが沢山あると タスク効率の低下になり ます。

メ モリ ーアクセス(メ モリ I/O,MBK入力,演算等)の連続、 WAITで条件成立している時(次記)などがあり ます。

a=0
DO
  WAIT a==0 /*SWAPしない
LOOP

MOVE
パルス（MPG-68K互換） XYU絶対座標移動

■書　式

MOVE x y u

  x,y,u:絶対座標 （パルス）

MOVE P(n)

  n:ポイン トデータ

     1≦n≦13000

■解　説

3軸同時移動のパルス発生コマンドです。MOVEは絶対座標移動です。 スピード上限及び加減速度はACCEL

コマンド既定されその範囲内でのスピード設定はFEEDコマンドで指定できます。

引き数と しては3軸分(X,Y,U)をスカラ値で与えるこ と も、 点データP(n)で指定するこ と もできます。 点デー

タP(n)を使用した場合はP(n)のX,Y,U成分だけが使用されZ成分は無視されます。

MOVL
パルス（MPG-314専用） 直線補間（絶対座標移動）

■書　式

MOVL x y u z

    x y u z:各軸座標時  パルス発生無=VOID

■解　説

直線補間パルス発生コマンドです。MOVL x y u zと しますが、現在位置との相対値を算出してパルス発生し

ますので、MOVSと同様現在位置が確定しているこ と と、実際の直線補間軸が何軸になるか注意が必要です。

例えば、 現在位置 100 200 300 400 で MOVL 200 300 400 500とする とそれぞれ100パルスずつの出力が必要

で、結果4軸直線補間とな りエラーとな り ます。MOVL 200 300 400 400ではZ軸の発生量は0とな り3軸直線補

間で移動可能です。 相対値を計算してからのパルス発生はRMVLと同じ扱いとな り ます。

パルス発生させない軸にはVOIDを設定します。

MOVL 100 200 300 VOID /* Z軸を除いた直線補間パルス発生

現在位置と比較してのパルス発生なので、実行前に現在位置が確定していなければなり ません。ただし、直

線補間を連続して実行する場合は発生終了待ちを省略できます。

40 RMVL 1000 2000 0 0
50 MOVL 100 200 VOID VOID

XYUを3軸直線補間した場合ACCELのパラ メータ優先順位は X>Y>Uです。

Uの出力数が最も多くなったと してもUが(DDAでいう と ころの)主軸にはならずXの移動開始 ・停止にUが同

期して高速化してしまいます。 XYサーボ、 Uステップのよ う な組み合わせでは3軸直線補間は困難です。 そ

の場合、 UはMOVSで制御しますが、 同時スタートでも終了はU軸非同期とな り ます。

ACCEL X_A XYspeed
ACCEL U_A Uspeed          /* U軸はこの値に従う
MOVL Xpos Ypos VOID VOID
MOVS U_A Upos
WAIT RR(X_A|Y_A|U_A)==0   /* 終了確認

MOVS
パルス（MPG-314専用） 軸独立パルス発生（絶対座標移動）

■書　式

単軸指定

MOVS n c
  n:軸指定予約定数 X_A～Z_A
  c:座標値

複数軸指定

MOVS x y u x
  x y u x:各軸座標値。 指定無=VOID予約定数。

■解　説

パルス発生コマンド です。速度、加速度等はACCELで設定します。入力方法には単軸指定、複数軸指定の二

通り あり ます。MOVSは直線補間はしません。複数軸指定でパルス発生をしない場合はVOIDをセッ ト します。

MOVS X_A 1000               /* X軸1000位置
MOVS Y_A -1000              /* Y軸-1000位置
MOVS 1000 -2000 VOID VOID   /* X軸1000 Y軸-2000
6-69



絶対座標移動は現在位置と比較するため、 直前のパルス発生が停止していなければな り ません。

WAIT RR(Y_A)==0             /*Y軸停止待ち
MOVS Y_A 10000

MOVT
パルス（MPG-314専用） 連続補間（絶対座標移動）
■書　式

MOVT n point [<CW,CCW>]

  n:軸指定予約定数。 2軸の論理和

  point:P(n)形式で点ﾃﾞｰﾀを指定

  CWまたはCCW:円弧補間方向

■解　説

RMVTコマンド の座標位置指定連続補間です。RMVTでは全て相対位置指定と なるため、実際のデータ入力が

大変困難です。MOVTはあらかじめ座標値を点データに指定しておく こ と により 所定の軌跡制御を実行しま

す。座標値は移動する順序に従って連続で登録します。これは、MOVTコマンド は指定された1つ前と の点と

の差を計算し、その値をRMVTと して実行するためです。点データは通常X,Y,U,Zの値ですが、MOVTでは第

一軸、 第二軸、 第一軸円弧中心、 第二軸円弧中心の順に入力します。 軸定数は必ず和で表記して下さい。

10 SETP 100 0 0 0 0
20 SETP 101 1000 0 0 0
30 SETP 102 1000 1000 1000 500
40 SETP 103 0 1000 0 0
50 SETP 104 0 0 0 500
60 PG &H410
70 ACCEL 2000
80 STPS 0 VOID 0 VOID
90 HOUT X_A;DS_DACL
100 MOVT X_A|U_A P(101)
110 MOVT X_A|U_A P(102) CCW
120 MOVT X_A|U_A P(103)
130 MOVT X_A|U_A P(104) CCW
140 MOVT X_A|U_A P(101)
150 HOUT X_A;EN_DACL

例えば、 110の文では点102を指定しています。 X_Aの移動量と してはX(102)-X(101),U_Aの移動量と しては

Y(102)-Y(101)が使用されます。CCWの円弧指定のためP(102)へ至る経路は円弧補間とな り、その円弧中心は

U(102)-X(101),Z(102)-Y(101)とな り ます。こ こではXU軸の円弧補間を含む軌跡制御ですが、座標のXYUZとい

う ラベルと適用される軸は無関係なこ とに留意して下さい。

次は同様の動作を実際のプログラムに応用したものです。 点3999を図面上の基準点と して4000～4005の各

点を設定しています。 このプログラムではD(0)～D(6)で動作モードを設定しておいてサブルーチン*UT1_dis

1で実際の動作を行います。

最初の MOVS 1000 1000 0 0は待機点で任意の点です。 次のMOVTで最初に点4000へ移動するこ とになって

いるので点3999が図面上の基準点にな り ます(MOVTは1つ前の点からの相対移動になるためです)。

点番号に相当する配列に0セッ トする と直線、 1をセッ トする とCW円弧補間、 2をセッ トする とCCW円弧補

間、 3で終端と しています。 このよ うにMOVTでは図面上の点を直接扱う こ とができます。

また、画像処理で補正する場合には、あらかじめAFFIN変換によ り各点を回転補正した上で作業点のXY座標

を修正すれば良いこ とにな り ます。

次のプログラムではMOVTとMOVTの間でいくつかのステート メン トが実行されます。 このため、点と点の

間があま りに短いと、 MOVTの連続実行が間に合わなくなる場合があ り ますので、 注意して くたさい。

その場合、位置制御そのものには矛盾が生じませんが、パルスが途切れるために機械系にスト レスを与えます。

こ こでは、 HOUT X_A;DS_DACL,HOUT X_A;EN_DACLとい う特殊なコマンドが使われています。

それぞれの意味は、 減速を無効、 減速を有効にする とい う ものです。

減速を無効にしておかないと通常のMOVS等と同様、最初のMOVTの終点で減速してしま う ために以後のパ

ルス発生が低速になってしまいます。しかし、終端では逆に減速を有効にしておかないと急停止になってし

まいますので最後のMOVTのあとで減速を有効にします。 MPG-314では ト レース移動を動的に決定できる

ために始点と終端をICに知らせるこ とが必要です。

P(103)
    0.1000

P(102)
 1000,1000

P(100)

0,0
  P(104)

1000,0
P(101)

  DIM D(100)
  SETP 3999 0 0 0 0
  SETP 4000 1000 0 0 0
  SETP 4001 2000 0 0 0
  SETP 4002 2000 1000 2000 500
  SETP 4003 1000 1000 0 0
  SETP 4004 0 1000 0
  SETP 4005 0 0 0 500
  SETP 4006 0 0 0 0
  SETP 4007 0 0 0 0
    D(0)=0
    D(1)=0
    D(2)=2
    D(3)=0
    D(4)=0
    D(5)=2
    D(6)=3
    D(7)=3
  PG &H410
  ACCEL 2000
  CLRPOS 
  MOVS 1000 1000 0 0
  WAIT RR(ALL_A)==0
  TIME 1000
  CP 
  PRINT "START"
  GOSUB *UT1_dis1
  WAIT RR(ALL_A)==0
  CP 
  END 
*UT1_dis1
  HOUT X_A;DS_DACL
  FOR Nans3=4000 TO 4999
    PRINT Nans3
      Turn=D(Nans3-4000)
    SELECT_CASE Turn
      CASE 0 : MOVT X_A|Y_A P(Nans3)
      CASE 1 : MOVT X_A|Y_A P(Nans3) CW
      CASE 2 : MOVT X_A|Y_A P(Nans3) CCW
      CASE_ELSE  : GOTO *UT1_dis2
    END_SELECT 
  NEXT Nans3
*UT1_dis2
  HOUT X_A;EN_DACL
  RETURN

MOVZ
パルス（MPG-68K互換） Z絶対座標移動

■書　式

MOVZ z

  z:絶対座標 （パルス）

MOVZ P(n)

  n:ポイン トデータ

     1≦n≦13000

■解　説

パルス発生コマンドです。MOVZは絶対座標移動です。スピード上限及び加速度はACCELコマンドで既定さ

れその範囲内でのスピード設定はFEDZコマンドで指定できます。 引き数と してはZ軸分(z)をスカラ値で与

えるこ と も、点データP(n)で指定するこ と もできます。点データP(n)を使用した場合はP(n)のZ成分だけが使

用され他の成分は無視されます。

MPC
システム MPC-816コンパチ

■書　式

MPC [1]

■解　説

MPC-684ではプログラム用コネクタの5,6番ピンのシ ョート /解放を検出し、 パワーオン リセッ ト時にプロ

グラムモードか自動実行モードかを選択します。しかし、MPC-816ではこの検出を随時行っておりプログラ

ム用コネクターを解放するこ とで直ちにプログラムを実行します。 この動作と MPC-684 を整合させるのが

このコマンドです。MPC<ENTER>でMPC-816と同じ仕様になり ます。元に戻すにはMPC 1<ENTER>と入力し

ます。

MPCINIT
メンテナンス RAMエリア初期化

■書　式

MPCINIT [n]

  n無し 初期化

  n=4 無入力警告メ ッセージ （仕事しないなら･･） の禁止　＜サポート :REV2.21以上＞

＊新しいボードは使用前にMPCINITとERASEを実行してして下さい。

■解　説

n無し MPC-684のRAM上のプログラム、 ポイン トデータ、 変数、 パラ メータエ リ アの初期化を行います。
MPC-684システムの入れ替え後や全てのデータを初期化する と きに実行します。

n=4 MPC-684はオンライン状態で一定期間入力が無いと警告メ ッセージを出力しますが、 パソコンと接

続したまま使用する場合には不都合が生じるこ とがあ り ます。 この警告メ ッセージは"MPCINIT 4"

と入力するこ とで禁止されます。

NEW
編集 プログラム初期化

■書　式

NEW

■解　説

プログラムエ リ アの初期化を行います。 点データはク リ アされません。

NEWP
編集 点データ初期化

■書　式

NEWP

■解　説

点データの初期化です。 点データの全ての座標値を0にします。

NEXT
制御文 繰り返し

FOR.NEXT参照

NOT
演算 補数

■書　式

NOT(n)

■解　説

nの補数を返します。

PRINT NOT(256)
-257
#PRX NOT(&H01)
  FFFFFFFE
#PRX NOT(&HFF)
  FFFFFF00

OFF
I/O 出力オフ

ON参照

ON
I/O 出力オン

■書　式

ON A1 [A2 A3 A4 A5 A6] 出力オン

OFF A1 [A2 A3 A4 A5 A6] 出力オフ

  A1～A6:出力ポートナンバー （変数も可）

     -1～-8192はメモ リー I/O

■解　説

出力ポートのビッ ト単位の操作です。

10  CONST sol1 0
20  CONST sol2 1
30 '
35  DO
40    ON sol1 : OFF sol2
50    TIME 500
60    OFF sol1 : ON sol2
65    TIME 500
70  LOOP
#

10  DO
20    FOR i=0 TO 47

30      ON i : TIME 500
40      OFF i : TIME 500
50    NEXT i
60  LOOP

OR
演算 論理結合式(論理和)

■書　式

（式） OR (式)

■解　説

IF SW(A)==1 OR SW(B)==1 THEN
  (制御文)
END_IF

DO
  (制御文)
LOOP UNTIL A==1 OR B==1

パルス発生の停止条件では論理結合はできません(次記の下線部)

MOVE -- UNTIL A==1 OR B==1

OUT
I/O パラレル出力

■書　式

OUT n m

  n:出力データ

     0≦n≦255

  m:出力ポートバンクナンバー

■解　説

出力ポートのバイ ト単位の操作です。nは0～255までのバイ ト値で、mは、ポートバンクの値です。例えばポー

トの0～7まではバンク0とな り ます。

10   '　バンク0～5を順にオン・オフします
20  DO
30    FOR i=0 TO 5
40     OUT 255 i : TIME 100
50     OUT 0 i : TIME 100
60    NEXT i
70  LOOP

mが-1から-1024はメモ リー I/Oです。

out &h0f -32
prx in(-32)
  000F

I/Oサイズ指定予約定数を使って出力バイ ト数を指定できます。

OUT n m~Lng /* 4バイト出力
OUT n m~Wrd /* 2バイト出力
OUT n m~Int /* 2バイト符号付出力

下記はメモ リ I/Oのパラレル入出力の例です。

#OUT &H12345678 -1~Lng/*メモリI/Oのバンク1～4へ出力
#PRX IN(-1~Lng)
 12345678

#OUT &H8000 -1~Wrd /* メモリI/Oのバンク1～2へ出力
#PR IN(-1~Wrd) /* 2バイト入力
32768
#PR IN(-1~Int) /* 2バイト符号付入力
-32768

P
パルス（MPG-68K互換） 点データ

■書　式

P(n)

     0≦n≦13000

■解　説

P(n)は特殊な予約配列で、点データを表します。点データは4つの座標値から成り立っています。従ってP(n)

をMOVE，MOVZ，GOコマンドで使用する とそれぞれベク トル値と して扱われ必要な座標成分が使用されま

す。又、printでは4つの成分を表示するこ と と しています。 この点データはX(n),Y(n),U(n),Z(n)とい う 4つの予

約配列と して使用するこ とが出来ます。これらは配列の為代入も可能です。点データにそのままデータを設

定するにはSETPコマンドがあ り ます。 ティーチングの場合は番号を指定するこ とによって直接現在位置を

点データに記録するこ とが出来ます。尚、nが0の時は現在位置を表します。各成分X(0),Y(0),Z(0),U(0)につい

ては現在位置の各成分の値を表します。 現在位置はMPG毎に異なるため現在どのMPGにアサインされてい

るかの注意が必要です。

#setp 10 100 200 300 400 /*P(10)を設定 
#print p(10) /*P(10)の表示
 100 200 300 400
#stp 50 50 50 50 /*現在位置の設定
#print p(0) /*現在位置の表示
50 50 50 50
#clp /*現在位置のクリアー
#print p(0) /*現在位置の表示
0 0 0 0
#SETP 1 X(0) Y(0) U(0) Z(0) /*現在位置をP(1)として登録
#MOVE P(10) /*P(10)にMOVE
#STP /*現在位置の表示
 X= 100 Y= 200 U= 300 Z= 0
#MOVE P(1) /*P(1)へMOVE
#STP
 X= 0 Y= 0 U= 0 Z= 0

P_HSW
I/O MPC-684 J4コネクタの入出力

P_SW参照

P_IN
I/O MPC-684 J4コネクタの入出力

P_SW参照

P_IN
I/O MPC-684 J4コネクタの高速入力

■書　式

P_IN(1,0)

パラレルラ ッチデータの読み出し。 0ならス ト ローブ未着。 ス ト ロー ブがあれば&H80以上の値にな

る。 &h7fで7ビッ トのバラレルデータを得るこ とができる。 一度読むと0に戻る。

P_IN(&H83,0)

ス ト ローブ(P_SW(7))オンでラ ッチ指定。 この時の返し値は-1。

P_IN(&H3,0)

ス ト ローブ(P_SW(7))オフでラ ッチ指定。 この時の返し値は-1。

■解　説

J4はDC24V入力のポートです。相手がTTL信号出力機器の場合、 フォ ト カプラなどによ り、オープンコレク

タに変換して使用ください。

また、J4-15,16からJ5-1へのパタンを一箇所ハンダ面でカッ ト し、RA5を 2.7K程度の抵抗に交換すれば、J4の

IOを5V対応にするこ と も可能です。 (ユーザ責任)

10 'p_sw(7) on strobe
20 PRINT P_IN(&H83,0)
30 'when p_sw off strb
40 ' PRINT P_IN(&H3,0)
50 *test_start
60 DO
70 aaa=P_IN(1,0)
80 IF aaa<>0 THEN : BREAK LOOP : END_IF
90 TIME 1000 : PRINT "Not Yet !!"
100 LOOP
110 PRX &H7F&aaa

P_LD
ファイルメモリ フラッシュ ROMから点データを読み込む

■書　式

P_LD [begin,end]

  begin:読込み開始点データ

  end:読込み終了点データ

■解　説

フラ ッシュ保存された点データを取り出して復旧します。引数が無いと全点データ、指定する とその範囲の

み復旧します。

#P_LD 100 110 ←点100から110のみを取り出してP(100)-P(110)に書き込みます。

P_SV参照

P_OFF
I/O MPC-684 J4コネクタの入出力

P_SW参照

P_ON
I/O MPC-684 J4コネクタの入出力

P_SW参照

P_OUT
I/O  MPC-684 J4コネクタの入出力

P_SW参照

P_SV
ファイルメモリ フラッシュ ROMに点データを書き込む

■書　式

P_SV

■解　説

点データをフラ ッシュエ リ アに書き込みます。書き込み中 （数秒） はマルチタスクが停止しますので、他の

タスクの作業が休止中に実施して ください。 

#P_SV
**-- ←保存中表示

P_LD参照

P_SW
I/O MPC-684 J4コネクタの入出力

■書　式

P_ON A

  A:ポート番号

     0≦A≦3

P_OFF A

  A:ポート番号

     0≦A≦3

P_OUT n

  n:出力パターン

     0≦n≦&HF

P_SW(n)

  n:ポー ト番号

     0≦n≦7

P_HSW(n)

  n:ポー ト番号

     0≦n≦7

P_IN(0)

■解　説

P_ON ポート （J4） のビッ トオン

P_OFF ポート （J4） のビッ トオフ

P_OUT ポート （J4） のパラレルアウ ト

P_SW(n) ポート （J4） のビッ ト入力 （5msecﾌｨﾙﾀｰ付き)

P_HSW(n) ポート （J4） のビッ ト入力 （ﾌｨﾙﾀｰ機能無し）

P_IN(n) ポー ト （J4） のパラレル入力

前記のコマンドはいずれもMPC-684に付属のI/Oポート J4に対するコマンドでそれぞれMIP/MOPに対する

ON/OFF/SW(n)などのコマンド ・関数に対して同等の機能を持ちます。ポート番号とピンアサインの関係は

次の通りです。

#P_ON 0 ’OP1をON
#PRX P_IN(0) ’パラレル入力上位バイトが出力、下位バイトが入力
 0100 ・・・出力はOP1がON、入力は全部OFF
#PRINT P_SW(0) ’IN1のビット入力
0 ・・・入力IN1はOFF
#P_OUT &H03 ’パラレル出力
#PRX P_IN(0)
 0300 ・・・出力はOP1,OP2がON入力は全部OFF

前記例のよ うに、P_IN(0)に対しては、下位バイ トが入力ポート、上位バイ ト (4bitのみ)が出力ポートのビッ ト

パタンとな り ます。 尚、 このコマンドはMPC-684に対して有効です。 MPG-68KのJ4に対しては、 HPT(入力

) HOUT(出力)を使います。

※MPG-68Kの原点入力のJ4コネク ターの入出力はHOUT,HPT(n)でおこないます。

PALET1
パルス（MPG-68K互換） パレット宣言

■書　式

PALET1 P(i) P(j) P(k) m n

PALET2 P(i) P(j) P(k) m n

PALET3 P(i) P(j) P(k) m n

PALET4 P(i) P(j) P(k) m n

  i,j,k:ポイン トナンバー

  m,n:マ ト リ クス数

■解　説

パレッ ト宣言です。P(i),P(j),P(k)によって構成された長方形（平行四辺形） をm,n分割したパレッ ト （マ ト リ ク

ス） と します。 パレッ トは1～4迄であ り設定されたパレッ トの点は関数PL1(n)～PL(n)にて使用します。 パ

レッ ト計算は4次元で実施されますので傾斜のあるパレッ ト も対応可能です。

ピン コマンド ・ 関数

 1

 2

 3

 4

 5

 6

 7

 8

 9

10

11

12

P_SW(0)/P_HSW(0)

P_SW(1)/P_HSW(1)

P_SW(2)/P_HSW(2)

P_SW(3)/P_HSW(3)      P_IN(0)   入力

P_SW(4)/P_HSW(4)

P_SW(5)/P_HSW(5)

P_SW(6)/P_HSW(6)

P_SW(7)/P_HSW(7)

P_ON 0/P_OFF 0

P_ON 1/P_OFF 1      P_OUT n   出力

P_ON 2/P_OFF 2

P_ON 3/P_OFF 3

10    PG &HE0 0
20    PALET1 P(1) P(2)P(3) 5 5 /*PALET1宣言
30    PALET2 P(11) P(12) P(13) 10 10 /*PALET2宣言
40      a=0 :   b=0
50   '
60    DO
70        a=a+1
80      IF a>25 THEN
90          a=1
100     END_IF
110     JUMP PL1(a) /*PALET1へJUMP
115  '
120       b=b+1
130     IF b>100 THEN
140         b=1
150     END_IF
160     JUMP PL2(b) /*PALET2へJUMP
170   LOOP
#

PALET2
パルス（MPG-68K互換） パレット宣言

PALET1参照

PALET3
パルス（MPG-68K互換） パレット宣言

PALET1参照

PALET4
パルス（MPG-68K互換） パレット宣言

PALET1参照

PAUSE
タスク操作 タスク一時停止

■書　式

PAUSE A

  A:タス クナンバー   1～31

■解　説

タスクの一時停止。 実行中のタスクを一時停止するものです。

10    FORK 1  *task1
20    DO
30      WAIT SW(192)==1
40      PAUSE 1
50      WAIT TASK(1)==-3
55      PRINT "task1 pause"
60      WAIT SW(192)==0
70     CONT 1
75     WAIT TASK(1)==0
77     PRINT "task1 continue"

80    LOOP
90   '
100  *task1
110   DO
120     FOR i=0 TO 47
130      ON i :TIME 100 :OFF i : TIME 100
140      NEXT i
150   LOOP

PEEK
メモリアクセス ユーザーメモリ読み出し

■書　式

PEEK(adrs)

  &HC0000≦adrs≦&HFFFFF

■解　説

ユーザメモ リの値をワード読みだしします。 adrsが奇数指定されても偶数に切り捨てされます。

PG
パルス（MPG-314専用） PG宣言

■書式

PG adrs [task]  

    adrs:MPG-314ボードアドレス  &H400,&H410～&H490

    task:タスク番号

    (パラ メータ省略で現在の設定値表示)

■解説

MPG-314の使用宣言文です。 adrsによって指定されたMPG-314をtaskで指定されたタスクに引き当てます。

例えば PG &H400 1 とすればタスク1で実行されるMOVEやACCELはアドレス &H400のMPG-314に対して

のコマンド とな り ます。 この宣言が一度実行される と RAM上に保存されリセッ トの度にインターフェース

は初期化されます。 PG<ent>とする と、 現在の引き当て状況のリ ス ト を表示します。

尚、 PGはティーチング中の切り替えコマンドではあ り ません。 この場合はPGSELを使用して ください。

#PG &H400 0 /* タスク0で&H400のMPG-314を制御
#PG &H410 1 /* タスク1で&H410のMPG-314を制御
#PG &H410 2 /* 1枚のMPGを複数のタスクで制御することもできる(単軸制御など)
#PG /* 引き当て確認
      指定されているPGアドレス-&H400

        TASK  0 for PG &h400    TASK  1 for PG &h410
        TASK  2 for PG &h410    TASK  3 for PG &h000
                            (以下略)

タスク番号を省略する とPGコマンドを実行したタスク自身への引き当てになり ます。

  FORK 1,*TASK1 /* TASK1起動
  PG &H400 /* &H400のMPG-314をTASK1へ引き当て
  ACCEL X_A|U_A -1 10000 /* このACCELは&H400のMPG-314へ発行される
  (略)
  END
*TASK1
  PG &H410 /* &H410のMPG-314をTASK1へ引き当て
  ACCEL X_A|U_A -1 10000 /* このACCELは&H410のMPG-314へ発行される
  (略)

1つのタスクの中でPG宣言を切り替えて使う こ と もできます。

PG &H400 /* 次からのコマンドは&H400のMPG-314へ
ACCEL X_A|U_A -1 10000
FEED X_A|U_A 0
PG &H410 /* 次からのコマンドは&H410のMPG-314へ
ACCEL X_A|U_A -1 10000
FEED X_A|U_A 0

PG
パルス（MPG-68K互換） PG宣言

■書式

PG adrs [task]  

    adrs:MPG-68K/405ボードアドレス  &HE0,&HE4,&HE8,&HEC...

    task:タスク番号

    (パラ メータ省略で現在の設定値表示)

■解説

MPG-68K/405の使用宣言文です。 adrsによって指定されたMPGをtaskで指定されたタスクに引き当てます。

例えば PG &HE0 1 とすればタスク1で実行されるMOVEやACCELはアドレス &HE0のMPGに対してのコマ

ンド とな り ます。

この宣言が一度実行される とRAM上に保存されリセッ トの度にインターフェースは初期化されます。PG<en

t>とする と、 現在の引き当て状況のリ ス ト を表示します。

尚、 PGはティーチング中の切り替えコマンドではあ り ません。 この場合はPGSELを使用して ください。

#PG &HE0 0 /* タスク0で&HE0のMPGを制御
#PG &HE4 1 /* タスク1で&HE4のMPGを制御
#PG &HE8 2 /* タスク2で&HE8のMPGを制御
#PG /*引き当て確認
      指定されているPGアドレス-&H0E0
        TASK  0 for PG &h0E0    TASK  1 for PG &h0E4
        TASK  2 for PG &h0E8    TASK  3 for PG &h000
                           (以下略)

PGSEL
パルス（MPG-68K互換） PG宣言

■書　式

PGSEL [n]

  0≦n≦23

■解　説

PGコマンドではタスク と スレーブボードを結合しますが、それでは調節時に支障をきたします。調整時のコ

マンド実行、 ティーチモードは全てタスク0によって実施されます。 PGSELはこの問題を解決するためにタ

スク0に一時的に他のタスクのMPCを割り付けものです。PGSEL<ENTER>の後<TAB>，<+>，<->のキーによっ

て希望のMPGを選択する と以降コマンドは選択されたMPGに対して有効とな り ます。 PGSELによる選択は

RUN、 リセッ トによ り ク リ アされます。ティーチモードにも この機能は含まれています。又、引き数を与え

れば直接目的のMPGを選択します。

pgsel
 PG[1,E4]
#

スレーブボードのアドレス

タスクナンバー 

  <TAB>， <+>キーでアップ

  <->キーでダウン

  <Q>で編集画面へ戻る

PL1
パルス（MPG-68K互換）  パレットポイント

■書　式

PL1(n)

PL2(n)

PL3(n)

PL4(n)

  n:パレタイズマ ト リ ッ クスナンバー。 ｎが負の値のと きはZIGZAG動作とな り ます。

■解　説

PALET1～4で設定されたパレッ トの点データを与えます。nの値は1から始ま りx方向(P(i)→P(j)でm分割)に増

加します。mに達する とx方向はリセッ ト され、y方向(P(i)→P(k)で分割)に1ピッチ増加します。計算式では次

のよ うにな り ます。 (m-1),(n-1)が最後にあるのは精度確保の為除算を最後にしている為です。

dx=P(j)-P(i)
dx=P(k)-P(i)
PL(n)=((n-1)%m*dx/(m-1)+((n-1)/m)*dy/(n-1)+P(i)

例えばn=1でPLはP(i)そのものとな り ます。 nがmの場合は次のよ うにP(j)そのものとな り ます。

PL(n)=(m-1)*dx/(m-1)+P(i)
PL(n)=dx+P(i)=P(j)

尚、 パレッ ト ポイン トの座標成分を必要とする場合は次のよ うにパレッ トデータを点データに移してから

使用して下さい。

SETP 3 PL1(n)
MOVE X(3) Y(3) 0

パレタイズマ ト リ ッ クスナンバーが負の値のと きは次の図の動きになり ます。

列間の移動が小さ く スピードアップにな り ます。

※注意
パレタイズマ ト リ ッ クスナンバーがカウン トオーバーしても、同じピッチで継続します。プログラムで進行

状態を管理して ください。

PL2
パルス（MPG-68K互換） パレットポイント

PL1参照

 

1    2    3    4 

5    6    7    8 

9   10   11   12 
 

-1   -2   -3   -4 

-8   -7   -6   -5 

-9  -10  -11   -12 

nが正数のとき              nが負の数のとき 

PL3
パルス（MPG-68K互換） パレットポイント

PL1参照

PL4
パルス（MPG-68K互換） パレットポイント

PL1参照

PLIST<PLS>
編集 点データ表示

■書　式

PLIST  （省略形　PLS）

■解　説

P(n)の一括表示です。 1画面単位で表示の継続中止を聞いてきますので、 中止の場合はq/Qを入力して下さ

い。 どれかのキーを押すと継続します。

PLSC
パルス（MPG-314専用） 一定速パルス発生

■書　式

PLSC n m [pls] [n1,m1,pls1]

  n:軸指定予約定数

  m:パルスレート (pps)

  pls:パルス量(省略時無限パルス発生)

※デューティーは50%固定

■解　説

一定速パルス発生です。加減速はあ り ません。引数の与え方で定量パルス発生にも、無限パルス発生にも使

用でます。注意と して指定パルスレート値によってMCX-314上のACCELパラ メータが破壊されるのでPLSC

実行後に通常のパルス発生コマンドを使用する場合は、 PRSET_ACCELを実行します。

移動量を規定しない無限パルス発生ではplsを省略します。 この場合のCW/CCW方向規定はmの正負によっ

て決定されます。 移動量を規定する場合はplsの値の正負によって方向が決定されます。

例では、X軸でパルス発生してXS2(IN1)入力で停止、その後PRSET_ACCELによってACCEL設定を回復し通常

のRMVSを使用しています。

PRSET_ACCELがないとRMVSでのパルス発生も加減速無しの一定速とな り ます。

10 PG &H410
20 ACCEL 10000
40 STOP X_A IN1_ON
60 PLSC X_A -100
70 WAIT RR(X_A)==0
80 PRSET_ACCEL X_A 
90 RMVS X_A 1000

POKE
メモリアクセス ユーザーメモリ書き込み

■書　式

POKE n adrs

  0≦n≦HFFFF

  &H200000≦adrs≦&2FFFFF (684)

■解　説

ユーザメモ リに対してワード単位でデータを書き込みます。 奇数アドレスを指定しても切り捨てによ り偶

数アドレス と して扱われます。

PRF
コプロ演算 内部データ表示

■書　式

PRF fn1,fn2,fn3,..

  fn1,fn2,fn3,..:FPnの番号  7を与える とアキュームレータ

■解　説

コプロの不動小数点レジスタFP0,FP1..のデータを表示します。

PRINT<PR>
RS-232 データ表示

■書　式

PRINT  (省略形　PR)

PRINT#0

PRINT#2

■解　説

CH1文字列 ・数値の表示 CH0文字列 ・数値の表示 CH2文字列 ・ 数値の表示

引き数間にタブ、スペースなどのデリ ミ タはあ り ません。また、PRINT#では改行コードすら出力されません。

従って、外部との通信はPRINTコマンドのみで自由なフォーマッ ト を構成するこ とが出来ます。また次の文

字列は特別な意味を持ちます。

¥n　ニューラインコード(CR-LF)

¥r 　 リ ターンコード

¥t　 タブコード

PRINT "HEX$(ASC(xyz$))=" HEX$(ASC(xyz$))
PRINT CHR$(ASC("A")) ASC(CHR$(ASC("A")))
PRINT 1 "¥t" 2 "¥t" 3

次の場合はCH0よ り&H1B &H2A &H30 &H31 &H32と出力されます。

PUT#0 &H1B &H2A
PRINT#0 "012"

PRINT#で通信相手側がCR,LFコード を必要とする場合(N88BASIC INPUTｺﾏﾝﾄﾞなど)には次のよう に記述します。

PRINT#0 ACCEL¥n 文字列“ACCEL”とCR･LFｺｰﾄﾞを出力
variable=123

PRINT#0 variable CHR$(13) CH0へ数値123とCRｺｰﾄﾞを出力
PRINT#2 variable CHR$(13) CHR$(10) CH2へ数値123とCR･LFｺｰﾄﾞを出力

PRINT#でのRS-232バッファの内容はRSコマンドで確認できます。

■パラ メーター間のスペースが無いと...
PRINT#2 a$ " ok¥n" とする と ころを a$" ok¥n" （a$と" ok¥n"の間の

ｽﾍﾟｰｽが無い） と してしま う と、 a$" ok¥n"は数値が0の1つの変数と し

て扱われてしまいます。MPCで①のプログラムを実行する とCH2に接

続したターミナル側では②の様になり ます。

①MPC側のプログラム

10        CNFG#2 "9600b8pns1NONE"
20        PRINT#2 "¥n"
30        DO 
40          INPUT#2 "a$=" a$
45          PRINT a$
50          PRINT#2 a$" ok¥n" <--送られてきた文字列を返したいのだが..
60        LOOP 

②ターミナル表示画面

a$=0a$=0a$=0a$=0a$=0a$=0a$=0a$=0a$=0a$= <--返ってくるのは0ばかり

③CH1でも同様です。

#PRINT a$ <--a$は1です。
1
#PRINT a$" ok¥n" <-- a$" ok¥n"は一つの変数として扱われます。
0
#PRINT a$ok¥n <-- a$ok¥nは一つの変数として扱われます。
0
#PRINT a$ ok¥n <--この場合はa$とok¥nという2つの変数になります。
10
#PRINT a$ "ok¥n" <--これがベスト。

※""の中のスペースは文字と して扱います。

PRINT#0
RS-232 データ表示

PRINT参照

PRINT#2
RS-232 データ表示

PRINT参照

実験機器構成

MPC-684

ﾊﾟｿｺﾝ

Windows

FTMW

ﾊﾟｿｺﾝ

Windows

ﾀｰﾐﾅﾙｿﾌﾄ

CH1 CH2

PROTOCOL
RS-232 MBK-RS（デジタルGPダイレクトアクセス）

■書　式

PROTOCOL MEWNET

■解　説

MPC-684のCH0で松下のMEWNETプロ ト コルに準拠した通信を行います。 デジタル社GPシ リーズとダイレ

ク ト アクセス方式で接続できます。

"PROTOCOL MEWNET"とプログラムに一行加筆する と有効になり ます。

停止する場合はPROTOCOL MEWNETの行を削除の上"PROTOCOL 0" も し くはMPCINITを実行します。

※注意

この通信は9600bpsで、 CPUにも負担がかかり ます。 高速通信、 高タク ト タイムが必要な場合はMBK-SHを

使って ください。

PRSET_ACCEL
パルス（MPG-314専用） ACCELパラメータ復旧

■書　式

PRSET_ACCEL n

  n:軸指定予約定数 X_A～Z_A

■解　説

ppsを指定する一部のコマンド(PLSC,HOME X_A)によって破壊されたACCELパラ メータを復旧します。

PRSET_ACCELはIN0～3条件停止も ク リ アします。

PRSET_ACCEL X_A /* X軸のパラメータ復旧
PRSET_ACCEL Z_A U_A /* Z,U軸のパラメータ復旧

PRX
RS-232 データHEX表示

■書　式

PRX A

  A:整数値， 変数

■解　説

PRINTと同様の目的のコマンドですが表示がヘキサ表現とな り ます。I/Oの状態を直接見たい時に有効です。

#a=&HFF
#PRINT a <--10進表示
 255
#PRX a <--ヘキサ表示
 00FF
#PRX IN(24)
 0055

PUT
RS-232 データ表示

■書　式

PUT [A1 A2 A3 A4 A5 A6]

PUT#0 [A1 A2 A3 A4 A5 A6]

PUT#2 [A1 A2 A3 A4 A5 A6]

  A1～A6:キャ ラ ク ター

■解　説

PUT

コンソールへ1文字ずつ出力します。

#put &h41 &h42 &h43 &hd &ha
ABC
#

PUT#0,PUT#2

A1～A6の数値をアスキーコード と してCH0も し くはCH2によ り出力します。 特にPRINT#ではヌルコードが

出力できないためにこのコマンドが有効です。 例えば、 NULL'A'と出力するには次の通りです。

PUT#0 0 &H41

PUT#0
RS-232 データ出力

PUT参照

PUT#2
RS-232 データ出力

PUT参照

Q_PAUSE
パルス（MPG-68K互換） クイックポーズ

■書　式

Q_PAUSE ク イ ッ クポーズ設定

Q_PAUSE -1 ク イ ッ クポーズ設定解除

注意） コマンドの後ろにコ メン ト を記述しないで下さい。

Q_PAUSE 'ｸｲｯｸﾎﾟｰｽﾞﾓｰﾄﾞ <--Q_PAUSE -1と同じになります。

■解　説

STOPによるパルス発生停止後にタスクのPAUSEを自動的に行います。コマンドQ_PAUSEを実行する と次の

よ うにSTOP コマンドに機能が追加されます。 これまでSTOP コマンドを実行する と発生中のパルスは停止

させられましたがパルス発生に続く コマンドはそのまま実行されていました。これに対して、Q_PAUSEモー

ドではSTOPコマンドの実行によ りパルス発生を中止し、 さ らにそのタスクをポーズと します。 状態の監視

はBSY()関数とTASK()関数によって行います。再開はCONTコマンドのみで有効とな り ます。途中で停止させ

られたパルス発生は強制的に再実行されます。この時RMOV，RMVZ，RM、RULSEの4つのコマンドは再実行

されません。これは相対移動の場合目的地が明確で無いこ とによ り ます。従ってパルス発生中のタスクの停

止 ・再開が複雑なインタロ ッ クによる こ とな く実現できます。 Q_PAUSEモードは"Q_PAUSE -1"で解除され

ます。 工場出荷時は"Q_PAUSE -1"の状態でこれまでのソフ トがそのまま動作します。 "Q_PAUSE"有効下で

はSTOP 3も有効とな り ます。 これはパルス発生の終了時 （1つのコマンド終了時） にただちにポーズとする

機能です。 STOP 1,2の途中停止無しのモードです。

･ STOPコマンドが有効になるのはPG宣言されているタスクだけです。 PG宣言されていないタスク

をポーズするのはPAUSEコマンドです。

･ CONTコマンドはEN_PGも併せて実行します。

･ UNTILによる条件停止ではPAUSE状態になり ません。

10       Q_PAUSE 
20       PG &HE4 0
30       PG &HE4 1
40       FORK 1 *TASK1
50       DO 
60         WAIT SW(192)==0
70         WAIT SW(192)==1
80         STOP 1 1 <--ﾀｽｸ1でMOVE中に減速停止をかけます
90         WAIT TASK(1)==-3 <--ﾀｽｸ1がPAUSE状態になるのを待ちます
100        PRINT X(0) Y(0) <--停止した所の座標値を表示
110        WAIT SW(192)==0
120        WAIT SW(192)==1
130        CONT 1 <--PAUSE中のﾀｽｸ1を再開します
140        WAIT TASK(1)<>-3
150      LOOP 
160      END 
170    *TASK1
180      SETPOS 0 0 0 0
190      DO 
200        MOVE 100000 100000 0 <--RMOV,RMVZ,RM,PULSEｺﾏﾝﾄﾞは無効
210        MOVE 0 0 0
220      LOOP 
230      END 
#RUN
40602 40602 <--SW(192)がONして減速停止した所の座標値です。
84677 84677    ﾀｽｸ1は停止した所でPAUSE状態になり、
84826 84826    CONTで再び再開してﾊﾟﾙｽ発生を行います。
50639 50639    到達座標はMOVEで与えられた座標です。
27944 27944

前記プログラムの文番号80のSTOP 1をSTOP 3にする と

80         STOP 3 1

#RUN
100000 100000  <--MOVEで与えられたﾊﾟﾙｽを出し終わってからﾀｽｸ1はPAUSE状態になります。
0 0
100000 100000

非常停止、 サイクルス ト ップSW、 ポーズSW監視タスクから別タスクのパルス発生を停止する。

  TIME 100
  SETIO 
  Q_PAUSE <--Q_PAUSEﾓｰﾄﾞ宣言
'  IN PORT
  CONST pause_sw 192 <--ﾘﾐｯﾄｽｲｯﾁ
  CONST emg_sw 193 <--非常停止ｽｲｯﾁ
  CONST cycle_sw 194 <--ｻｲｸﾙ停止ｽｲｯﾁ
'  OUT PORT
  CONST pause_led 0
  CONST emg_led 1
  CONST cycle_led 2
  FORK 1 *PULSE
'************************************
' 非常停止、PAUSE SW監視
'************************************

  PG &HE0
*MAIN
  DO 
    IF SW(pause_sw)==1 THEN
      GOTO *PAUSE
    END_IF 
    IF SW(emg_sw)==1 THEN
      GOTO *EMG
    END_IF 
    IF SW(cycle_sw)==1 THEN
      GOTO *CYCLE
    END_IF 
  LOOP 
*PAUSE
  PRINT "PAUSE SW停止処理"
  STOP 1 1 <--減速停止
  WAIT TASK(1)<0 <--ﾀｽｸが実行中でなくなるのを待つ
  GOSUB *TASK_STAT
  PRINT "現在点は" P(0)
  ON pause_led
  WAIT SW(pause_sw)==0
  OFF pause_led
  CONT 1
  WAIT TASK(1)>=0
  GOSUB *TASK_STAT
  GOTO *MAIN
*EMG
  PRINT "EMG SW処理"
  STOP 2 1 <--急停止
  WAIT TASK(1)<0 --+ ﾀｽｸ1が実行を停止したら
  QUIT 1 --+ QUITします。
  WAIT TASK(1)==-2
  GOSUB *TASK_STAT
  DO 
    ON emg_led
    TIME 100
    OFF emg_led
    TIME 100
  LOOP 
*CYCLE
  PRINT "CYCLE停止処理"
  STOP 3 1 <--MOVEに与えられたﾊﾟﾙｽを出力してから停止
  WAIT TASK(1)<0
  GOSUB *TASK_STAT
  PRINT "現在点は" P(0)
  ON cycle_led
  WAIT SW(cycle_sw)==0
  OFF cycle_led
  CONT 1
  WAIT TASK(1)>=0
  GOSUB *TASK_STAT
  GOTO *MAIN
*TASK_STAT <--これはﾀｽｸ1の状態を表示するｻﾌﾞﾙｰﾁﾝです
  SELECT_CASE TASK(1)
    CASE -1 : PRINT "ﾀｽｸ未使用"
    CASE -2 : PRINT "ﾀｽｸQUIT状態"
    CASE -3 : PRINT "ﾀｽｸPAUSE状態"
    CASE_ELSE  : PRINT "ﾀｽｸ実行中"
  END_SELECT 
  RETURN 
'*************************************
'パルス発生タスク
'*************************************
*PULSE
  PG &HE0
  ACCEL 10000
  CLRPOS 
  DO 

    MOVE 100000 100000 100000
    TIME 100
    MOVE 0 0 0
    TIME 100
  LOOP 

■実行結果
前記プログラムをRUNして順番にSW(pause_sw)をON/OFF→SW(cycle_sw)をON/OFF→SW(emg_sw)をONして

みます。

#RUN
PAUSE SW停止処理 <--SW(pause_sw)==1
ﾀｽｸPAUSE状態
現在点は63855 63855 63855 0 <--STOP 1 1は減速停止後ﾀｽｸ停止
ﾀｽｸ実行中 <--SW(pause_sw)==0でﾀｽｸ再起動,ﾊﾟﾙｽ再出力
CYCLE停止処理 <--SW(cycle_sw)==1
ﾀｽｸPAUSE状態
現在点は100000 100000 100000 0 <--STOP 3 1 はMOVE終了後にﾀｽｸ停止
ﾀｽｸ実行中 <--SW(cycle_sw)==0
EMG SW処理 <--SW(emg_sw)==1
ﾀｽｸQUIT状態

QUIT
タスク操作 タスク停止

■書　式

QUIT A

QUIT

  A:タス クナンバー　1～31

■解　説

QUIT AではAで指定されたタスクを停止します。

QUITのみの場合は全タスク （0を除く） の停止とな り ます。

QUIT 1 2 3 ・ ・ ・ ・ ・ タス ク1,2,3を停止

タスク0 （メ インタスク） は停止できません。

【Question】

a) も しSTOPコマンド及びBSY()を使用しないでパルス発生中のタスクをQUITした場合はど うな り ますか？

b) パルス発生の停止を確認後、 タスクをQUITしたと き、 MPCの現在位置とパルスは一致していますか。

c) a,bの場合、 再度FORKして、 パルスを続行できますか？

【Answer】

a) パルス発生中のタスクに （STOPしないで） QUITをかける と、 MPGに対する通信が折られてしま うため、

次から使えなくな り ます。 パワーオン リセッ トが必要となってしまいます。

b)途中停止した場合でもパルス数と現在位置は一致します。 （メ カ的にズレがないと して）

c) タスクの停止とMPGの状態は、 正規の手続き （通信中にQUITするなどは御法度です） さえ踏まれていれ

ば無関係なので、 もちろん停止、QUIT後の再FORKも大丈夫です。ただ、実際にパルス発生中であるかど

うかの確認はBSY()だけでは難し く、 問題を起こ しやすいのでクイ ッ クポーズを使って下さい。 例えば現

在位置を取得するX(0)もMPGと通信をしますが、 BSY()では時間が短すぎて検出しにく く、 停止の確認後

QUITとい う技がう ま くできません。

Q_PAUSE参照

_RET_VAL
制御文 戻り値の受け取り

■書　式

_RET_VAL var1 var2 - -

  var:変数

■解　説

RETURN戻り値を受け取り ます。

RAM
デバッグ RAMモード

■書　式

RAM

■解　説

フラ ッシュ ROM書き込みを無効にします。 プログラム実行時(RUN)のフラ ッシュ ROMへの書き込みをキャ

ンセルします。 編集後の実行が効率的になり ます。 また、 起動時のフラ ッシュ ROMからのプログラムの読

み込みと変数初期化もキャンセルされます。RAMモードではプログラムはプロテク ト されません。デバッグ

が終了したらROMコマンドで元に戻しRUNまたはFIXでフラ ッシュ ROMへ書き込んで下さい。

#RAM ←RAMﾓｰﾄﾞ
**
#
#MPC-68K ADVFSC(r)s REV-2.52i
 BASIC like + multi tasking
 Created by ACCEL Co.'91~97
･････!! ROM化できません。ERASEしてください。 ←RAMﾓｰﾄﾞ起動時のﾒｯｾｰｼﾞ
#

#ROM ←ﾌﾗｯｼｭROMﾓｰﾄﾞ
**

#

RANGE
パルス（MPG-314専用） 動作範囲の制限

■書　式

RANGE n max min

  n:軸選択予約定数 X_A～Z_A

  max:動作範囲上限

  min:動作範囲下限

■解　説

MPG-68K互換RANEGコマンドは暗黙でSLMT_ON(ソフ ト リ ミ ッ ト )を有効にしますが、 このRANGE設定を使

うためにはINSET_314コマンドで明示的にSLMT_ONを有効にします。

(このRANGEは入力条件の設定には関係しません）

RANGE X_A 1000 -1000
INSET_314 X_A SLMT_ON

RANGE
パルス（MPG-68K互換） 動作範囲の制限

■書　式

RANGE n

  n=0 領域設定を解除します。

  n=1 現在位置を最大領域と して設定します。

  n=2 現在位置を最少領域と して設定します。

  n=11 最大領域の値を現在位置に複写します。 現在位置が失われるので注意

  n=12 最少領域の値を現在位置に複写します。 現在位置が失われるので注意

■解　説

このRANGEコマンドはMPG-68K互換コマンドで、 ソフ ト リ ミ ッ ト (SLMT_ON)を自動的に有効にします。

次のプログラムの110行でSLMT_ONをorしているのは、 INSET_314で入力条件を設定した時RANGE有効状態

を保つためです。

(INSET_314はパラ メータで与えられたもの以外はリセッ ト します）

10 SETP 1 1000 500 400 200
20 SETP 2 -1000 -500 -400 -200
30 PG &H410
40 ACCEL 8000
50 SETPOS X(1) Y(1) U(1) Z(1)
60 RANGE 1
70 SETPOS X(2) Y(2) U(2) Z(2)
80 RANGE 2
90 CLRPOS
100 CP
110 INSET_314 ALL_A INP_OFF|SLMT_ON

また、RANGEによって停止したかど うかは、RR()関数とSLMP@(正方向リ ミ ッ ト )SLMM@(負方向リ ミ ッ ト )を

組み合わせて検出するこ とができます。

RR(X_A,SLMP@¦SLMM@)

RR(Y_A,SLMP@¦SLMM@)

RR(U_A,SLMP@¦SLMM@)

RR(Z_A,SLMP@¦SLMM@)

REG
MPG-3202 X3202レジスタ読み込み

■書　式

REG(reg)

  reg: X3202のアドレス と レジスタ ・ カウンタセレク ト コードの和。 またはステータスレジスタアドレス。

■解　説

MPG-3202のパルス発生IC 「X3202」 のレジスタを読込みます。 regにステータスレジスタアドレスを指定す

る と動作状態を知るこ とができます。

MPG-3202関係コマンド ・ 関数:CMND,REG,REG3,ST_REG

参照: 「MPG-3202 製品別マニュアル」　 ･･･web参照または営業係へ請求下さい。

PRINT REG(&H121) /* X3202#1のﾚｼﾞｽﾀ&H21読込み
WAIT REG(-1)=&H20 /* X3202#1のｽﾃｰﾀｽﾚｼﾞｽﾀ読込み。この場合動作完了待ち

REG3
MPG-3202 X3202レジスタ読み込み

■書　式

REG3(reg)

  reg: X3202のアドレス と レジスタ ・ カウンタセレク ト コードの和。 またはステータスレジスタアドレス。

■解　説

MPG-3202のパルス発生IC 「X3202」 のレジスタを読込みます。 3byte符号付きのレジスタの読込みに使用し

ます。

MPG-3202関係コマンド ・ 関数:CMND,REG,REG3,ST_REG

参照: 「MPG-3202 製品別マニュアル」　 ･･･web参照または営業係へ請求下さい。

PRINT REG3(&H121) /*X3202#1のﾚｼﾞｽﾀ&H21読込み

RENUM<RNM>
編集 文番号ふりなおし

■書　式

RENUM

■解　説

文番号のふりなおし10毎にふりなおされます。

list 0
5     FOR i=0 TO 47
7       ON i : TIME 100
10      OFF i : TIME 100
15    NEXT i
#rnm
#list 0
10    FOR i=0 TO 47
20      ON i : TIME 100
30      OFF i : TIME 100
40    NEXT i

RETURN<RET>
制御文 リターン

■書　式

RETURN

■解　説

サブルーチン リ ターン GOSUB で呼ばれたサブルーチンは RETURN によってメ インルーチンへ戻り次のス

テップが実行されます。

RETURN
制御文 リターン 戻り値渡し

■書　式

RETURN [arg1 arg2 - -]

  arg:変数、 定数

■解　説

サブルーチンの結果を引き渡します。 ローカル変数と組み合わせる と タスク間でのサブルーチンの共有が

可能とな り ます。 受け取りは_RET_VALはで行います。

10 GOSUB *SUB
20 _RET_VAL A /* 戻り値の受け取り
30 PRINT A
40 END
50 *SUB
60  C!=123 /* ﾗﾍﾞﾙ!はﾛｰｶﾙ変数
70 RETURN C! /* C=戻り値
RUN
123

RLS
タスク操作 セマフォ開放

RSV参照

RM
パルス（MPG-68K互換） 4軸相対座標移動

■書　式

RM x y u z

  x,y,u,z:相対移動量 （パルス）

■解　説

4軸同時の相対パルス発生です。x y u zによって指定された値だけパルス発生します。RMもGO同様RMOVに

比べてスピードが遅くなっています。 座標値は移動した分だけ加算されます。

RMOV
パルス（MPG-68K互換） XYU軸相対座標移動

■書　式

RMOV x y u

  x,y,u:相対移動量 （パルス）

■解　説

3軸同時の相対パルス発生です。 x y uで指定された値だけパルス発生します。 座標値は移動した分だけ加算

されます。

RMVC
パルス（MPG-314専用） 無限パルス発生

■書　式

RMVC n m [n1,m1,n2,m2]

  n:軸指定予約定数

  m:方向指定 +1または-1

■解　説

移動量を指定しない無限パルス発生です。加減速はACCEL、速度はFEED指定に従います。 nは軸指定、mは

+1,-1で方向指定のみ。引数を続けて与えて連続で起動するこ とができます。通常ニアオリジンなどへ高速復

帰する場合などに用います。

次はX_IN0をニアオ リ ジンと想定して加減速付き高速移動でパルス発生する例です。

10 PG &H410
20 ACCEL 4000
30 HOUT 2 (X_A;NOP,1)
40 RMVC X_A -1
45 WAIT RR(X_A)==0
50 HOUT 0 (X_A;NOP,1)

RMVL
パルス（MPG-314専用） 直線補間（相対座標移動）

■書　式

RMVL x y u z

  x y u z:各軸パルス (最大3軸まで)

■解　説

直線補間パルス発生コマンドです。RMVL x y u zと しますが、4軸直線補間はハード的に不可能なので0以外

の値を設定できるのは3軸までです。

直線補間の場合の速度加速度等はACCEL設定に従いますが各軸の値が異なる場合は"主軸"の値を適用しま

す。主軸はX>Y>U>Zの優先順位で選択されます。例えばYUZの組み合わせではY軸が主軸とな り ます。 (一般

的なDDA演算の主軸とは意味が異なり ます)

RMVL 1000 2000 0 0 /* X軸CW1000 Y軸CW2000パルス
RMVL 0 2000 2000 200 /* YUZ軸でパルス発生

RMVS
パルス（MPG-314専用） 軸独立パルス発生（相対座標移動）

■書　式

単軸指定

RMVS n pls

  n:軸指定予約定数 X_A～Z_A

  pls:パルス数

複数軸指定

RMVS x y u z

  x y u z:各軸パルス数

■解　説

パルス発生コマンドです。 速度、 加速度等はACCEL設定に従います。

入力方法には単軸指定、 複数軸指定の二通りあ り ます。 RMVSは直線補間はしません。

例1）

RMVS X_A 1000 /* X軸CW1000パルス
RMVS Y_A -1000 /* Y軸CCW1000パルス
RMVS 1000 -2000 0 0 /* X軸CW1000 Y軸CCW2000 

例2)

10  PG &H400
20  ACCEL X_A 5000 500 100
30  FEED X_A 0
35  INSET_314 X_A ALM_ON /*アラームonで有効
40  DO
50    RMVS X_A 10000 /*すぐ次の行へ
60    WAIT RR(X_A)==0 /*パルス終了待ち
70    TIME 1000
80    RMVS X_A -1000
90    WAIT RR(X_A)==0
100   TIME 100
110 LOOP

RMVT
パルス（MPG-314専用） 連続補間（相対座標移動）
■書　式

直線補間

RMVT n pls1 pls2

  n:軸指定予約定数。 2軸の論理和

  pls1,pls2:移動先の相対座標

円弧補間

RMVT n pls1 pls2 <CW,CCW> cnt1 cnt2

  n:軸指定予約定数。 2軸の論理和

  pls1,pls2:移動先の相対座標

  CW,CCW:円弧補間方向指定定数

  cnt1,cnt2:円中心

■解　説

連続移動補間コマンドです。 軌跡制御に使用します。

入力フォーマッ トはRMVT n pls1 pls2も し くはRMVT n pls1 pls2 <CW,CCW> cnt1 cnt2です。nは軸定数で二つ

の軸定数の和を使用します。 pls1,pls2は行き先の相対座標です。 CWも し くはCCW定数に続いて円中心を相

対位置で与える と目的位置まで円弧移動します。 連続補間の途中で軸定数を変更するこ とはできません。

10 PG &H400
15 CLRPOS
20 ACCEL 4000
25 HOUT X_A;DS_DACL
30   Aset=X_A|Y_A
40 RMVT Aset 800 0
50 RMVT Aset 0 1600 CCW 0 800
60 RMVT Aset 0 -1600 CCW 0 -800
70 RMVT Aset 800 0
80 HOUT X_A;EN_DACL

RMVT コマンドはパルス発生中に次のコマンドを設定するこ とができるのでコマンドの切れ目でパルス出

力が途絶えるこ とがあ り ません。 25 のステート メ ン トは通常のパルス発生で自動減速を有効と しているの

を解除します。 これを略すと最初のコマンドの到達時に減速し、 残りのパルス発生が全て最低速とな り ま

す。 逆に80は減速機能を有効にして最終位置到達時に減速停止させるためのものです。

800,1600

800,800

800,0 1600,00,0

RMVZ
パルス（MPG-68K互換） Z軸相対座標移動

■書　式

RMVZ z

  z:相対移動量 （パルス）

■解　説

Z軸の相対パルス発生です。 zで指定された値だけパルス発生します。座標値は移動した分だけ加算されます。

ROM
デバッグ フラッシュ ROMモード

■書　式

ROM

■解　説

フラ ッシュ ROM書き込みを有効にします。

RAM参照

RR
パルス（MPG-314専用） レジスタ読み取り

■書　式

RR0を読み取る

RR(0) 

パルス終了確認

RR(n) 

  n:軸指定予約定数。 指定軸がパルス終了する と0を返す

ボードを指定してRR0を読み取る

RR(adrs)  

  adrs:MPG-314ボードアドレス

軸とレジスタを指定して読み取る(RR0以外の読み取り )

RR(n,a) 

  n:軸指定予約定数

  a:レジスタ番号。 オフセッ ト アドレスを加えるこ とによ りMPGのアドレス指定が可能

■解　説

MCX314のレジスタを読み取るこ とで、 パルス発生状態やエラー内容を知るこ とができます。

WAIT RR(X_A)=0 /* X軸パルス発生終了待ち
A=RR(&H410)&&HF0 /* アドレス&H410のMPG-314のｴﾗｰｽﾃｰﾀｽを読み取る。
A=RR(Y_A,1) /* Y軸のRR1を読み取る。(終了ステータスの参照)
A=RR(Y_A,1+&H410 /* &H410のY軸のRR1を読み取る。(終了ステータスの参照)

RR()関数に状態定数 (EMG@,ALM@,LMTP@,LMTM@,ALLI@,SLMP@,SLMM@) や停止条件定数 (EMG_,ALM_,L

MTP_,LMTM_,IN3_,IN2_,IN1_,IN0_,ALLE_) をセッ トする こ とによ り各入力状態の検出、またパルス発生が停止

した原因を特定するこ とができます。@の付いた定数グループは、状態検出です。たとえばALM入力は論理

設定で、 ON/OFFいずれかの入力でアラーム受付とな り ますが、 この条件が成立している時にRR(X_A,ALM

@)で検出するこ とができます。 これに対してALM_はパルス発生がアラーム入力によって停止したかど うか

を判断するためのもので、 RR(X_A,ALM_) が 0 であればアラーム入力で停止していないという こ とにな り ま

す。RR(X_A,ALM@¦LMTP@)のよ うに状態検出は論理和をとって検出するこ とができますが、状態入力グルー

プと停止条件グループを混在させるこ とはできません。

リ ミ ッ トの検出

#INSET_314 X_A LMT_ON /*リミットオンで検出
#PRX RR(X_A,LMTP@|LMTM@) /*コネクタJ2 2,3番ﾋﾟﾝ入力
 000C /*両方on

※表示されるビッ ト と コネク タピン番号は整合しません。

RS
RS-232 バッファ表示

■書　式

RS n

  n:チャンネルナンバー

     n=0または2

■解　説

対応するチャンネルナンバーのRS-232の リ ングバッファの内容を参照します。 入出力と も表示されるので

実際にどのよ うなキャラ ク タが通信されたのかをモニタするこ とができます。MPC-684のCH2，CH0のバッ

ファは入出力と も256byteです。

予約定数

状態監視グループ

EMG@ EMGがイネーブル検出

ALM@ サーボアラーム出力ON検出

LMTP@ リ ミ ッ ト入力(+)検出

LMTM@ リ ミ ッ ト入力(-)検出

SLMP@ ソフ ト リ ミ ッ ト (+)検出

SLMM@ ソフ ト リ ミ ッ ト (-)検出

停止条件グループ

EMG_ 入力によって停止

ALM_ アラーム入力によって停止

LMTP_ リ ミ ッ ト入力によって停止

LMTM_ リ ミ ッ ト入力によって停止

IN3_ STOP条件のIN3によって停止

IN2_ STOP条件のIN2によって停止

IN1_ STOP条件のIN1によって停止

IN0_ STOP条件のIN0によって停止

RSE
RS-232 RS-232エラー

■書　式

RSE(n)

  n:チャンネルナンバー

     n=0または2

■解　説

RS-232受信時のエラーを返す関数です。外部機器と通信を行う時に通信状況のチェッ クが行えます。エラー

が発生した場合の初期化はCNFGコマンドで行います。

戻り値 0 正常終了

1 パリ ティーエラー

2 オーバーラン

4 フレーミ ングエラー

8 ブレークコード検出

*LOOP
  CNFG#2 "9600b7pns1NONE"
*LOOP1
  INPUT#2 A$
  SELECLT_CASE RSE(2)
    CASE 0 : PRINT A$
    GOTO *LOOP1
    CASE 1 : PRINT "ﾊﾟﾘﾃｨｰｴﾗｰ"
    CASE 2 : PRINT "ｵｰﾊﾞｰﾗﾝ"
    CASE 4 : PRINT "ﾌﾚｰﾐﾝｸﾞｴﾗｰ"
    CASE 8 : PRINT "ﾌﾞﾚｰｸｺｰﾄﾞ"
    CASE_ELSE : PRINT "ｴ!? ﾅﾆ?"
  END_SELECT
  PRINT#2 "TRY AGAIN!¥n"
  GOTO *LOOP

RSE
RS-232 CH1キャラクタ入力

■書　式

RSE(-1)

■解　説

CH1(プログラムポート )から1キャラ ク タ入力します。

  getchar=RSE(1)

RSV
タスク操作 セマフォ獲得

■書　式

RLS(n) 　 　 I/Oのビッ ト評価及びビッ トオフ （セマフォ解放）

RSV(n) 　 　 I/Oのビッ ト評価及びビッ トオン （セマフォ獲得）

  n: メモ リー I/Oナンバー

     -8192≦n≦-1

■解　説

これらはマルチタスク下でのセマフォの概念に対応するものです。 ADVFSCはセマフォ と して、 実在のI/O

も し くは内部のメモ リ上のデータ （メモ リ I/O） を使用できます。メモ リ I/Oはnを負の値と したものです。セ

マフォ とは、複数のタスクが1つの出力ポート などを共通に使用する場合にその使用状況が混乱しないよ う

にするためのものです。例えば、1つのRS-232ポートに対して2つのタスクがデータの出力をする場合キャラ

ク タが交互にだされた り してメ ッセージが意味をなさな くなる こ とがあ り ます。 例えば次のプログラムを

実行する と出力される メ ッセージはかなり期待外れのものです。

           time0=timer
           time=timer
        FORK 1 *aho
        DO
           WAIT time<>timer
           PRINT "abcdefg" "ABCDEFG"
              time=taimr
        LOOP
*aho
        DO
           WAIT time0<>timer
           PRINT "123456" "78901234"
              time0=taimr
        LOOP
#run
123456789abcdefgABC01234
DEFGH
123456789abcdefgABC01234
DEFGH
  [25]  ・・・・・プログラムブレーク
#

これに対して次のプログラムは期待通りに動作します。 これは、 メモ リ I/O-1をセマフォ とするこ とによっ

て交通整理がなされたためです。なぜ、通常のON/OFF/SW(n)の組み合わせを使用しないかという と、 この

セマフォの獲得でデッ ド ロ ッ クを防ぐには、 テス ト＆セッ ト とい う一括動作が必要なためです。

RSV(n)はこのためのものです。

【RSV】

           time0=timer
           time=timer
        FORK 1 *aho
        DO
           WAIT time<>timer
           WAIT RSV(-1)==0
           PRINT "abcdefg" "ABCDEFGH"
              dummy=RLS(-1)
              time=taimr
        LOOP
*aho
        DO
           WAIT time0<>timer
           WAIT RSV(-1)==0
           PRINT "123456" "78901234"
              dummy=RLS(-1)

              time0=taimr
        LOOP

■タスク間のセマフォの受け渡しについて

TASK Aがセマフォを解放して、必ず次にTASK Bがセマフォを獲得する、という と きはTASK AはSW()でメモ

リー I/Oを読みTASK Bの獲得を確認する。

TASK Aはほかのタスクのために1時的にセマフォを解放するが、 他にセマフォを獲得するタスクが無けれ

ば再びTASK Aがセマフォを獲得したい。そのためTASK Aがセマフォを解放した状態で必ずタスクを切り替

えて他のタスクを実行するよ うにしたい、 とい う場合はSWAPコマンドをいれます。 SWAPのかわりにTIME

を使っても有効ですが、 時間的に遅くな り ます。

RUN
デバッグ  プログラム実行

■書　式

RUN [n m]

■解　説

プログラムの実行。 mはブレークポイン トです。 ブレークポイン ト よ りの再実行にはCNTを使用します。

S_MBK
MBK-SH データエリア書き込み

■書　式

ワード(2バイ ト )書き込み

S_MBK outdata dtadr

  outdata: データ 0～65535

  dtadr: データアドレス 0～7899

符号付きワード(2バイ ト )書き込み

S_MBK outdata dtadr̃Int

  outdata: データ ±32767

  dtadr: データアドレス 0～7898

  Int: 予約定数

ロング(4バイ ト )書き込み

S_MBK outdata dtadr̃Lng

  outdata: データ 0～4294967295

  dtadr: データアドレス 0～7898

  Lng: 予約定数

    TASK A
WAIT RSV(-1)==0
       |
       |
dummy=RLS(-1)
WAIT SW(-1)==1
WAIT RSV(-1)==0

  TASK B
    |
    |
    |
WAIT RSV(-1)==0
    |
dummy=RLS(-1)

WAIT RSV(-1)==0
       |
       |
dummy=RLS(-1)
SWAP
WAIT RSV(-1)==0

*LOOP
     |
     |
IF RSV(-1)<>0 THEN : GOTO *LOOP : END_IF
     |
dummy=RLS(-1)
GOTO *LOOP

■解　説

MBK-SHのデータエ リ アにデータを書き込みます。

S_MBK 2 8 /* 2をDT0008(切り替え画面番号)に書き込む=2ﾍﾟｰｼﾞを表示
S_MBK 123456789 699~Lng   

/* 123456789をDT0699(下位)とDT0700(上位)にﾛﾝｸﾞ書き込み
S_MBK 1234 20 /* 1234をDT0020にワード書き込み
S_MBK -30000 21~Int /* -30000をDT0021に符号付きワード書き込み

S_MBK
MBK-SH 文字列転送

■書　式

S_MBK a$ dtadr count

  a$: 文字列

  dtadr: データアドレス 0～7899

  count: 文字数

■解　説

文字列をcount数分MBKの指定アド レスにブロッ ク転送します。count数が文字列より 大きい時は0で埋めます。

a$="123567890ABC"
#S_MBK a$ 102 10 /* 10文字=5ワード使用
#PRX MBK(102)
 3231 /* &H32='2'、&H31='1'
#PRX MBK(103)
 3533
#PRX MBK(104)
 3736
#PRX MBK(105)
 3938
#PRX MBK(106)
 4130 /* &H41='A'、&H30='0'
#PRX MBK(107)
 0000

SELECT_CASE <SLC>
制御文 多値分岐

■書　式

SELECT_CASE （省略形 SLC）

CASE

CASE_ELSE （省略形 CEL）

END_SELECT （省略形 ESL）

＊必ずCASE_ELSEも記述してください

■解　説

選択制御文 （多値分岐） 。 IF文などの条件文では、 そ うである ・ そ うでないという、 2つの可能性のみの場

合分けていますが、 実際の制御では1つの入力でいくつもの分岐が発生するこ とがあ り ます。例えば、 IN(n)

でデジスイ ッチの内容を読んでその内容に従っていくつかの違った動作をする場合です。例1ではデジスイ

ッチの分岐、例2は文字列の場合分けです。 この例の中にあるCASE文の後ろの: （コロン） はマルチステート

メン トのデリ ミ タ と してのものです。

例1）

   a=IN(0)&&HF+IN(0)&&HF0/16*10
SELECT_CASE a
   CASE 1 :GOTO *work1
   CASE 2 :GOTO *work2
   CASE 10 :GOTO *work10

   CASE_ELSE :GOTO *abort
END_SELECT

例2）

   INPUT "a$=" a$
   SELECT_CASE  a$
      CASE "123" : PRINT ”数字だよ”
      CASE "abc" : PRINT ”小文字だよ”
      CASE "ABC" : PRINT ”大文字だよ”
      CASE_ELSE  : PRINT ”良くわかんない”
      SELECT_CASE a$
         CASE "aho" :PRINT ”阿呆みたい”
         CASE "baka" :PRINT ”馬鹿みたい”
         CASE_ELSE  :PRINT ”さっぱりよねー”
      END_SELECT
   END_SELECT

SENSE_SW
I/O 入力検出で出力操作

■書　式

SENSE_SW inp outp

  inp: 検出入力ポート

  outp:操作出力ポート

■解　説

検出入力がonする と操作出力をon/offします。

SENSE_SW ?(-1)&?(192) (OFF_P,0)  
/* SW(-1)&SW(192)<>0ならば出力0をオフ。オンの場合はON_P

SENSE_SW ?(-1)&?(-2) (X_A;STP_I,&H410)
/* SW(-1)&SW(-2)<>0 ならば out (X_A;STP_I),&h410 
/*  つまりMPG-314 のアドレス&H410に対する停止命令

アクシ ョ ンは括弧でく く って表記しないと正し く動作しません。また、アクシ ョ ンは引数の許されるだけ記

述できます。 ON_P,OFF_Pは予約定数です。

SET
パルス（MPG-68K互換） インチング量設定

■書　式

SET n [x y u z]

  0≦n≦3

＊パラ メーターを省略する と現在の設定値が表示されます。

■解　説

このコマンドではティーチモードでのインチング量を既定します。 ティーチモードではX～Zのキーインに

よってJOG移動しますがこの移動量は0～3のキーによって選択するこ とが出来ます。 SETコマンドのnはこ

の0～3に対応しています。そして、 それぞれの移動量を定義するこ とが出来ます。 SETは次の例のよ うに引

き数がないと0～3のJOG量を表示します。 又、 番号のみ指定すればその対応するJOG量だけ表示します。

#SET 1 5 5 5 10
#SET
 dx= 5 dy= 5 du= 5 dz= 10
#SET
 dx= 10 dy= 10 du= 10 dz= 10
 dx= 5 dy= 5 du=5 dz= 10
 dx= 100 dy= 100 du=100 dz= 100
 dx= 500 dy= 500 du=500 dz= 500
#

SETF
コプロ演算 データ引渡し

■書　式

SETF fn1,fn2,fn3,..

  fn1,fn2,fn3,..:変数,定数

■解　説

変数、定数値をコプロに引き渡します。与えられた引数は順にコプロの不動小数点レジスタFP0,FP1..に引き

渡されます。

SETF 0 3 4 /* FP0に0,FP1に3,FP2に4をセットします。

SETIO
I/O 出力初期化

■書　式

SETIO [n m]

■解　説

出力の一括OFFに使用します。

SETP
パルス（MPG-68K互換） 点データ設定

■書　式

SETP [n x y u z]

  n:ポイン トナンバー

  x,y,u,z:座標値 （パルス）

     0≦n≦13000

＊パラ メーターを省略する と現在の設定値が表示されます。

■解　説

点データの設定です。通常はSETP 1 100 200 300 400とい う よ うに点番号と設定すべき値を与えます。FTMW

でサポート されている点データの保存はこのSETPコマンドの書式でファ イルが作成されます。SETPは入力

書式によって次の使い分けがあ り ます。

#SETP 1 1000 1000 1000 2000
#SETP 1
 X= 1000 Y= 1000 U= 1000 Z= 2000
#SETP 100
 X= 0 Y= 0 U= 0 Z= 0
#SETP 0
 X= 12 Y= 12 U= 12 Z= 12

最初の例はこのコマンドの目的である点データの設定です。 次に点番号だけ与えればその点番号のデータ

を表示します。さ らに引き数が何も無なければ現在位置の変更とな り ます。又、SETPでは点データのコピー

もできます。

SETP 3 P(5) ･･･P(5)のﾃﾞｰﾀをP(3)にｺﾋﾟｰ
SETP 10 PL2(5) ･･･ﾊﾟﾚｯﾄの点PL2(5)をP(10)にｺﾋﾟｰ

SETPOS
パルス（MPG-68K互換） 現在値変更

■書　式

SETPOS x y u z

  x,y,u,z:座標値 （パルス）

＊パラ メーターを省略する と現在の設定値が表示されます。

■解　説

現在位置の変更です。現在位置がx y u zで指定された値になり ます。引き数を与えないで実行する と現在位

置が表示されます。

SFTL
演算 配列変数ローテート

■書　式

SFTL ary(n)

SFTR ary(n)

  ary:配列名

■解　説

DINで配列宣言された配列要素に対してデータのローテート を実施します。ary(n)のaryは配列名、nは数値で

いくつまでのデータをローテートするかを指定します。 尚、 SFTL/SFTRはx (),y(), u(), z()には有効ではあ り

ません。 例えば、 SFTL aho(3)では次のよ うにな り ます。

aho(0) <- aho(1)
aho(3) -> aho(2)

SFTL、 SFTRは閉じたシフ ト を行います。

10        DIM aho(10)
20        FOR i=0 TO 10
30            aho(i)=i+10
40        NEXT i
45        SFTL aho(3)
50        FOR i=0 TO 10
60          PRINT "i=" i "aho=" aho(i)
70        NEXT i
#
RUN
i=0 aho=11
i=1 aho=12
i=2 aho=13
i=3 aho=10
i=4 aho=14
i=5 aho=15
i=6 aho=16
i=7 aho=17
i=8 aho=18
i=9 aho=19
i=10 aho=20

SFTR
演算 配列変数ローテート

SFTL参照

SHMZ
パルス（MPG-68K互換） 原点復帰設定

SHOM参照

SHOM
パルス（MPG-68K互換） 原点復帰設定

■書　式

SHMZ pat spd

SHOM pat spd

  pat:原点復帰出力パターン

  spd:原点復帰スピード （pps）

＊パラ メーターを省略する と現在の設定値が表示されます。

■解　説

原点復帰の動作モード設定です。SHMZはZ軸原点復帰、SHOMはXYUの3軸原点復帰モード設定です。spdは

ppsで表現されるスピード とな り ます。patは原点復帰のパルス方向を定めます。patの各ビッ トが原点復帰時

に出力されるパルスパターンとな り ます。 patと原点復帰出力パルスの関係は次の表の通りです。

例えばXYZそれぞれCCW方向にパルスを出力して原点復帰する場合はSHOM &H2A 1000と します。 1000は

1kppsを表します。 SHOM,SHMZは引き数無しで実行する と現在設定された値を表示します。

X,Y,U軸がCCW方向に1000ppsで原点復帰。 Z軸はCCW方向に500ppsで原点復帰。

SHOM &H2A 1000
SHMZ &H2 500

※注意

SHOMはZCW,ZCCWについても有効です。 SHMZを設定してもその後にSHOMを実行する とSHOMの設定内

容が有効になり ます。 SHOMとSHMZを併用する場合はSHOM、 SHMZの順に記述して下さい。

SHMZ &H2 1000 <--Z軸の設定を行っているが...
SHOM &H2A 1000 <--このSHOMでZ軸の設定がｸﾘｱｰされるのでこの順番はダメです。

SIN
演算 三角関数

COS参照

MSB 7 6 5 4 3 2 1 0 LSB

SHMZ UCCW UCW YCCW YCW XCCW XCW

SHOM - - - - - - ZCCW ZCW

SLOW
RS-232 CH1キャラクタ送信間隔

■書　式
SLOW n
＊パラ メーターを省略する と現在の設定値が表示されます。

■解　説
CH1のキャラ ク タ送出スピードの低速化。 パワーオン リセッ ト時にn=0に設定されます。

SQ
演算 自乗

■書　式
SQ(n)
  n:定数， 変数

■解　説

nの自乗を返します。

#A=4
#B=SQ(A)
#PRINT B
16

SQR
演算 平方根

■書　式

SQR(n)
  n:定数， 変数

■解　説
nの平方根を返します。

PRINT SQR(SQ(100)) これは100となります。

小数点以下は切り捨てます。

ST_REG
MPG-3202 X3202レジシタ書き込み

■書　式

ST_REG reg data
  reg:X3202のアドレス とレジスタ ・ カウンタセレク ト コードの和
  data:設定データ

■解　説
MPG-3202のパルス発生IC 「X3202」 のレジスタ設定を行います。
MPG-3202関係コマンド ・ 関数:CMND,REG,REG3,ST_REG
参照: 「MPG-3202 製品別マニュアル」　 ･･･web参照または営業係へ請求下さい。

STOP
パルス（MPG-314専用) 条件停止

■書　式

STOP n cond

  n:軸指定予約定数 X_A～Z_A

  cond:入力指定 

       INx_ON～INx_OFFの論理和 または STP_I,STP_D

■解　説

nを論理和で指定すれば指定軸の同時停止が可能です。

condにINxを指定した場合、 STOP n VOIDで解除するまで有効とな り ます。

STP_I,STP_Dをセッ トする と発生中のパルスを停止します。

STOP Z_A IN0_ON|IN3_ON/* Z軸IN0オンIN3オンになったら停止。
/* 一度設定すると STOP Z_A 0実行まで解除されない。

STOP X_A|Y_A STP_D /* XY軸同時減速停止。

STOP
パルス（MPG-68K互換） パルス発生停止

■書　式

STOP m n

  m:停止パターン

  1＝減速停止(ACCELコマンドで設定された加減速テーブルに従い停止します。）

  2＝即停止

  3＝パルス発生終了時にタスクをPAUSEします

  (STOP 3はREV 1.2以上でQ_PAUSEモード指定時のみ有効です。 )

  n:タスクナンバー

     0≦n≦31

＊どの場合も座標値は整合します。

■解　説

マルチタスクで使用します。 パルス発生中のタスクに他のタスクから停止をかけるものです。 STOPでは減

速停止と即停止の選択が出来ます。 STOP 1で減速停止、 STOP 2で即停止です。

FORK 1 *aho
    MOVE 100000 10000 1000
    END
*aho
    IF SW(0)==1 THEN
       STOP 1 0
    END_IF
    IF BSY(0)<>0 THEN
       END
    END_IF
    GOTO *aho

この例では移動直前に停止タスクを起動しています。 停止タスクはパルス動作の監視も併せて行っており

パルス停止後監視タスク も停止するよ うになっており ます。 こ こでSTOPの引き数は減速モードの停止と タ

スク番号となっています。

※MOVE、MOVZ、JUMP、GOにはQ_PAUSEモードでの停止がシンプルで合理的（お勧め）です→Q_PAUSE参照

■次のプログラムではSTOPコマンドを発行してBSY()関数でパルス停止を確認、タスクをQUITしています。

パルス発生中のタスクをいきな り QUITしてもパルスは止ま り ません。 パルスの停止にはこの他にDS_PGコ

マンドがあ り ます。

10       FORK 1 *TASK1
20       WAIT SW(192)==0 AND SW(193)==0
30       WAIT SW(192)==1 OR SW(193)==1
40       IF SW(192)==1 THEN : STOP 1 1 : END_IF  <--減速停止
50       IF SW(193)==1 THEN : STOP 2 1 : END_IF　<--即停止
60       WAIT BSY(1)<>0
70       PRINT "SW(192)=" SW(192) "SW(193)=" SW(193) "BSY=" BSY(1)
80       QUIT 1
90       END
100    *TASK1
110      PG &HE0
120      ACCEL 10000
130      FEED 0
140      DO
150        MOVE 0 0 0
160        WAIT BSY(-1)==1　<--正常停止でない時ﾌﾟﾛｸﾞﾗﾑは先へ進まない
170        MOVE 50000 0 0
180        WAIT BSY(-1)==1
190      LOOP
#RUN
SW(192)=1 SW(193)=0 BSY=256
#RUN
SW(192)=0 SW(193)=1 BSY=512

■STOPコマンドによるパルス停止例

10       DO
20         FORK 1 *TASK1
30         WAIT SW(192)==0
40         WAIT SW(192)==1
50         PRINT "STOP"
60         STOP 1 1 <--STOPして
70         WAIT BSY(1)<>0 <--停止を確認して
80         PRINT BSY(1)
90         QUIT 1 <--QUITする
100        WAIT SW(193)==1
110      LOOP
120    *TASK1
130      DO
140        MOVE 100000 0 0
145        WAIT BSY(-1)==1 <--正常停止以外はここで停止する。この様な連続した
150        MOVE 0 0 0    LOOPの場合、もしここにWAIT BSY()がないとTASK 0か
155        WAIT BSY(-1)==1    らSTOPされてMOVEｺﾏﾝﾄﾞから抜けでても、すぐに次の
160        LOOP    MOVEｺﾏﾝﾄﾞを実行してしまいﾊﾟﾙｽが止まらない。

■STOPコマンドで停止してBSYで確認しない場合 （正常に動作しない）

10       DO
20         FORK 1 *TASK1
30         WAIT SW(192)==0
40         WAIT SW(192)==1
50         PRINT "STOP"
60         STOP 1 1
90         QUIT 1 <--STOPのあといきなりQUITしているのでﾊﾟﾙｽが止まら
100        WAIT SW(193)==1    なくなる
110      LOOP
120    *TASK1
130      DO
140        MOVE 100000 0 0 <--MPGはﾊﾟﾙｽを出し切るまで止まらない
145        WAIT BSY(-1)==1    そしてMPGを管理していたﾀｽｸが止まってしまったの
150        MOVE 0 0 0    で8251ｴﾗｰとなる
155        WAIT BSY(-1)==1
160      LOOP

#RUN
STOP

   タスク停止 *1 [140]MPG通信エラ－8251側、STOPしないでQUITしてませんか？

   ↑ﾀｽｸ1を再FORKしてもMPGがｴﾗｰをおこしてしるので動かない。

■DS_PGによるパルス発生の停止

10       DO
15         EN_PG 1
20         FORK 1 *TASK1
30         WAIT SW(192)==0
40         WAIT SW(192)==1
50         PRINT "STOP"
60         DS_PG 1 <--ﾊﾟﾙｽ発生禁止
70         WAIT BSY(1)<>0
80         PRINT BSY(1)
90         QUIT 1
100        WAIT SW(193)==1
110      LOOP
120    *TASK1
130      DO
140        MOVE 100000 0 0 <--MOVE実行中にDS_PGされるとこのﾀｽｸはMOVEｺﾏﾝﾄﾞ中で
145        TIME 1000    ﾊﾝｸﾞｱｯﾌﾟする。
150        MOVE 0 0 0    ﾊﾟﾙｽ発生以外のｺﾏﾝﾄﾞ実行中にDS_PGされるとこのﾀｽｸ
155        TIME 1000    は動くがﾊﾟﾙｽは出ない。
160      LOOP
#run
STOP
1024
STOP
1024
STOP
1024

■次のプログラムはタスク1でパルス発生中にタスク1を再FORKしています。

このプログラムはRMOVですがこのほかMOVE、JUMP、GO、ACCELなどパルス発生関係コマンド でも同じです。

LIST 0
10       PG &HE0 1
20       DO 
30         FORK 1 *TASK1
40         TIME 1000
50       LOOP 
60     *TASK1
70       DO 
80         RMOV 100000 100000 1000000
90       LOOP 
#RUN
    タスク停止 *1 [80]MPG通信エラ－です68301側
    タスク停止 *1 [80]MPG通信エラ－です68301側
    *0   [40]
    *1   [80]

■次のプログラムはパルス発生中のタスクをSTOPせずにQUITしています。

STOPせずにQUITする とパルスは最後まで出ます （この場合はX,Y,Uと も1000パルス） 。 しかしMPCとMPG

の内部の通信が途切れてしま うので再びFORKしてパルスを発生しよ う と してもエラーになり ます。

LIST 0
10       PG &HE0 1
20       DO 
30         FORK 1 *TASK1
40         TIME 50
50         QUIT 1
60         TIME 2000
70       LOOP 

80       END 
90     *TASK1
100      DO 
110        RMOV 1000 1000 1000
120      LOOP 
#RUN
    タスク停止 *1 [110]MPG通信エラ－8251側、STOPしないでQUITしてませんか？
    *0   [60]
    *1  停止　

STPS
パルス（MPG-314専用） 現在位置指定

■書　式

STPS  n pos

  n:軸指定予約定数 X_A～Z_A

  pos:指定する位置

STPS x y u z

  x y u z:各軸の指定する位置。 指定無=VOID

■解　説

前者は選択した軸のみ現在値を設定します。 後者はVOID定数以外の軸の現在位置を設定します。 パルス発

生中の軸に値を設定しよ う とする と停止待ちとな り ます。

STPS U_A 1000 /* U軸の現在位置を1000にする
STPS VOID 1000 VOID 2000 /* YZ軸の現在位置を設定

STPS
パルス（MPG-314専用） カウンタ設定

■書　式

STPS n c

  n:軸選択予約定数 X_C～Z_C

  c:カウンタ初期値

■解　説

カウンタ値は X(-1)、CP -1などで読み取れます。カウンタを使用するにはMPG-314のOC5～OC8ソケッ トに

TLP2630を実装します。

STR$
文字列  数値から文字列に変換

■書　式

STR$(n)

■解　説

nの数字列を与えます。

#a=1000
#a$="BAKA"+str$(a)+"AHO"
#print a$
BAKA1000AHO

STRCPY
文字列 文字列の複写

■書　式

STRCPY s1$ s2$ [n m]

■解　説

文字列のコピー （制御機能付き）

STRCPY A$ B$ n m A$のn文字目からm文字をB$にコピー

nは0文字目から数える。n及びmはそれぞれ省略するこ とが出来ます。nもmも省略された場合には、 s1$よ り

s2$へすべての文字がコピーされます。nのみ与えたと きはn文字目からすべてとな り ます。次の3つはすべて

同じ働きをします。

B$=A$
STRCPY A$ B$
STRCPY A$ B$ 0

mを省略せずにし指定する と コピーされる文字の数はmによって定められます。 次の例では文字の長さを調

べる関数、 LEN()を使用して、 いろいろなパターンで文字のコピーを実施しています。

例1）

10          a$="01234567890"
20        PRINT LEN(a$)
30        FOR i=0 TO LEN(a$)
40          STRCPY a$ b$ i
50          PRINT b$
60        NEXT i
RUN
11
01234567890
1234567890
234567890
34567890
4567890
567890
67890
7890
890
90
0

例2）

10          a$=time$ /*time$は予約変数です。
20        STRCPY a$ h$ 0 2
30        STRCPY a$ m$ 3 2
40        STRCPY a$ s$ 6 2
50        PRINT h$ "時" m$ "分" s$ "秒"
#
RUN
02時36分36秒

   n=2 開始位置

ABCDEFGHIJK

　 　m=7 文字数

SV_M
MBK-SH メモリ一括コピー (MBK→MPC)

■書　式

SV_M array mbktop count [opt]

  array:MPC点データの先頭(XYUZ(1～13000))、 配列変数の先頭

  mbktop:MBKメモ リ先頭アドレス。 20≦mbktop

  count:コピー数

  opt: 4=ロングワードで1アドレスおき、 Lng=ロングワードで連続

■解　説

配列からMBKのデータエリ アにコピーします。パラメ ータはアドレスの使用範囲を超えねよう に設定して下さい。

SV_M X(9020) 20 683 /* DT20から点データX(9020)へ683個ワードコピー
SV_M test(10) 500 10 /* DT500から配列変数test(10)へ10個ワードコピー
SV_M X(9020) 20 342 4 /* DT20から点データX(9020)へ342個ロングワードコピー
SV_M X(9000) 500 5 Lng /* DT500から点データX(9000)へ5個ロングコピー (連続)
SV_M test(0) 500 5 Lng /* DT500から配列変数test(0)へ5個ロングコピー (連続)

SW
I/O ビット入力

■書　式

SW(n)

  n:ポー トナンバー

■解　説

I/Oポートのビッ ト入力。 I/Oを2度読みし結果が一致するまで読み続けます。 これは、 チャタやノ イズによ

る誤読み取り を防ぐためのものです。 返す値は1がON状態、 OFFで0とな り ます。 ポート番号に出力ポート

の状態を得るこ と もできます。

SWAP
タスク操作 実行権の放棄

■書　式

SWAP

■解　説

タス クスワ ップ。 他のタス クに実行権を譲り ます。 このコマンドはプログラム実行の高速化に必要です。

ADVFSCのマルチタスクはタイムシュア リ ングによるものですが、これはプログラムの実行効率を著し く低

下させます。マルチタスクでタスクを5個実行すれば、1つのタスクに割り当てられる時間は1/5とな り ます。

つま りプログラムの実行速度は1/5となってしまいます。このため、ADVFSCではタイマー待ちの状態では不

要なタスクをス リープさせた り WAIT などのよ う な条件待ちでは実行中のタスクを強制的に切り換えてプ

ログラムの実行効率を高めています。つま り、プログラムの実行中にあま り速度を要求されない仕事やレベ

ルの低い仕事をさせないよ うにしているわけです。

*WAIT
         IF A=0 THEN
            GOTO *WAIT
         END_IF

この例では、変数Aの状態が1となるのを待っています。こ う した仕事をポーリ ングと言いますが、これに他

の仕事と同じ時間を与えるのは効率上好まし くないこ とです。 なぜならAを1にセッ トするのは他のタスク

ですから、 条件不成立の場合にこれ以上このタスクを実行するこ とは意味があ り ません。 こ う した場合に

SWAPを用います。

*WAIT
         IF A=0 THEN
            SWAP
            GOTO *WAIT
         END_IF

これでポーリ ングして条件不成立ごとに、他のタスクに実行権が引き渡され、無駄なくAが1になるこ とを待

つこ とが出来ます。 なお、 これと等価のコマンドはWAITです。

         WAIT　A=1

WAITには条件不成立の場合のSWAPが組み込まれています。 タイマー及びタイマーの組み込まれている関

数などではSWAPと等価のこ とが起こっています。

SWP
演算 上下位バイト交換

■書　式

SWP(n)

■解　説

バイ ト スワ ップ。 上位バイ ト と下位バイ ト を入れ換えます。 684と8086では上位下位のアドレス順位が反対

です。

#prx swp (&hff00)
 00FF
#prx swp (&h00ff)
 FF00
#

SYSCLK
タイマー システムクロック

■書　式

SYSCLK

■解　説

パワーオンよ り5msec毎に+1される変数です。

FOR～NEXT 10000回の所要時間を計測します。

5           SYSCLK=0
10        FOR I=1 TO 10000
30        NEXT I
40        PRINT SYSCLK
RUN
94 ←94×5=470msec

TAIL
編集 文番号の最大値

■書　式

TAIL

■解　説

文番号の最大値を表示します。プログラムコーディ ング中の後ろにプログラムをつけ加える場合に使用します。

LIST 0
10   FOR i=0 TO 47
20     ON i : TIME 50 : OFF i
30   NEXT i
#tail
30
#

TAN
演算 三角関数

COS参照

TASK
デバッグ タスク状態表示

■書　式

TASK(n)

  n:タスクナンバー

      0≦n≦31

■解　説

nはタスク番号で、指定されたタスクの状態を知らせる関数です。この値が0であればそのタスクは実行中で

あるこ と になり ます。0以外の正の数であればそのタスクは休止中であるこ と を意味します。この数の意味は

タイマー値です。1000であればあと 1秒は休止しているこ と になり ます。TASK(n)の値の意味は次の通り です。

戻り値  0 実行中

-1 未使用

-2 QUIT状態

-3 PAUSE状態

TASKN
タスク操作 タスク番号取得

■書　式

TASKN

■解　説

タスクナンバーを知る変数です。

10        FORK 13 *JOB
20        DO  : LOOP
30      *JOB
40        PRINT "I am " TASKN
50        END
RUN
I am 13

TEACH
パルス（MPG-68K互換） ティーチングモード

■書　式

TEACH    （省略形 T）

■解　説

TEACHコマンドはTで実行できます。ティーチモードでは点データの教示、及び調整時に必要なコマンドが

1文字キーインで操作できるよ うになっています。

x

X

y

Y

u

U

z

Z

XCW方向にインチング移動 

XCCW方向にインチング移動

YCW方向にインチング移動

YCCW方向にインチング移動

UCW方向にインチング移動

UCCW方向にインチング移動

ZCW方向にインチング移動

ZCCW方向にインチング移動

0,1,2,3

P,p

L,l

O,o

F,f

S,s

H,h

J,j

A,a

R,r

インチング量の選択0～3(SETコマンド参照)

点データの指定

LIMZ設定(JUMP,JMPZ参照)

出力ポートのON

出力ポートのOFF

I/Oの値表示

原点復帰入力ポートの表示(HPT(0)参照)

指定点へJUMP移動

指定点へJMPZ移動

RANGE設定(RANGEコマンド参照)

TAB

+

-

MPG切り替え(タスク番号インク リ メン ト )

MPG切り替え(タスク番号インク リ メン ト )

MPG切り替え(タスク番号デク リ メ ン ト )

Q,q テ ィーチモードの終了

THEN
制御文 条件分岐

IF参照

TIME
タイマー 時間待ち

■書　式

TIME

■解　説

時間待ちです。 単位はmsecで指定しますが実態は5msec単位です。 TIME 3はTIME 0と同じ意味になり ます。

TIME 1000 1秒停止、この間タスクはスリープです。

time$
カレンダー 時刻文字列取得
■書　式

time$

■解　説

MBK-SHまたはMBK-RSとデジタルGPが接続されている と、GPの時刻が文字列と して入り ます。小文字で書

いて下さい。

timer
カレンダー 時間取得

■書　式

timer

■解　説

時間を全て秒になおした数です。 00:01:00の場合timerは60とな り ます。

#PR time$
00:01:08
#PR timer
68

time$,CLK参照

TMON
メンテナンス タスクモニタ

■書　式

TMON

■解　説

各タスクのスタ ッ ク とプログラムカウンタの値を表示します。使用されていないタスクではFFFFFFFFが表

示されます。

TMOUT
タイマー 入力時間設定

■書　式

TMOUT n

  n:タ イマー値 （sec）

■解　説

TMOUTは関数WS0(),WS1()のタイムアウ トの設定です。 WS0,WS1は指定されたポートがONも し くはOFFに

なる迄ポーリ ングする条件待関数です。 ポーリ ング最大時間をTMOUTによって定めるこ とができ、 返され

る値はポートの状態ではなく、 時間内に条件が整ったかど うかです。 時間内に条件が充たされれば0、 そ う

でなければ1を返します。nの単位は秒です。TMOUTの時間は全タスク中最後に実行されたTMOUTが有効に

なり ます。

TOFF
デバッグ トレースモード

TON参照

TON
デバッグ トレースモード

■書　式

TON ト レースモード

TOFF ト レースモード解除

■解　説

ト レースモードでは実行中のプログラムの文番号を画面に表示します。 タスク0のみ有効です。

10     FOR i=0 TO 47
20       ON i : TIME 50 : OFF i
30     NEXT i
#ton
#run
[10]
[20]
[20]
[20]
[30]

U
パルス（MPG-314専用） カウンタ値読み込み

X参照

U
パルス（MPG-68K互換） U軸点データ

X参照

UNTIL
制御文 条件文

■書　式

DO～LOOP UNTIL 条件式

MOVE x y u UNTIL 条件式

RMOV x y u UNTIL 条件式

PULSE axis acnt [t1 t2] UNTIL 条件式

■解　説

DO～LOOPやパルス発生コマンドの繰り返し条件、パルス停止条件を与えるこ とができます。条件式には普

通のI/O、 メモ リー I/O、 変数が使えます。

■UNTIL使用上の注意
DO～LOOPの場合は論理結合ができます。

10   DO
20     ON 0
30     TIME 100
40     OFF 0
50     TIME 100
60     LOOP UNTIL A==1 AND B==1
70   END

パルス発生コマンドの停止条件では論理結合が出来ません。2つ以上の条件を結合したいと きは別のタスク

で結合します。また一致の条件を入力する場合は必ず” ==” と入力して下さい。DO～LOOPの場合は” =” と

入力してもMPCが自動的に” ==” と変更しますがパルス発生コマンドの場合は行われません。

10       OFF -1
20       FORK 1 *TASK1
30       TIME 2000
40         IF SW(192)==1 AND SW(193)==1 THEN : ON -1 : END_IF
50       END
60     *TASK1
70       DO
80         CLRPOS
90         MOVE 10000 10000 10000 UNTIL SW(-1)==1 <--ｲｺｰﾙは必ず2つ入力
100        IF SW(-1)==1 THEN : GOTO *END : END_IF
110        MOVE 0 0 0 UNTIL SW(-1)==1
120        IF SW(-1)==1 THEN : GOTO *END : END_IF
130      LOOP
140    *END
150      PRINT "ｵﾜﾘ"
#

■UNTILでのパルス停止は減速停止になり ます。

UNTIL使用例

10         B$="" :   A$="" <--文字列初期化
20       DO UNTIL A$=="*" <--"*"が来るとDO LOOPを終了
30           B$=B$+A$ <--文字列の合成
40           A$=INPUT$(1) <--RS-232　CH1から1文字入力
50       LOOP
60       PRINT B$
#RUN
ABCD <-ﾀｰﾐﾅﾙから"ABCD*"と入力した後
#

_VAR
制御文 引数からの受け取り

■書　式

_VAR var1 var2 - -

  var:変数

■解　説

GOSUBの引数を受け取り ます。

GOSUB参照

VAL
文字列 数字文字列から数値

■書　式

VAL(a$)

  a$:数字文字列 （ﾍｷｻ表記も可能）

■解　説

数字文字列a$を数値に変換します。

#a$="1000"
#a=VAL(a$)+2000
#print a
3000

ヘキサ表記文字列の変換例

#a$="12BC"
#a$="&H"+a$
#a=val(a$)
#pr a
4796

SW(-1)==1

減速停止

ACCELコマンドで設定

した加減速テーブル

VER
メンテナンス 改版データの表示

■書　式

VER

■解　説

搭載されているROMの版数を表示します。 ターミナルソフ トのオープニングにも表示されます。

MPC-684 ADVFSC(r)e REV-3.82k ････ｱﾄﾞﾊﾞﾝｽﾄﾞﾌｪｰｼﾞｯｸ版数
 BASIC like + multi tasking ････ﾏﾙﾁﾀｽｸﾍﾞｰｼｯｸﾗｲｸｲﾝﾀﾌﾟﾘﾀ
 Created by ACCEL Co.'~2001 ････（株）アクセル製

VER$
文字列 改版データの取得

■書　式

VER$   (必ずラージアルファベッ ト )

■解　説

バージ ョ ンデータが入っている予約変数です。

#a$=VER$
#pr a$
MPC-684 ADVFSC(r)e REV-3.82k
 BASIC like + multi tasking
 Created by ACCEL Co.'~2001

VLIST
編集 プログラムリファレンスの表示

■書　式

VLIST

■解　説

プログラムで使用しているラベル、 変数などの情報を表示します。

WAIT
制御文 条件待ち

■書　式

WAIT 条件式

■解　説

条件待ち。マルチタスク下での効率のよいポーリ ングを実行します。 タスク処理の効率についてはSWAPを

参照して下さい。

① WAIT P_SW(i01)<>1
② WAIT A==1
③ time=timer
   WAIT timer<>time

①の例はi01で指定された入力ポートが0になるのを待ち②の例ではAが1になるのを待ちます。 ③の例では

RTCを使用した1秒タイマーとなっています。

WEND
制御文 条件ループ

WHILE参照

WHILE
制御文 条件ループ

■書　式

WHILE 条件式～WEND

■解　説

条件待ち制御文。WHILEはDO～LOOPのサブセッ トです。この場合は条件が成立している間という意味を特

定しています。プログラムは読む人に意味を分かりやすく記述するのが大切ですが、このよ う な意味のはっ

き り した制御文を使用するこ とは、 この目的のためです。

a=1
     b=0
WHILE a<4
     DO UNTIL b>4
        PRINT a""b=b+1
     LOOP
     a=a+1
     b=0
WEND

WIR
バスアクセス ワード読み取り

■書　式

WIR(adrs)

  0≦adrs≦&HFFFF （偶数のみ）

■解　説

アドレスadrsのワード読み込みです。こ こでのアドレスは98バス側からみたアドレスであ り、68000側と奇数

と偶数が入れ替えられます。

WOW
バスアクセス ワード書き込み

■書　式

WOW n adrs

  0≦n≦&HFFFF

  0≦adrs≦&HFFFF （偶数のみ）

■解　説

アドレスadrsのワード書き込みです。 WIRと同様98バス側からみたアドレス とな り ます。

WS0
I/O タイムアウト付き入力

■書　式

WS0(n)

WS1(n)

  n:入力ポート番号

■解　説

WS0はnで指定されたポートの値が0(OFF)になるのを待ちます。WS1(n)は1(ON)になるのを待ちます。待ち時

間はTMOUTで設定するこ とができ、その時間を越える と1を返します。その時間内に条件が成立した場合は

0を返します。

TMOUT 5
   n=192
IF WS1(n)==1 THEN ’sw(n)が5秒以内にオンに
   GOTO *aho ’ならない時はgoto *aho
END_IF
PRINT　”スイッチ” n ”はオンです”
END
*aho
PRINT　”タイムアウト”

また、 WS1()、 WS0()は条件のAND、 ORもできます。 しかしMPC-684では1行 （命令） が8文字列以内とい制

約があ り ますから、論理結合も3つまでです。次のプログラムは出力0<->入力192,1<->193,2<->194を接続し

て実行しました。

①この場合は192,193の両方がタイムアウ トにならなければ*AHOには行きません。

10   *MAIN
20     TMOUT 1
30     IF WS1(192)==1 AND WS1(193)==1 THEN
40       GOTO *AHO
50     END_IF 
60     PRINT "Ok"
70     END 
80   *AHO
90     PRINT "Time out!!"
#OFF 0
#OFF 1
#RUN
Time out!!
#ON 0
#OFF 1
#RUN
Ok
#OFF 0
#ON 1
#RUN
Ok
#ON 0
#ON 1
#RUN
Ok
#

②どちらか、 または両方がタイムアウ トする と*AHOに行きます。

30     IF WS1(192)==1 OR WS1(193)==1 THEN

③1つ以上タイムアウ トすれば*AHOに行きます。

30     IF WS1(192)==1 OR WS1(193)==1 OR WS1(194)==1 THEN

WS1
I/O タイムアウト付き入力

WS0参照

X
パルス（MPG-314専用） カウンタ値読み込み

■書　式

X(-1)

Y(-1)

U(-1)

Z(-1)

■解　説

MPG-314に対して-1を与える とカウンタ値を返します。

カウンタは4バイ ト長(±2147483647)です。

#PG &h400
#STPS X_C 2147483647
#PR X(-1)
2147483647

X
パルス（MPG-68K互換） X軸点データ

■書　式

X(n)

Y(n)

U(n)

Z(n)

  n:ポイン トナンバー

     1≦n≦13000

■解　説

予約配列です。点データP(n)の各座標成分は予約配列と して表現されます。nが0の時は現在位置を表します

が、代入の場合は現在位置の変更とはな り ません。ホス トCPUのパルス発生モードの現在位置が変更される

だけで、MPG側は変更されません。点データ とかかわりな く独立した配列と して使用できます。又、 NEWP

でク リ アされます。

#PRINT X(0) ’現在位置のX成分表示
 3764
#PRINT Y(99) ’P(99)のY成分表示
 100
#Y(99)=5000
#PRINT Y(99)
 5000
#

Y
パルス（MPG-314専用）  カウンタ値読み込み

X参照

Y
パルス（MPG-68K互換） Y軸点データ

X参照

Z
パルス（MPG-314専用） カウンタ値読み込み

X参照

Z
パルス（MPG-68K互換） Z軸点データ

X参照

